
Sissejuhatus VBAsse 
Rakenduste arendussüsteem VBA – Visual Basic for Applications, põhineb üldotstarbelisel programmeeri-

missüsteemil Visual Basic (VB), mis leiab laialdast kasutamist eriti rakendustarkvara ja veebirakenduste 

loomisel. Kasutusulatuselt on ta umbes viiendal–kuuendal kohal. Temast eespool on vaid Java ja C-pere 

keeled (C, C++, C#), mis on eeskätt süsteemprogrammeerimise keeled. Enam-vähem samas ulatuses on PHP 

ja Pythoni kasutamine. Basicu variante ja dialekte on üsna palju. Neist nn tööstusliku tarkvara loomiseks 

leiab aga kasutamist peamiselt Visual Basic ja selle alusel loodud VBA. 

VBA esindab dokumendipõhiste rakenduste loomise ja arendamise vahendit. Kui pidada silmas 

programmeerimiskeelt, siis on see VB-s ja VBA-s praktiliselt sama. Kõik põhilaused, andmete käsitlus, 

andmete liigid ja tüübid on mõlemas süsteemis ühesugused. Erinevus seisneb peamiselt selles, et VBA on 

mõeldud kasutamiseks koos mõne rakendustarkvaraga ning selle abil loodud rakendused saavad töötada 

ainult vastava programmi keskkonnas. Visual Basicu näol on aga tegemist üldotstarbelise 

programmeerimissüsteemiga ning selle abil luuakse autonoomseid rakendusi. Samal ajal saab VB-d 

kasutada ka dokumendipõhiste rakenduste loomisel.  

VBA peamised rakendusvaldkonnad on üldotstarbelised rakendusprogrammid: tabeli- ja tekstitöötluse 

programmid, esitlus- ja graafikasüsteemid, andmebaasirakendused. VBA leiab laialdast kasutust eeskätt 

Microsofti toodetes – Excel, Word, PowerPoint, Access, Visio, Project. VBA on kasutusel ka sellistes 

süsteemides nagu Corel Office, Corel Draw, AutoCAD, Imagineer jm. VBAga lähedased arendusvahendid on 

ka süsteemides OpenOffice ja IBM SmartSuite ja paljudes teistes. 

Siin vaadeldakse VBA kasutamist MS Exceli keskkonnas. Tahaks rõhutada, et eesmärgiks ei ole Exceli 

programmeerimine, vaid Scratchis omandatud programmeerimise oskuste ja teadmiste süvendamine ja 

laiendamine tekstipõhiste, objektorienteeritud programmeerimist toetavate vahendite abil. VBA ja Excel 

pakuvad selleks suurepäraseid võimalusi. Olgu märgitud, et VBA võeti esmakordselt kasutusele just Excelis 

(1995). 

Võrreldes enamike teiste programmeerimiskeeltega on VBAl (võiks öelda Visual Basicul) oluliselt lihtsam 

struktuur ja süntaks ning samal ajal on siin rikkalik valik objektorienteeritud vahendeid töötamiseks 

graafikaobjektide, tabelite ja massiividega. VBA lausete ja Scratchi käsuplokkide vahel on suur sarnasus, 

eriti protsesside juhtimise ja graafikaandmete kasutamise osas. Lihtsuses saab Visual Basicuga võistelda ehk 

ainult Python. Kuid Pythonis on mõnevõrra keerulisem ja töömahukam kasutajaliideste loomine, eriti kui 

kasutatakse tabeleid, massiive ja graafikaobjekte.  

Exceli töölehed ja töövihikud pakuvad lihtsaid ja mugavaid vahendeid kasutajaliideste loomiseks ja 

andmete, sh graafikaobjektide, salvestamiseks ja säilitamiseks. Exceli töölehte võib teatud määral võrrelda 

Scratchi lavaga. Kuid võrreldes viimasega on töölehel praktiliselt piiramatud mõõtmed ning kindlad 

mõõtühikud, mis võimaldavad teha ka mõõtkavas joonistusi, kasutades ühikutena näiteks sentimeetreid. 

Töölehe lahtrid aga võimaldavad salvestada suuri andmehulki üksikväärtustena, tabelitena ja massiividena. 

Kusjuures kõik andmete paigutamisega ja vormindamisega seotud küsimused saab lihtsalt lahendada Exceli 

vahenditega, mis vähendab oluliselt vajadust tegeleda sellega loodavas programmis. 

Töötamiseks VBAga Exceli keskkonnas siin pakutavas lähenemisviisis ei pea eriti palju tundma Excelit (kuigi 

halba see muidugi ei tee). Praktiliselt piisab oskustest määrata nimesid lahtritele ja lahtriplokkidele, lisada 

ja eemaldada töölehti ja lahtriplokke, salvestada ja avada töövihikuid. Veidi tuleb kokku puutuda ka 

joonestamisvahendite kasutamisega ja vormindamisega. 

Eeldatakse, et lugeja juba omab baasteadmisi programmeerimisest ning on töötanud Scratchiga. VBA ja 

programmeerimise üldisi mõisteid ja vahendeid püütakse selgitada ja kinnistada Scratchist omandatud 

teadmiste ja oskuste abil.  

Kiirtutvus VBAga  

Programmide ja protseduuride liigid ning struktuur 

VBA toega rakendus võimaldab hallata kasutajaliidest ning määrata operatsioone ja tegevusi andmete ja 

baasrakenduse (Excel, Word, AutoCAD jm) objektidega:  

 andmed – arvud, tekstid, kuupäevad jms, 

 objektid – Exceli töövihikud, töölehed, lahtrid, graafilised kujundid, Wordi dokumendid jms. 

Excelis saab kasutada kahte liiki VBA programme: 

 makrod – võimaldavad määratleda suvalisi tegevusi, käivitatakse näiteks käsunupuga, 

 töölehefunktsioonid – võimaldavad leida ainult väärtusi, saab kasutada Exceli valemites. 

Üldjuhul võib programm koosneda mitmest protseduurist. Protseduure on kahte liiki: 

 Sub – protseduurid ehk alamprogrammid, 

 Function – protseduurid ehk funktsioonid. 

Mitmeprotseduurilises programmis on üks protseduur alati peaprotseduur, teised on alamprotseduurid. 

Viimased võivad omakorda kasutada teisi alam-

protseduure jne. Programmi täitmine algab pea-

protseduurist sellekäivitamisel. Alamprotseduur hakkab 

tööle siis, kui selle poole pöördutakse peaprotseduuris või 

teises alamprotseduuris. Makro peaprotseduur on alati 

Sub-protseduur ja töölehefunktsioonil Function-protse-

duur. Alamprotseduurid võivad olla mõlemal suvalised. 

Lihtsamal juhul koosneb makro ühest Sub-protseduurist, 

töölehefunktsioon ühest Function-protseduurist.  

NB! Töölehefunktsioone saab kasutada ainult Excelis.  

Protseduur koosneb lausete (korralduste) jadast. Lausete 

abil määratakse vajalikud tegevused ja nende täitmise 

järjekord, esitatakse programmi ja protseduuride struktuur, kirjeldatakse andmeid jm. 

Laused koosnevad võtmesõnadest (kindla tähendusega ja kasutamiskohaga sõnad: Sub, Dim, End, If jms), 

nimedest, avaldistest jm.  

Protseduuride tüüpstruktuur 

Makro peaprotseduur 

Parameetriteta Sub-protseduur 

Sub-tüüpi alamprotseduur 

Võivad olla sisend- ja  
väljundparameetrid 

Function-protseduur – suvaline 

Võivad olla ainult  
sisendparameetrid 

  Sub nimi ( ) 

     laused  [ ja kommentaarid ] 

  End Sub 

  Sub nimi ([ parameetrid ]) 

   [laused ja kommentaarid] 

   End Sub 

Function f_nimi ([parameetrid ]) 

    [ laused  ja kommentaarid ] 

    f_nimi = tagastatav väärtus 

End Function 

NB! Makro peaprotseduuril ei 
tohi olla parameetreid! Nime 
järel peavad olema tühjad sulud! 

Pöördumine Call-lausega:  

 [Call] nimi ([ argumendid ]) 

Pöördumine funktsiooniviidaga: 

  f_nimi ([ argumendid ]) 

NB! Siin ja edaspidi tähendavad nurksulud, et nende vahel olev element võib puududa. Näiteks ei pea igal 

protseduuril olema parameetreid. Sellisel juhul puuduvad pöördumisel ka argumendid. 

Andmevahetuseks protseduuride vahel saab kasutada parameetreid ja argumente. Sub-protseduuridel 

võivad olla sisend- ja väljundparameetrid. Viimastest igaüks tagastab ühe väärtuse. Funktsioonil võivad olla 

ainult sisendparameetrid: tulemus ehk tagastatav väärtus (saab olla ainult üks) omistatakse funktsiooni 

NIME NIMELISELE muutujale ja selle kaudu tagastatakse väljakutsuvale protseduurile. Argumentide arv, liik ja 

järjekord peab vastama parameetritele. Andmevahetuseks võib kasutada ka globaalseid muutujaid. 

Protseduuride sees kasutatakse sisemuutujaid ehk lokaalseid muutujaid. Scratchi skriptid vastavad enam-

vähem protseduuridele. 

Protseduurid salvestatakse moodulilehtedele ehk moodulitesse, mis asuvad töövihiku eraldi osas 

VBA-projektis, kasutades Visual Basicu redaktorit (VBE – Visual Basic Editor), mis toetab programmide 

sisestamist ja redigeerimist, kontrollides operatiivselt keele süntaksit, pakkudes spikreid, abiinfot jms. 

Makrode käivitamiseks on mitmeid erinevaid võimalusi. Makro saab käivitada alati Exceli menüü 

vahekaardilt Developer (Arendaja). Käivitamise kiirendamiseks ja hõlbustamiseks võib paigutada töölehele 

käsunupu või mingi graafikaobjekti ning siduda makro sellega. Töölehefunktsiooni käivitamine toimub 

Exceli valemist funktsiooniviida abil. 

Protseduuride täitmist korraldab spetsiaalne programm – VBA interpretaator (translaator). Saades 

korralduse mingi protseduuri täitmiseks, eraldab interpretaator protseduurile täitmise ajaks arvuti mälus 

tööpiirkonna (programmiplokk ja andmeplokk). Programmiplokki tuuakse vastava protseduuri tekst ning 

kontrollitakse selle süntaksit. Vea korral väljastatakse veateade ja täitmine katkestatakse. Kui vigu ei ole, 

eraldatakse andmeplokis mälupesad (väljad) protseduuri poolt kasutatavate muutujate väärtuste ajutiseks 

säilitamiseks. Seejärel tõlgib interpretaator protseduuri laused masinkeelde, suunab selle täitmiseks 

protsessorisse ning tagastab tulemused andmeplokki. 

Näide: Tutvus  

Inimese pikkuse ja massi (kaalu) alusel leitakse tema kehamassi indeks:  

kind = mass/pikkus2 , mass (kaal) – kg, pikkus – m  

ja selle alusel antakse hinnang tema nö saleduse kohta, arvestades järgmisi kriteeriume: 

ülekaalsuur  - 30 ülekaal; - 3025 normaalne;- 2581 kõhn;18  indindindind kkkk  

Näite abil teeme tutvust VBA põhimõistete ja vahenditega: andmed ja objektid, protseduurid, laused, 

andmete sisestamine ja väljastamine, valikute ja korduste kirjeldamise lihtsamad võimalused jms. 

Vaadeldakse kolme põhivarianti: 

 Makro. Interaktiivne andmevahetus dialoogibokside abil,  

 Makro. Algandmed ja tulemused töölehe lahtrites, andmevahetus töölehtedega, 

 Töölehefunktsioonid. Andmed ja valemid töölehel. 

Variant 1. Ülesande püstitus ja kasutajaliides  

Programm kuvab tutvustava teate ning küsib ja loeb kasutaja nime. Seejärel küsitakse, kas kasutaja tahab 

lasta leida oma kehamassi indeksi. Kui jah, siis programm küsib ja loeb kasutaja poolt sisestatud pikkuse ja 

massi, leiab ja kuvab kehamassi indeksi ning annab hinnangu selle kohta. Kui kasutaja ei soovi sellist infot, 

väljastab programm kahetsuse ja lõpetab töö.  

 

Kasutajaliides asub töölehel, sisaldades rohkem 

elemente kui otseselt vaja. Piisaks ju ainult käsu-

nupust programmi käivitamiseks, kuid isegi selleta 

saaks hakkama. Töölehel on ka kaks graafika-

objekti: Juku ja Kraps. Nende abil tutvustatakse 

graafikaobjektide kasutamise põhimõtteid. Enne 

programmi töö lõppu teevad nad salto ning nende 

asukoht töölehel muutub. 

Kui lugeja tahab proovida antud programmi käima panna, võib skeemil näidatud pildikeste asemel võtta 

suvalised graafikaobjektid. Võib näiteks joonistada mingi kujundi, kasutades Exceli korraldust Insert Shapes 

või lisades pildi failist korraldusega Insert Picture. Oluline on, et nimed vastaksid programmis olevatele 

nimedele (vt programmi viimased read). Nime panemiseks või muutmiseks teha pilt aktiivseks ja tippida 

nimekasti valemirea vasakus servas vajalik nimi, seejärel vajutada kindlasti klahvile Enter. Vt jaotist 

„Graafilised kujundid – klass Shape“.  

Programm 

Programm (makro) koosneb kolmest protseduurist: peaprotseduur Tutvus ning alamprotseduurid saledus 

(Function-protseduur) ja Liigu (Sub-protseduur). Viitamiseks graafikaobjektile kasutatakse konstruktsiooni 

Shapes(“nimi“). 

Sub Tutvus() 'peaprotseduur 

    Dim nimi, v, pikkus, mass, indeks 'muutujad 

    MsgBox  "Tere! Olen Juku!"   ' teate kuvamine 

    nimi = InputBox("Sinu nimi => ")   ' nime lugemine 

    If nimi = "" Then MsgBox "Kedagi ei ole!": End 

    v = InputBox("Leian keha indeksi (1 - jah  / 0 - ei )? ") 

    If Val(v) <>  1 Then  ' valiku algus 

        MsgBox nimi + "! Kahju! Ehk mõtled veel?" 

        Exit Sub 

    Else                  ' alternatiiv 

        pikkus = InputBox (nimi & "! Sinu pikkus (cm) ") 

        mass = InputBox ("aga mass/kaal (kg) ") 

        indeks = mass / (0.01 * pikkus) ^ 2 

        MsgBox nimi & "! Sinu indeks on:" & Round(indeks,1) 

        MsgBox nimi & "! Oled " & saledus (pikkus, mass) 

    End If 

    MsgBox "Kohtumiseni! Igavesti Sinu, Juku ja Kraps!" 

    Call Liigu (Shapes("Juku"), 200, 100)  'AP kutsung 

    Liigu Shapes ("Kraps"), 100, 50 ' Call on ära jäetud  

End Sub 

 

Function saledus(L, m) 

' hinnang massiindeksi alusel 

 ' L – pikkus, m – mass 

   Dim indeks 

   indeks = m / (L / 100) ^ 2 

   If indeks < 18 Then 

       saledus = "kõhn" 

   Else 

       If indeks <= 25 Then 

           saledus = "normis" 

       Else 

           If indeks <= 30 Then 

               saledus = "ülekaaluline" 

           Else: 

               saledus = "suur ülekaal" 

           End If 

       End If 

   End If 

End Function 

 

Parameetrite L ja m abil antud 

pikkuse ja massi väärtustest 

arvutatakse kehamassiindeks. 

 Selle alusel leitakse hinnang, mis 

omistatakse funktsiooni NIME 

NIMELISELE muutujale.  

Valik tehakse mitmetasemelise 

If-lause abil 

Sub Liigu (kuju, x, y)  

 ' kuju pöörlemine ja asukoht 

    Dim i 

    For i = 1 To 360 

        kuju.IncrementRotation 1 

        DoEvents 

   Next i 

   kuju.Left = x * Rnd()  ' vasak  

   kuju.Top = y * Rnd()  ' ülemine 

End Sub 

Parameetriteks on graafikaobjekt kuju ja selle sihtkoha 

koordinaadid (x, y). Konkreetne kuju ja koordinaatide maksimaalsed 

väärtused antakse pöördumisel argumentide abil. 

Kuju teeb pöörde 360. For…Next-lause määrab korduse:  i muutub 

1 kuni 360 ja iga kord täidetakse korduses olevad laused.  

Meetod IncrementRotation muudab kuju pöördenurka  

(siin iga kord 1). Käsk DoEvents tagab, et pöörlemine oleks nähtav.  

Omaduste Left ja Top väärtused määravad kuju asukoha. 

Funktsioon Rnd() annab juhuarvu vahemikus 0 kuni 1. 

Programmide sisestamine, redigeerimine ja käivitamine 

Soovitame kopeerida programmi VBA redaktorisse, panna see käima ning katsetada ja täiendada. Alguses 

võib graafikaobjektid ja protseduuri Liigu ära jätta. Viimased kaks rida peaprotseduuris võib sellisel juhul 

muuta kommentaarideks, lisades nende ette ülakomad. Programmi peaprotseduuri ja alamprotseduuri 

Liigu võiks salvestada lehemoodulisse (vt allpool), millel asub kasutajaliides, funktsiooni saledus, aga 

üldmoodulisse, mis oleks kasulik ka järgmiste variantide jaoks.  

VBA projekti struktuur  

Programmi protseduurid salvestatakse VBA projekti moodulitesse Visual Basicu redaktori (VBE – Visual 

Basic Editor) abil. Redaktori aktiveerimiseks saab kasutada mitmeid erinevaid viise. Esmakordsel 

pöördumine võib kasutada käsku (nuppu) Visual Basic vahekaardilt Developer. Redaktori saab käivitada ka 

klahvikombinatsiooniga Alt+F11. VBE keskkonda pääseb, klõpsates hiire parempoolse nupuga lehelipikul ja 

valides ilmuvast objektimenüüst korralduse View Code, mis viib antud töölehe moodulisse. Kui Visual Basic 

redaktorit on antud seansis juba kasutatud, asub tegumiribal ikoon, mis võimaldab kohe pääseda 

redaktoriaknasse. 

VBA projekti põhikomponentideks on moodulid, kuhu saab salvestada protseduure. Kaheks peamiseks 

mooduli liigiks on objektimoodulid ja üldmoodulid.  

Oma objektimoodul on igal lehel ja ka töövihikul. Need luuakse ja eemaldatakse koos töövihiku ja lehe 

loomisega. Lehemoodulisse salvestatakse tavaliselt sellised 

protseduurid, mis on tihedalt seotud antud lehega, 

kasutades näiteks ainult antud lehe objekte, lehe ja selle 

alamobjektide sündmusprotseduure jms.  

Üldmoodulisse võib salvestada suvalisi protseduure. 

Tavaliselt paigutatakse neisse üldise iseloomuga protse-

duure, mis on seotud mitme lehega. Üldmooduleid algselt 

töövihikus ei ole, kuid neid saab alati lisada korraldusega 

Insert Module ja nende arv ei ole piiratud. Vaikimisi on 

üldmoodulite nimed Module1, Module2, …, mida saab 

muuta omaduste aknas (Properties Window). 

Üldmooduleid saab ka eemalda.  

NB! Töölehefunktsioonid peab tingimata salvestama üldmoodulitesse.  

Redaktori kasutajaliides 

 

Redaktori akna ülemises servas on menüü ja standardriistariba. Ekraani vasakus servas asub 

projektihalduri aken, milles on kuvatud moodulite puukujuline struktuur (kaustad ja moodulid) iga projekti 

(avatud töövihiku) jaoks. Alguses on objektimoodulite (lehtede ja töövihiku) moodulite kaust Microsoft 

Excel Objects ja seejärel (kui on) – üldmoodulite kaust Modules. Vajaliku mooduli aktiveerimiseks peab 

tegema topeltklõpsu selle nimel. Samas piirkonnas on alguses tavaliselt ka põhiobjektide (lehed, töövihik, 

moodulid) omaduste (atribuutide) aken Properties Window (pildil ei ole näidatud), mille võib esialgu 

sulgeda. Vajadusel saab selle kuvada korraldusega View, Properties Window. 

Põhiosa ekraanist võtab enda alla koodi- ehk programmiaken, kuhu sisestatakse mooduli protseduurid. 

Protseduuride arv ühes moodulis ei ole piiratud. Pildil on aktiivne töölehe Sheet1 (Tutvus) moodul, milles 

on makro Tutvus protseduurid. 

Lehemooduli jaoks on projektihalduris toodud kaks nime: VBA sisenimi ja välisnimi ehk lihtsalt nimi 

(sulgudes). Sisenimi on algselt esitatud kujul SheetN, kus N on lehe järjenumber, mida suurendatakse 

automaatselt lehtede lisamisel. Sisenimesid saab muuta omaduste (Properties) aknas. Välisnimi on näha 

lehe nimelipikul ning seda saab muuta Exceli aknas või VBA protseduuris. Kui nimesid ei ole muudetud, 

langevad sise- ja välisnimi kokku. 

Programmi teksti sisestamisel kontrollib VBA redaktor tavaliselt automaatselt ridade kaupa lausete süntaksi 

õigsust. Kui antud rea mingis lauses on süntaksivigu, siis kuvatakse üleminekul uuele reale vastav teade, 

muudetakse reas oleva teksti värvust (vaikimisi punaseks) ning paigutatakse kursor vea eeldatavale 

asukohale, mis ei pruugi olla alati päris täpne. 

Tüüpilised süntaksivead on võtmesõnade vale esitus, tehtemärkide ärajätmine avaldistes (eeskätt 

korrutamistehtes), eraldajate (tühikud, komad, sulud, jutumärgid, …) ärajätmine vajalikes kohtades, 

tühikute kasutamine nimedes või võtmesõnades jms.  

VBA redaktori tugi on väga kasulik objektide omaduste ja meetodite sisestamisel. Kui peale objektiviita 

sisestada punkt, pakub redaktor antud klassi objektide omaduste ja meetodite loendit vajaliku elemendi 

valimiseks, hoides nii kokku aega ja vähendades vigade tekkimise võimalust. 

Informatsiooni objektide (klasside) ning nende omaduste ja meetodite kohta saab objektisirvija 

Object Browser abil ning VBA abiinfosüsteemist Microsoft Visual Basic Help. Viimasest saab informatsiooni 

ka programmeerimiskeele lausete, funktsioonide ja muude elementide kohta. Informatsiooni saamiseks 

konkreetse võtmesõna, omaduse, meetodi jmt kohta, võib kursori viia vastava elemendi kohale ja vajutada 

klahvile F1. 

objektimoodulid 

üldmoodulid 

Makrode käivitamine 

Makro käivitamine toimub enamasti Exceli aknas, kusjuures tavaliselt peab muutma aktiivseks töölehe, 

millel paiknevad makro poolt kasutatavad objektid ja andmed.  

Käivitamiseks võib alati kasutada dialoogiboksi Macro, mille saab kuvada käsuga Macros vahekaardil 

Developer (Arendaja), kui see ei ole peidetud, kuid võib kasutada ka kiirkäsku Alt+F8. Seejärel kuvatakse 

dialoogiboks Macro. Kuvatud loetelust saab valida vajaliku makro ning peale klõpsamist nupule Run asub 

VBA selle täitmisele.  

Makro kiireks käivitamiseks võib paigutada töölehele makroga seotud käsunupu või mingi graafikaobjekti. 

Makro käivitatakse hiireklõpsuga vastaval nupul või graafikaobjektil.  

Käsunuppe on kahte tüüpi: Exceli vormi (Form) käsunupud ja ActiveX 

käsunupud. Viimased on uuemad ja pakuvad rohkem võimalusi, kuid ActiveX 

käsunupu loomine ja sidumine makroga on mõnevõrra keerulisem. Alguses võiks 

proovida Form käsunuppe, kuid üsna pea võiks üle minna ActiveX 

käsunuppudele. Nii ühe kui teise loomiseks kasutatakse vahekaardi Developer 

(arendaja) grupi Controls (juhtelemendid) käsku Insert (lisa). Selle klõpsamisel 

kuvatakse kaks tööriistakasti: Forms Controls (vormi juhtelemendid) ja ActiveX 

Controls (ActiveX juhtelemendid), kus saab teiste Windowsi ohjurite (juhtijate) 

seast valida käsunupu. 

Makro sidumine Form käsunupuga 

Valida tööriistakastist käsunupp ja hiireklikiga lisada see töölehele. Ilmub dialoogiboks Assign Macro 

(makro omistamine), kus kuvatakse olemasolevate makrode loetelu, millest saab:  

 valida vajalik makro ja klõpsata nuppu OK, 

 muuta nupul olevat teksti (ei ole kohustuslik). 

Makro sidumine ActiveX käsunupuga 

Valida tööriistaboksist käsunupp ja hiireklikiga lisada see töölehele. Nupul on tekst CommandButtonN, kus 

N on 1, 2, … , tähistades sama tüüpi nupu järjenumbrit antud töölehel. Kui tegemist on esimese (ainukese) 

nupuga, siis N on 1 ning sisse lülitub disainirežiim (vt nupp Design Mode).  

Nupu sidumine loodud makroga 

1. Tehes topeltklõpsu makronupul, kuvatakse antud lehe moodul, millesse on lisatud sündmusprotseduuri 

mall: 

Private Sub CommandButtonN_Click() 
 

End Sub 

2. Protseduuri alguse ja lõpu vahele lisatakse käivitatava protseduuri nimi. Seejärel tuleks siirduda tagasi 

töölehele. 

3. Muuta nupul olevat teksti (muutmine ei ole kohustuslik, kuid reeglina seda tehakse). 

4. Klõpsata nuppu Properties (Atribuudid), mille tulemusena kuvatakse nupu omadused (Properties). 

5. Asendada reas Caption tekst CommandButtonN oma soovitud tekstiga. 

Makro sidumine graafikaobjektiga 

1. Lisada töölehele suvaline kujund (Insert , Shapes), mille võib ka importida. 

2. Teha kujundil hiire paremklikk ning valida ilmuvast objektimenüüst korraldus Assign Macro. 

3. Kuvatud dialoogiboksist Assign Macro (nagu Forms käsunupu korral) valida makro ja klõpsata 

nuppu OK. 

Kui Forms käsunupuga või graafikaobjektiga on seotud makro, käivitab hiireklikk sellel vastava makro 

koheselt. Nupu või graafikaobjekti  asukoha, suuruse vms muutmiseks on vaja hiirekliki tegemise ajal hoida 

all klahvi Ctrl. 

Lausete struktuur ja põhielemendid 

Laused on korraldused (käsud), mis määravad tegevusi. Lausetel on keelereeglitega määratud kindel 

otstarve, struktuur – süntaks ja täitmise reeglid – semantika. Lausete põhielemendid on järgmised:  

 Võtmesõnad: kindla asukohaga, kujuga ja tähendusega sõnad (Sub, Function, If, Else, …)  

 Funktsiooniviidad: Round(indeks,1), MsgBox("Tere! "), InputBox("mass"), saledus(pikkus, mass) 

 String- ja arvkonstandid: "normaalne", "Sinu nimi => ", "juku", 100, 0.01, -31.71 jms  

NB! Stringkonstandid paigutatakse jutumärkide vahele  

NB! reaalarvudes on murdosa eraldajaks punkt 

 Protseduuride, muutujate, parameetrite nimed: saledus, Liigu, v, pikkus, L, m,  x1, x_1, a13s1  

Nimede esitamiseks kehtivad ühtsed reeglid: 

Nimi võib koosneda ühest tähest või tähtede, numbrite ja allkriipsude jadast, mis 

algab tähega. Nimes ei tohi olla tühikuid. Suur- ja väiketähti ei eristata.  

 

 Avaldised: mass / (0.01*pikkus) ^2, nimi & "! Sinu pikkus (cm) ", nimi = " ", indeks < 18 

 Avaldiste põhiliikideks on arvavaldised, stringavaldised ja loogikaavaldised (erijuht: võrdlus) 

 Tehtesümbolid ehk operaatorid: ^, *, /, +, -, =, <= 

 Piirajaid ja eraldajad: (  ) , ”  : . 

Laused jagunevad lihtlauseteks ja liitlauseteks. Liitlause võib sisaldada teisi liht ja/või liitlauseid. 

Liitlauseteks on vaadeldavas näites if-laused peaprotseduuris ja funktsioonis saledus ning for-lause 

protseduuris Liigu. Kõik teised on lihtlaused. 

Muutujate deklareerimine. Dim-lause 

Dim nimi, v, pikkus, mass, indeks, lahter 

See lause määratleb – öeldakse ka deklareerib – muutujad. Lihtsamal juhul antakse Dim-lauses muutujate 

nimede loetelu, kusjuures iga muutuja jaoks võib näidata tema väärtuste tüübi – muutuja tüübi.  

Näiteks: 

 Dim nimi As String, v As Integer, pikkus As Double, mass, indeks 

Muutujate tüübid näidatakse vastavate võtmesõnadega: muutuja nimi tüübiks on string (tekst või sõne),  

v tüübiks on täisarv, pikkus tüübiks on reaalarv. Kui muutuja tüüp pole Dim-lauses antud nagu ülaltoodud 

näites muutujad mass ja indeks, valib süsteem (VBA interpretaator) selle ise.  

Enamasti ei ole lihtsamates programmides muutujate tüübi määramine vajalik ning sellest 

saadav mälumahu kokkuhoid märkimisväärne. Hiljem näeme, et tüüpide määramisest on 

teatud kasu objektimuutujate jaoks programmide sisestamisel ja redigeerimisel. 

Muutujate deklareerimine ei ole Visual Basicus kohustuslik, kuid muutujate loetelu andmine 

(ka ilma tüüpi määramata) aitab lihtsustada võimalike vigade peilimist. 

Vt jaotist „Andmete liigid ja tüübid“. 

 

Enne protseduuri täitmist eraldab VBA interpretaator igale muutujale oma tööpiirkonnas mäluvälja ehk 

pesa. Programmi täitmise ajal saab vastavate lausete toimel salvestada nendesse pesadesse väärtusi. 

Vajadusel saab hiljem lugeda neid väärtusi näiteks uute väärtuste leidmiseks. 

 

Scratchis luuakse muutujad „käsitsi“, kasutades grupi Muutujad käsku Tee muutuja. 

Laval kuvatakse monitor, kus näeb muutuja jooksvat väärtust. Monitori saab ka 

peita. Muidu on sama lugu nagu enne – ikka mälupesa, ei midagi muud.  

Andmete sisestamine ja väljastamine dialoogiboksidega 

Suurem osa näite peaprotseduuri lausetest on seotud andmete sisestamisega ja väljastamisega 

dialoogibokside abil (vt jaotist „Dialoogibokside kasutamine“ ). Siin näidatakse nende lihtsamaid võimalusi. 

Andmete kuvamiseks kasutatakse protseduuri MsgBox, millel lihtsamal juhul on järgmine kuju: 

MsgBox avaldis 

Üldjuhul on tegemist arv- või stringavaldisega, kuid väga tihti on 

tegemist lihtsalt stringiga, mille abil kuvatakse teade, näiteks:  

 MsgBox  "Tere! Olen Juku!"   või   MsgBox  "Hello, World!“ 

Kõige tüüpilisem on stringavaldiste kasutamine, mis moodustatakse  

stringkonstantidest ning arv- või stringmuutujatest sidurdamistehte 

abil.  

Sidurdamistehte operaator on & või + (erinevus on vaid eri tüüpi operandide kasutamisel). 

MsgBox nimi + "! Sinu indeks on:" & Round(indeks, 1)  või  MsgBox "pindala = " & S 

 

Scratchis kasutatakse sel puhul käsku  

Stringavaldise saab panna kokku plokiga  

Andmete lugemiseks klaviatuurilt kasutatakse omistamislause erijuhtu, kus paremas pooles on VBA 

sisefunktsioon InputBox: 

 muutuja = InputBox (teade, …) 

Esimene (kohustuslik) argument teade esitatakse enamasti stringkonstandi abil, kuid see võib olla ka 

stringavaldis. Näiteid vaadeldavast protseduurist: 

nimi = InputBox ("Sinu nimi => ") : pikkus = InputBox (nimi & "! Sinu pikkus (cm) ") 

 

Lause täitmisel programmi töö peatakse ja kuvatakse 

dialoogiboks, milles on teade ja väli, kuhu saab 

sisestada väärtuse. Peale nuppu OK klõpsamist, 

omistatakse antud väärtus muutujale ja programmi 

töö jätkub. Kui väärtust ei sisestata ja klõpsatakse OK 

või Cancel, omistatakse muutujale tühi väärtus! 

 

Sisendboksist loetud arv on salvestatud muutuja väljas tekstivormingus (vt jaotist 

„Dialoogibokside kasutamine“). Selliselt esitatud arvuga võib tekkida probleeme 

aritmeetika ja võrdlemise operatsioonide täitmisel. Seetõttu on sageli otstarbekas 

teisendada väärtus arvuvormingusse, kasutades näiteks funktsiooni Val. Nii on seda 

tehtud näiteks võrdluses lauses If Val(v ) <> 1 Then. Kui on kindlalt teada, et väärtust ei 

kasutada liitmisoperatsioonides ja võrdlustes, võib väärtuse jätta ka teisendamata. 

 

 

 

Scratchis kasutatakse andmete lugemiseks klaviatuurilt plokki . Väärtus 

salvestatakse sisemuutujas vastus. Vajaduse korral saab selle omistada suvalisele 

muutujale, kasutades käsku   

Omistamisest ja omistamislausest 

Omistamist kasutatakse programmides väga sageli.  

Lause üldkuju enamikes tekstipõhistes programmeerimiskeeltes on: 

 muutuja = avaldis 

Lause täitmisel leitakse avaldise väärtus ja saadud tulemus omistatakse antud muutujale: st salvestatakse 

muutuja väljas ehk pesas. Avaldise erijuhtudeks on konstant, muutuja ja funktsiooniviit.  

Üldkujul kujul esinevad omistamislaused näite peaprotseduuris ja funktsioonis saledus: 

indeks = mass / (0.01 * pikkus) ^ 2 ja  indeks = m / (L / 100) ^ 2 

Omistamislause erijuhtu vaadeldi andmete sisestamisel funktsiooniga InputBox. Samuti on erijuhuks 

omistamislause saledus = konstant funktsioonis saledus. Siin omistatakse väärtus funktsiooni NIME 

NIMELISELE muutujale.  

Üheks võimaluseks on ka väärtuse omistamine objekti omadusele, nagu näiteks 

kuju.Left = x * Rnd() 

– objekti omaduse Left väärtuseks võetakse avaldise väärtus. 

 

Scratchis on kaks omistamiskäsku:  

  ja  

Teine on erijuht, mille saab panna kirja ka nii: [võta muutuja = (muutuja + avaldis)] 

Valikud ja If-lause 

Kõikides programmeerimiskeeltes on valik (If-lause) üks peamisi protsesside juhtimisega seotud tegevusi. 

Enamikes keeltes on vähemalt kaks If-lause varianti, mis võimaldavad teha valiku ühest (laused kas 

täidetakse või jäetakse vahele) või valiku kahest (täidetakse üks kahest käsuplokist). Paljudes 

programmeerimiskeeltes on vahendid mitmese valiku kirjeldamiseks. Visual Basicus on olemas kõik kolm 

varianti, mis esinevad ka järgnevas näites.  

Peaprotseduuris on kasutusel valik ühest ja valik kahest. Valiku korral ühest on kaks varianti: üherealine 

lause ja mitmerealine lause. Esimest varianti tasub kasutada ainult erijuhul – If -lausesse kuulub ainult üks 

või kaks lihtlauset. 

Valik ühest 

Üherealine If-lause 

  If nimi = "" Then MsgBox "Kedagi ei ole!": End 

Mitmerealine If-lause  

  If v = "" Then  

      MsgBox "Kahju!" 

      Exit Sub 

  End If 

Üherealine If-lause 

If tingimus Then laused 

Mitmerealine If-lause 

If tingimus Then  

   laused 

End If 

 

Valik kahest 

  If  Val(v ) = 0 Then  ' kui tingimus tõene 

      MsgBox nimi + "! Ehk mõtled veel?" 

     Exit Sub 

  Else                        '  tingimus väär 

      pikkus = InputBox (…) 

        …   

  End If 

  If  tingimus Then   

      laused_1 

  Else                         

      laused_1 

  End If 

  

Mitmese valikuga on tegemist funktsioonis saledus. Sellist valikut saab kirjeldada mitmel erineval viisil, kuid 

alati saab kasutada ühes If-lauses teisi If-lauseid. Praegu on esimese taseme If-lause If-harus üks lihtlause, 

else-harus (ehk else-osalauses) on kasutatud järgmise taseme If-lauset ja viimases veel ühte. Sisemisi 

lauseid võib jaotada harude vahel ka teisiti. Antud juhul võib näiteks panna esimese taseme If-harusse ning 

ka else-harusse veel If-lause.  

Visual Basicus on aga veel üks võimalus, milleks on ElseIf-osalausete kasutamine.  

Allpool on toodud mõned vaadeldava If-lause võimalikud esitamise variandid.  

Mitmene valik  

   If indeks < 18 Then 

       saledus = "kõhn" 

   Else 

       If indeks <= 25 Then 

           saledus = "normis" 

       Else 

           If indeks <= 30 Then 

               saledus = "ülekaaluline" 

           Else 

               saledus = "suur ülekaal" 

           End If 

       End If 

   End If 

   If indeks <= 25 Then 

       If indeks < 18 Then 

           saledus = "kõhn" 

       Else 

           saledus = "normis" 

       End If 

   Else 

       If indeks <= 30 Then 

           saledus = "ülekaaluline" 

       Else 

           saledus = "suur ülekaal" 

       End If 

   End If 

If indeks < 18 Then 

    saledus = "kõhn" 

 ElseIf indeks <= 25 Then 

    saledus = "normis" 

 ElseIf indeks <= 30 Then 

     saledus = "ülekaaluline" 

 Else 

     saledus = "suur ülekaal" 

 End If 

 

Kasutajafunktsioonid 

Kasutaja saab luua oma funktsioone, mille kasutamine programmides toimub analoogselt sise-

funktsioonidega. Funktsiooni peaprotseduuriks on Function-protseduur: 

Function nimi ( [parameetrid] ) 

 protseduuri keha 

 nimi = avaldis 

End Function 

Parameetritega esitatakse funktsiooni sisendandmed (kui vaja).  

Näiteks on funktsiooni päis järgmine: 

Function saledus(L, m) 

Funktsiooni nimeks on saledus, parameetriteks: L – pikkus (cm) ja m – mass (kg). 

Tüüpiliselt funktsioon leiab ja tagastab ühe väärtuse. Tagastatav väärtus omistatakse funktsiooni NIME 

NIMELISELE muutujale. Selliseid omistamisi võib olla mitu, kuid tagastatakse ainult üks väärtus (viimasena 

omistatu). Funktsiooni töö lõpetab tavaliselt lause End Function, kui täitmisjärg jõuab selleni. Funktsiooni 

töö saab katkestada lausega Exit Function. Tulemus ja täitmisjärg tagastatakse kohta, kust toimus 

pöördumine funktsiooniviida abil.  

Funktsiooniviit esitatakse kujul: nimi (argumendid), kus nimi on funktsiooni nimi ning argumendid annavad 

parameetrite väärtused. Argumentide arv, tüübid ja järjekord peavad vastama parameetritele.  

Funktsioon saledus tagastab väärtuse, mille leiab (valib) If-lause muutuja indeks väärtuse alusel. Viimane 

omakorda sõltub parameetrite L ja m väärtustest. Pöördumine funktsiooni poole toimub peaprotseduuri 

lausest: 

MsgBox nimi & "! Oled " & saledus (pikkus, mass) 

Parameetritele L ja m vastavad argumendid on esitatud muutujate pikkus ja mass abil.  

Oma funktsioone saab kasutada ka töölehe valemites. Sellisel juhul peab protseduur asuma üldmoodulis.  

Objektid, omadused ja meetodid 

Protseduuris Liigu kasutatakse graafikaobjekte (vt „Graafilised kujundid – klass Shape“). Igal objektil on 

teatud hulk omadusi ja meetodid, mida saab kasutada tegevuste määramiseks. Omaduste ja meetodite 

valik ning nende nimetused sõltuvad objekti tüübist ehk klassist. Graafikaobjektide ehk kujundite (klass 

Shape) omadusteks on näiteks nimi (Name), vasaku ja ülemise serva koordinaadid (Left ja Top) jms. 

Objektide asukoha muutmiseks, pööramiseks, kopeerimiseks, … on mitmeid meetodeid.  

Viitamiseks objektidele on erinevaid võimalusi. Graafikaobjektide puhul on põhivariant: 

 Shapes(nimi), 

kus nimi on objektile töölehel määratud nimi, mis esitatakse stringkonstandi abil nagu näiteks 

Shapes(“Juku“).  

Viitamiseks saab kasutada ka muutujaid. 

Sub-tüüpi alamprotseduurid 

Sarnaselt Function-protseduuridega on Sub-tüüpi alamprotseduurid mõeldud ühe kindlapiirilise 

(alam)ülesande täitmiseks. Erinevalt peaprotseduurist võivad neil olla ka parameetrid. Kirjeldatud 

protseduur algab lausega: 

 Sub nimi ( [parameetrid] ) 

Protseduuride käivitamiseks peaprotseduurist, teisest alamprotseduurist või ka funktsioonist, kasutatakse 

pöördumislauset ehk Call-lauset: 

 Call nimi ([argumendid]) või nimi [argumendid]. 

Esitatud variandid on samaväärsed. Siin nimi on protseduuri nimi; argumendid (mis võivad ka puududa) 

peavad vastama parameetritele. Võtmesõna Call ärajätmisel peab ära jätma ka sulud. 

Allpool esitatud näites on kasutusel protseduur Liigu (kuju, x, y), mille parameetrid on järgmised:  

kuju – graafikaobjekt, x ja y – kuju sihtkoha koordinaadid (arvud). Protseduuri poole pöördutakse kaks 

korda lausetega Call Liigu(Shapes("Juku"), 200, 100) ja Liigu Shapes("Kraps"), 100, 50. 

 

Scratchi skriptid on üsna sarnased VBA protseduuridega ning plokk [teavita teade ja oota] on 

sarnane pöördumislausega (Call-lausega), kuid päris samad need ei ole, sest VBA pöördumis-

lausega saab käivitada vaid ühe konkreetse protseduuri. 

Scratchi praeguses versioonis ei ole parameetreid ja argumente. 

Scratchis saab samaaegselt (paralleelselt) käivitada mitu protseduuri: 

 

Scratchis kasutaja funktsioone praegu veel ei ole. 

Makro Tutvus. Variant 2 – tööleht 

Võimalus kasutada töölehti andmete säilitamiseks ja kasutajaliideste tegemiseks lihtsustab rakenduste 

loomist, vähendades selleks kuluvat aega, eriti kui on vaja kasutada tabeleid. Viimased on aga kõikides 

eluvaldkondades üheks peamiseks vahendiks andmete korrastatud esitamiseks. Olgu märgitud, et sageli 

kujutavad endast andmebaasid omavahel seotud tabeleid. 

Kasutajaliides 

Allpool toodud näite abil tutvustame lihtsamaid võimalusi kasutajaliidese loomiseks töölehel.  

Kõigepealt tuleb valida kohad (lahtrid ja lahtriplokid) algandmetele ja tulemustele, kasutades mitme-

suguseid abi- ning kujunduselemente nagu pealkirjad, päised, selgitused, teated jms. Lisaks nimetatutele 

võib kasutada ka mitmesuguseid vahendeid andmete sisestamiseks ja protseduuride käivitamiseks. 

Nendeks võivad olla  kerimisribad, märkeruudud, käsunupud, …  

Kuna programmides kasutatakse lahtritele ja lahtriplokkidele viitamiseks nimesid, siis üheks oluliseks tööks 

enne programmi kirjutamist on nimede määramine. 

 

Vaadeldaval juhul võiks kasutajaliides olla 

näiteks järgmine. Algandmete jaoks on 

lahtrid nimi, pikkus ja mass, tulemuste 

jaoks indeks ja hinnang. Pikkuse ja massi 

väärtuste sisestamiseks on võimalus 

kasutada kerimisribasid. Kasutades prog-

rammis viitamiseks lahtritele nimesid, ei 

ole kasutajaliidese täpsel asukohal erilist 

tähtsust. 

Inimese nime lülitamine algandmete hulka on üsna tinglik, sest ülesande lahendamiseks ei ole see otseselt 

vajalik. Formaalselt pole ülesande lahendamise seisukohast olulised ka pealkirjad lahtrite kohal. Need on 

vajalikud kasutajale, mõistmaks, kus mingid andmed on. Loomulikult ei ole mingit praktilist väärtust Juku ja 

Krapsu pildikestel, kuid mõnikord kasutatakse selliseid „kaunistusi“. Siin on need selleks, et näidata, kuidas 

saab protseduure siduda graafikaobjektidega. 

 

 

 

Kerimisriba (ScrollBar) saab lisada töölehele vahekaardi Developer 

(Arendaja) grupi Controls (Juhtelemendid) käsuga Insert (Lisa). Samas on 

ka mitmed teised kasutajaliidese komponendid nagu märkeruudud, 

komboboksid, käsunupud jms.  

Käsuga Properties (Atribuudid) kuvatavas aknas saab määrata omaduste 

väärtused: minimaalne (Min) ja maksimaalne (Max) väärtus, lahter 

valitud väärtuse salvestamiseks (Linked Cell), muutmise sammud (Small 

Change, Large Change). 

Nimede määramisest lahtritele 

Nimede määramiseks lahtritele ja lahtriplokkidele on mitmeid võimalusi. Üheks lihtsamaks, kuid mitte alati 

kõige efektiivsemaks, on nimeboksi kasutamine.  

Nimeboks (Name Box) asub vasakul valemiriba kõrval. Sisuliselt võiks seda nimetada nime ja aadressi 

boksiks, sest nime puudumisel kuvatakse selles lahtri aadress. Kui lahtril (või lahtriplokil) on nimi, kuvatakse 

nimi. Boksi saab kasutada ka nimede määramiseks. Nimede esitamise reeglid on peaaegu samasugused 

nagu muutujate puhul: nimi võib koosneda ühest 

tähest või tähtede, numbrite ja allkriipsude jadast, mis 

algab tähega, kuid on ka piiranguid – nimi ei tohi kokku 

langeda lahtri aadressiga nagu näiteks A1, X13, AS365 

vmt. Nimedena ei tohi kasutada üksikuna tähti c ja r, 

sest need on kasutusel kiirtähisena aktiivse lahtri rea 

(Row) või veeru (Column) valimiseks. 

Nimed saab määrata korraga ka mitmele lahtrile, kasutades nende kohal või kõrval olevaid tekste. Selleks 

on vahelehel Formulas (Valemid) grupis Defined Names (Määratletud nimed) käsk Create from Selection 

(Loo valikust). Eelnevalt tuleb valida sidus lahtriplokk: nimetatavad lahtrid ja lahtrid selgitavate tekstidega. 

Peale käsu sisestamist kuvatakse boks Create Names from Selection (Nimede loomine valikust), kus saab 

näidata (kinnitada) nimedeks kasutatavate tekstide asukoha. 

Alloleval pildil on näidatud nimede määramine kõrvuti asuvatele lahtritele.  

 

Vahelehel Formulas asub ka nupp Name Manager (Nimehaldur) vastava dialoogiboksi kuvamiseks, milles 

saab nimesid määrata ja täita ka muid tegevusi nendega (eemaldada, muuta, …).  

Nimed võivad olla globaalsed või lokaalsed. Globaalsed nimed kehtivad terves töövihikus ning sama nime 

võib töövihikus olla ainult üks. Lokaalne nimi kehtib ainult ühel kindlal töölehel. Selliseid nimesid võib 

töövihikus olla ka mitu. Esimest korda nime määramisel on see vaikimisi globaalne. 

Programm 

Allpool olev näiteprogramm koosneb kahest protseduurist: peaprotseduur Tutvus_2 ja funktsioon saledus, 

mis on põhimõtteliselt samad nagu eelmises variandis. Siin on toodud versioonis kasutatakse If-lauses 

ElseIf-osalauseid.  

Kui lugeja tahab programmi katsetada, võiks selleks kasutada sama töövihikut, kus on eelmine variant. 

Programmi peaprotseduuri võiks salvestada selle lehe moodulis, kus asub kasutajaliides või üldmoodulis. 

Funktsiooni võiks paigutada üldmoodulisse kui see seal juba ei ole. 

Peaprotseduuris on neli muutujat: nimi, L, mass, indeks. Protseduur loeb kõigepealt töölehelt kasutaja 

nime ning kuvab tervituse. Seejärel loeb töölehe lahtritest pikkuse ja massi väärtused ning omistab need 

muutujatele L ja mass. Omistamislause indeks = mass / (L / 100) ^ 2 leiab kehamassiindeksi väärtuse ja 

salvestab selle muutujas indeks. Järgnev lause kirjutab väärtuse muutujast töölehe lahtrisse. 

Peaprotseduuri viimase lause paremas pooles on funktsiooniviit pöördumiseks funktsiooni saledus poole, 

mis tagastab hinnangu, mis väljastatakse otse töölehele. 

NB! Muutujate nimed võivad, aga ei pea kokku langema lahtrite nimedega. 

Sub Tutvus_2() 

   ' Kehamassi indeks. Pealik 

    Dim nimi, L, mass, indeks  ' muutujad 

    nimi = Range("nimi").Value   ' loeb väärtuse lahtrist nimega nimi ja salvestab muutujas nimi 

    MsgBox "Tere, " & nimi & "! Hakkan arvutama!"  ' kuvab teate dialoogiboksis 

    L = Range("pikkus").Value   ' loeb väärtuse lahtrist pikkus ja salvestab muutujas L 

    mass = Range("mass")        ' loeb lahtrist mass, omaduse Value võib jätta ära 

    indeks = mass / (L / 100) ^ 2  ' avutatakse avaldise väärtus, tulemus omistatakse muutujale indeks 

    Range("indeks").Value = indeks ' muutuja indeks väärtus kirjutatakse töölehe lahtrisse indeks 

    Range("hinnang") = saledus(L, mass)  ' funktsiooni poolt leitav väärtus kirjutatakse lahtisse hinnang 

End Sub 



20 

Function saledus (L, m):  

   ' hinnang massiindeksi alusel 

    ' L - pikkus, m - mass 

   Dim indeks 

   indeks = m / (L / 100) ^ 2 

   If indeks < 18 Then 

       saledus = "kõhn" 

   ElseIf indeks <= 25 Then 

       saledus = "normis" 

   ElseIf indeks <= 30 Then 

        saledus = "ülekaaluline" 

   Else 

         saledus = "suur ülekaal" 

   End If 

End Function 

' Lugeja võiks proovida käivitada need protseduurid 

Sub Vaheta(K1 As Shape, K2 As Shape) 

   Dim x1, x2 

   x1 = K1.Left:  x2 = K2.Left 

   K1.Left = x2:  K2.Left = x1 

   poorle K1:  pöörle K2 

End Sub  
 

Sub pöörle (kuju As Shape) 

    Dim i 

    For i = 1 To 360 

        kuju.IncrementRotation 1 

        DoEvents 

   Next i 

End Sub 

Põhilisteks tegevusteks peaprotseduuris on väärtuste lugemine töölehelt ja tulemuste väljastamine 

(kirjutamine) töölehele. Rakenduste loomisel tabelarvutusprogrammide keskkonnas omab see olulist 

tähtsust, sest töölehed on mugavad ja lihtsalt hallatavad vahendid andmete salvestamiseks ja säilitamiseks. 

Lugemise ja kirjutamise põhivariandid (esinevad ka näites) on järgmised: 

 lugemine:  muutuja = Range(“nimi“)[.Value] 

 kirjutamine:   Range(“nimi“)[.Value] = avaldis 

Formaalselt on tegemist lahtri omaduse Value (väärtus) lugemise ja muutmisega. Kuna Value on lahtri jaoks 

nn vaikimisi võetav omadus, võib selle ära jätta, sest nii saab programmi teksti lühendada.  

Vt ka jaotist „Andmete lugemine töölehelt ja kirjutamine töölehele“. 

Viiteid lahtritele (formaalselt lahtri omadusele Value) võib otse kasutada ka avaldistes: 

Range("indeks").Value = Range("mass") / (Range("pikkus") / 100) ^ 2 

Selline lause on pikem ja ebaülevaatlikum ning selle täitmine nõuab ka oluliselt rohkem aega. Praegu ei ole 

see oluline, kuid suurte andmehulkade korral peab sellega arvestama. Otstarbekas on kinni pidada 

järgmisest põhimõttest: andmete lugemine ja kirjutamine peab olema eraldatud andmete töötlemisest. 

Tutvus. Variant 3 – töölehefunktsioonid 

Selles variandis kasutame töölehefunktsioone – VB funktsioone –, mille poole pöördutakse töölehe lahtrites 

olevatest valemitest. 

Olgu näiteks vaja leida kehamassi indeks ja hinnang selle väärtuse kohta. Hinnangu leidmiseks on olemas 

funktsioon saledus. Indeksi leidmiseks varem eraldi funktsiooni ei olnud ning see tuleb koostada. 

Function kind(L, m) 

    kind = m / (L / 100) ^ 2 

End Function 

Tegemist on väga lihtsa üherealise, sisuga funktsiooniga. Vastavad arvutused 

võiks realiseerida ka Exceli valemiga. 

NB! Et funktsiooni saaks kasutada töölehel, peab see olema salvestatud üldmoodulis. Eeldades, et lahtritele 

on määratud nimed, võib funktsioonide abil esitada arvutused nii nagu näidatud allpool.  

Töölehefunktsioonide kasutamine tabelis 

Töölehefunktsioone on mugav kasutada ka Exceli tabelis. Olgu vaja leida kehamassiindeks ja hinnang mitme 

isiku jaoks (näit meeskond, klass vmt). Lisaks on vaja leida pikkuste, masside ja indeksite aritmeetilised 

keskmised. 

 
Function p_kesk (piirkond As Range) 

    ' piirkonna keskmine 

    Dim lahter, n, S 

    n = piirkond.Cells.Count 

    S = 0 

    For Each lahter In piirkond 

        S = S + lahter.Value 

    Next lahter 

    p_kesk = S / n 

End Function 

 

Lause For Each täitmisel korratakse For ja Next vahel olevaid lauseid antud kogumi (siin lahtriplokk ehk 

piirkond) iga elemendi jaoks. 

Keskmise leidmiseks on funktsioon p_kesk, mida saab kasutada töölehelt. Selle peab salvestama 

üldmoodulisse. Ka Excelis on olemas funktsioon keskmise leidmiseks: AVERAGE(prk). 

Viidad funktsioonidele kind ja saledus sisestatakse tabeli esimese rea (rivi) lahtritesse neli ja viis, kasutades 

argumentide esituseks aadresse. Valemid kopeeritakse allapoole, järgmistesse lahtritesse. Viidad 

funktsioonile p_kesk sisestatakse vastava veeru (tulba) alla, esitades argumendi lahtrite vahemikuna: 

C5:C10, D5:D10, E5:E10. 

=kind(C5;D5) 

Makro kasutamine tabelis 

Rakendusi Exceli keskkonnas on mõnikord otstarbekas realiseerida selliselt, et Exceli enda vahendeid 

(valemeid, funktsioone jmt) ei kasutata. Kõik töötlusega seotud põhitegevused täidetakse VBA 

protseduuride abil. Kasutatakse ainult töölehti andmete salvestamiseks ja säilitamiseks ning Exceli 

vormindamise ja kujundamise vahendeid.  

 

Näites on toodud eelmises jaotises vaadeldud 

ülesanne kehamassi indeksi, saleduse ja mitme 

isiku keskmiste leidmiseks. Selleks kasutatakse 

funktsioone kind, saledus ja p_kesk, mis olid 

mängus eelmistes näidetes. Need käivitatakse 

makrost. Töölehel ei ole nüüd üldse valemeid, 

sest kõik arvutused teeb VBA makro. 

Kasutaja sisestab nimed, pikkused ja massid 

tabelisse ja käivitab makro. Isikute arv (st ridade 

arv tabelis) võib ka olla suvaline. 

Taolistes ülesannetes on olulised järgmised küsimused: 

 kuidas programm teeb kindlaks tabeli asukoha ja mõõtmed: ridade ja veergude arv, 

 kuidas viidata tabelile ja selle elementidele: lahtrid, read (rivid) ja veerud (tulbad). 

Ülalnimetatud küsimuste lahendamisel on oluline koht lahtritele või lahtriplokkidele määratud nimedel. 

Programmi laused on lühemad, kui tabelile (lahtriplokile) saab viidata objektimuutuja abil, mis määratakse 

Set-lauses. 

Näites eeldatakse, et nimi andmed on määratud tabeli kehale (ilma päiseta) ning tulpadele pikkus, mass ja 

indeks.  

Tabeli sidumine muutujaga võimaldab viidata tabeli lahtritele indeksite abil: 

          muutuja(rivi, tulp) 

kus rivi ja tulp on indeksid: rivi (rea) ja tulba (veeru) numbrid: 1, 2, … 

Sub Statistika() 

    Dim T As Range, m, i 

    Set T = Range("andmed") 

    m = T.Rows.Count 

    For i = 1 To m 

        T(i, 4) = kind(T(i, 2), T(i, 3)) 

        T(i, 5) = saledus(T(i, 2), T(i, 3)) 

     Next i 

     T(m + 2, 1) = "keskmine" 

     T(m + 2, 2) = p_kesk(Range("pikkus")) 

     T(m + 2, 3) = p_kesk(Range("mass")) 

     T(m + 2, 4) = p_kesk(Range("indeks")) 

End Sub 

Muutuja T tüübiks on Range ja sellega seotakse piirkond 

(lahtriplokk) nimega andmed. Kasutades omadusi Rows ja 

Count tehakse kindlaks rivide arv (m) tabeli kehas. 

For-lause määrab korduse: i väärtust (rea järjenumbrit) 

muudetakse 1 kuni n ning iga i väärtuse korral täidetakse For 

ja Next vahel olevad laused. 

T(m+2, tulp) määrab lahtri, mis asub tulbas tulp (1, 2, 3, 4) ja 

rivis m+2. Tabeli all jäetakse tühjaks üks rida.  

Funktsioon p_kesk, kasutades argumendina tulba (lahtri-

ploki) nime, leiab selle keskmise. 

Näide. Korrutamistabel. For-lause 

Programm (makro) koosneb ühest Sub-protseduurist ning imiteerib korrutamise trenni. Kõigepealt 

küsitakse ülesannete arvu n, seejärel esitatakse n korrutamisülesannet ning hinnatakse vastuseid ja tehakse 

kindlaks valede vastuste arv. Programm fikseerib ka aja. 

nimi pikkus mass indeks hinnang

Juku 193 71 19,1 normis

Kalle 171 97 33,2 suur ülekaal

Mart 191 82 22,5 normis

Pets 203 120 29,1 ülekaaluline

Sass 197 64 16,5 kõhn

Tom 177 87 27,8 ülekaaluline

keskmine 188,67 86,83 24,7

Makro tabelis Arvuta

Sub Korrutustabel()     ' Korrutamise harjutamine 

  Dim n, vigu, a, b, k, _ 

           vastus, algaeg, aeg 

   n = InputBox("Mitu ülesannet? ") 

   vigu = 0 

   Randomize 

   algaeg = Timer()  ' jooksev arvutiaeg 

   For k = 1 To n     ' korduse algus 

       a = 2 + Round(7 * Rnd()) 

       b = 2 + Round(7 * Rnd()) 

       vastus = InputBox(a & " * " & b) 

       If vastus = "" Then  ' kui tühi väärtus 

            MsgBox "Kahju, et loobusid!" 

            Exit Sub   '  katkestab korduse 

       End If 

       If Int(vastus) <> a * b Then 

           MsgBox "Vale!" 

           vigu = vigu + 1 

       End If 

   Next k 

   If vigu = 0 Then 

      MsgBox "Tubli! Kõik oli õige!" 

   Else 

       MsgBox ("Vigu oli " & vigu) 

   End If 

   aeg = Round(Timer() - algaeg, 2) 

   MsgBox ("Aeg: " & aeg & " sek") 

End Sub 

Programmis on kasutusel muutujad: 

 n – ülesannete arv, 

 a, b – juhuslikud arvud vahemikus 2 kuni 9 küsimuste esitamiseks, 

 vastus – kasutaja poolt pakutav vastus, 

 vigu – vigade arv,  

 k – juhtmuutuja For-lauses,  

 aeg – ülesandele kulunud aeg. 

Korduse kirjeldamiseks kasutatakse programmis nn juhtmuutujaga kordust ehk For-lauset, mille enam-

kasutatav variant on järgmine: 

For muutuja = algväärtus To lõppväärtus 

laused 

Next muutuja 

muutujat nimetatakse siin juhtmuutujaks. Täitmisel muude-

takse järjest juhtmuutujat algväärtusest kuni lõppväärtuseni 

(sammuga üks) ja iga juhtmuutuja väärtuse korral täidetakse 

For ja Next vahel olevad laused. 

See kordus on oma olemuselt üsna sarnane Scratchi plokile [korda n], kuid viimasest veidi üldisem. Scratchi 

plokist erineb For-lause peamiselt automaatselt muudetava juhtmuutuja olemasolu poolest. Antud 

programmis juhtmuutujat k küll ei kasutata, kuid süntaksi reeglite tõttu peab juhtmuutuja alati olema. Töös 

tabelite ja massiividega kasutatakse juhtmuutujat sageli järjenumbrina (rea või veeru numbrina).  

Kuna antud programmi kasutajaliides põhineb dialoogibokside kasutamisel ning ei pruugita Exceli 

vahendeid, saab selle panna tööle ka teistes rakendusprogrammides, mis toetavad Visual Basicut.  

Näide Pranglimine. Do Until-lause  

See programm võimaldab harjutada liitmist. 

Programm esitab etteantud aja jooksul (praegu 45 sek) liitmise ülesandeid ja kontrollib vastuseid, andes 

õigete vastuste eest punkte. Kasutaja saab valida kolme taseme vahel, millest sõltub arvude suurus ja ühe 

tehte punktide arv (vt protseduuri Tee_tase). 

Töölehel on lahtrid nimedega: tase, punkte, aeg ja rekord. Programmi muutujad on loetletud Dim-lauses. 

Töölehel on ka väsimatud abilised – Juku ja Kraps, kes raiskavad veidi aega karistuseks vale vastuse eest. 

 

Sub Prangi_1() 

  ' Liitmise trenn  

  Const max_aeg = 45   ' maksimaalne aeg 

  Dim algaeg, a, b, vastus, mini, maxi, p, punkte 

   Range("punkte") = "" 

   Tee_tase Range("tase"), mini, maxi, p 

   Randomize 

   algaeg = Timer():  Range("aeg") = 0 

   Do Until Range("aeg") > max_aeg 

        Range("aeg") = Timer() - algaeg 

        a = mini + Int(Rnd() * (maxi - mini + 1)) 

        b = mini + Int(Rnd() * (maxi - mini + 1)) 

        vastus = InputBox(a & " + " & b) 

        If vastus = "" Then MsgBox "Kahju!": End 

        If Int(vastus) = a + b Then 

             Range("punkte") = Range("punkte") + p 

        Else 

             MsgBox "Vale!": Call Ai_ai 

        End If 

   Loop 

   MsgBox "Aeg on läbi " 

   If Range("punkte") > Range("rekord") Then 

      Range("rekord") = Range("punkte") 

      MsgBox "Õnnitlen! Uus rekord!" 

   End If 

End Sub 

Sub pöörle(kuju As Shape) 

    Dim i 

    For i = 1 To 360 

        kuju.IncrementRotation 1 

        DoEvents 

   Next i 

End Sub 

Sub Tee_tase(tase, mini, maxi, p) 

   If tase = 1 Then 

       mini = 1: maxi = 10: p = 2 

   ElseIf tase = 2 Then 

        mini = 10: maxi = 20: p = 3 

   Else 

       mini = 20: maxi = 30: p = 5 

   End If 

End Sub 

Sub Ai_ai() 

   pöörle Shapes("Juku") 

   pöörle Shapes("Kraps") 

End Sub 

Põhiosa peaprotseduurist moodustab kordus, mida juhitakse Do Until tingimus … Loop-lause abil. Lauseid 

korduses täidetakse seni, kuni tingimus (jooksev aeg on suurem kui max_aeg) saab tõeseks.  

Näide Jalka 

Rakendus, milles on tegemist peamiselt objektidega – kujundid (graafikaobjektid) ja töölehe lahtrid. 

Võrdluseks on toodud ka Scratchi programm. 

  
Rakenduses imiteeritakse jalgpallitrenni. Piltidel on toodud VBA (vasakul) ja Scratchi kasutajaliidesed. 

Programmi põhiobjektideks on Juku (pealelööja rollis) ja pall. Lisaks on kasutusel suhteliselt passiivne 

objekt värav – ruuduline ristkülik. Programmis kontrollitakse palli sattumisi värava piirkonda. Töölehel on 

„vaatlejana“ ka Kraps, kes on Scratchi rakenduses pealelööja rollis. Tulemuste fikseerimiseks kasutatakse 

töölehe lahtreid nimedega lööke, tabas ja protsent. Scratchis on samade nimedega muutujad.  

Sub Trenn() 

   Dim pall As Shape, J As Shape 'objektimuutujad 

   Set pall = Shapes("pall") ' objektide sidumine  

   Set J = Shapes("Juku")   'muutujatega, sama 

   Range("lööke") = 0: Range("tabas") = 0 ' algväärtused 

   Randomize  ' juhuarvude generaator algseisu 

   Do  ' lõputu korduse algus 

        pall.Left = Rnd() * 400             ' palli vasak serv  

        pall.Top = 150 + Rnd() * 150   ' palli ülemine serv 

        Call Hyppa(pall, 4, 30)  ' kutsub protseduuri Hyppa 

        J.Left = pall.Left : J.Top = pall.Top ' Juku palli juurde 

        paus 0.5 

        pall.Left = Rnd * 400 : pall.Top = Rnd * 150 ' löök 

        Range("lööke") = Range("lööke") + 1 

        If On_sees(pall, Shapes("värav")) Then 

             Range(“tabas“) = Range(“tabas“)+ 1 

             Call Hyppa(J, 2, 40)  ' kutsub protseduuri Hyppa 

             Call Hyppa(Shapes("Kraps"), 3, 20) 

        End If 

        [protsent] = [tabas] / [lööke] * 100  ' [ ]=Range( ) 

        paus 2 

   Loop  ' korduse lõpp 

End Sub 

 

Programmis on realiseeritud järgmine stsenaarium (algoritm). Käivitamisel nupuga Läks korratakse lõputult 

järgmisi tegevusi. Pall viiakse väljaku alumisse poolde, kus see teeb mõned hüpped. Juku „jookseb“ palli 

juurde, lühikesele pausile järgneb „löök“ – pall viiakse väljaku ülemisse poolde ning juhuk kui pall sattus 

väravasse, suurendatakse tabamuste arvu ning Juku ja Kraps teevad mõned hüpped. Arvutatakse 

tabamuste protsent ning kõik kordub. Programmi töö saab katkestada nupuga Stopp. 

Palli asukoha muutmiseks kasutatakse selle omadusi Left ja Top. Nende muutmiseks kasutatakse juhuarve. 

VBAs peab kasutama otsest viitamist objektidele ning nende omadustele ja meetoditele. Scratchis kuulub 

iga skript kindlale objektile (spraidile) ja saab määrata ainult selle tegevusi. VBA protseduur aga on 

objektidest sõltumatu ja saab määrata suvaliste objektide tegevusi ja muuta nende omadusi. 

VBA programm koosneb peaprotseduurist Trenn ja kolmest alamprotseduurist: Hyppa, paus ja On_sees.  

Peaprotseduuris on Dim-lausega määrtletud muutujad pall ja J. Tegemist on objektimuutujatega 

(viitmuutujatega), mida kasutatakse viitamiseks objektidele pall ja Juku. Objekt seotakse muutujaga 

Set-lausega. Standardne viitamisviis graafikaobjektidele on Shapes(“nimi“).  

Lausetega Range("lööke") = 0: Range("tabas") = 0 antakse töölehe lahtritele nimedega lööke ja tabas 

väärtus null. Standardne viitamisviis nimedega lahtritele on Range ("nimi"). Alternatiivina võib kasutada ka 

konstruktsiooni [nimi] (vt lause [protsent] = [tabas] / [lööke] * 100 protseduuri lõpus).  Sel juhul ei tohi 

lahtri nimi kokku langeda muutuja nimega samas protseduuris. 

Protseduuri põhiosa moodustab liitlausega määratletud Do…Loop lõputu kordus. Palli viimiseks platsi 

alumisse poolde kasutatakse lauseid: 

        pall.Left = Rnd() * 400            ' palli vasak serv juhuarv 0-st kuni 400-ni 

        pall.Top = 150 + Rnd() * 150  ' palli ülemine serv 150-st kuni 300-ni. 

Mõõtühikuks on point (punkt – 1/72 tolli ehk umbes 0,35 mm). Objekti omaduse muutmiseks kasutatakse 

omistamislauset kujul:  

objekt.omadus = avaldis 

Objekt on esitatud muutuja abil (pall), alternatiivina võiks olla Shapes(“pall“). Omadused esitatakse 

nimedega. VB sisefunktsioon Rnd() tagastab juhusliku reaalarvu vahemikus 0 kuni 1. Tühjad sulud 

funktsiooni nime järel võib ka ära jätta. 

Palli viimine platsi ülemisse poolde toimub omaduste ja juhuarvude abil analoogselt: 

pall.Left = Rnd * 400 : pall.Top = Rnd * 150    ' löök, palli  ülemine serv 0-st 150-ni 

NB! Ühel real võib olla mitu lauset. Eraldajaks on koolon. 

Juku viimiseks palli juurde võrdsustatakse vastavad omadused: J.Left = pall.Left : J.Top = pall.Top 

Peale palli viimist platsi ülemisse poolde suurendatakse 

löökide arvu ja kontrollitakse, kas pall on värava sees või 

mitte. Selleks kasutatakse abifunktsiooni On_sees(O_1, O_2), 

mis tagastab väärtuse True (tõene), kui objekt O_1 on 

täielikult objekt O_2 sees. Oma olemuselt vastab funktsioon 

Scratchi plokile <puudutab objekt>, ainult siin peab O_1 olema 

täielikult O_2 sees. 

Protseduur illustreerib funktsioonide kirjeldamise põhi-

mõtteid. VBA funktsioon saab leida ja tagastada ühe ja ainult 

ühe väärtuse. Tagastatav väärtus omistatakse funktsiooni 

nimele. Funktsioonil on tüüpiliselt olemas parameetrid, mis 

esindavad sisendandmeid. Viimased saavad väärtused pöördu-

misel vastavatelt argumentidelt.  

Function On_sees(O_1, O_2) 

' Tõene, kui objekt O_1 on O_2 sees 

  Dim v1, p1, y1, a1, v2, p2, y2, a2 

    v1 = O_1.Left: p1 = v1 + O_1.Width 

    y1 = O_1.Top: a1 = y1 + O_1.Height 

    v2 = O_2.Left: p2 = v2 + O_2.Width 

    y2 = O_2.Top: a2 = y2 + O_2.Height 

    If v1 > v2 And p1 < p2 And _ 

        y1 > y2 And a1 < a2 Then 

      On_sees = True 

    Else 

      On_sees = False 

    End If 

End Function 

Pauside tekitamiseks kasutatakse abiprotseduuri paus, mille poole saab pöörduda lausega: 

 paus pikkus 

pikkus  pausi pikkus sekundites. Protseduur vastab täielikult Scratchi käsule [oota]. 

Sub paus(pp) 

' paus pikkusega pp sek 

   Dim pl  ' pausi lõpp 

   pl = Timer() + pp 

   Do Until Timer() > pl 

      DoEvents 

   Loop 

End Sub 

Protseduur esindab parameetritega protseduuri. Parameeter pp saab 

väärtuse pöördumisel vastavalt argumendilt, mis annab konkreetse pausi 

pikkuse sekundites.  

Muutuja pl väärtuseks võetakse pausi lõpu jaoks väljaarvutatud aeg. VBA 

sisefunktsioon Timer(), mis on Scratchi ploki [taimer] analoog, annab arvuti 

jooksva kellaaja. Lause Do Until … Loop määrab tingimusliku korduse, mis 

vastab [korda kuni ...] plokile Scratchis. Igal kordamisel täidetakse lause 

DoEvents, mis võimaldab Windowsil pausi ajal täita ootel olevad tegevused. 

Kordus lõpeb, kui aeg ületab pausi lõpuaja pl. 

Parameetritega protseduuri näiteks on ka protseduur Hyppa, mille abil imiteeritakse objektide, milleks siin 

on Juku, Kraps ja pall, hüppamist – objekt tõuseb korduvalt (n korda) ülespoole ja laskub alla. 

Sub Hyppa(kuju, n, h) 

' objekt kuju hüppab 

   Dim i 

   For i = 1 To n        

       kuju.IncrementTop  -h 

       paus 0.2 

       kuju.IncrementTop h 

       paus 0.3 

    Next i 

End Sub  

   

 

Protseduuril on kolm sisendparameetrit: kuju – graafikaobjekt, n – hüpete (korduste) arv, h – hüppe 

kõrgus. Parameetrid saavad väärtused vastavatelt argumentidelt. Protseduuri poole pöördumine toimub 

siin kolm korda lausetega:  

Call Hyppa(pall, 4, 30)  :   Call Hyppa(J, 2, 40)  :   Call Hyppa(Shapes("Kraps"), 3, 20) 

Esimesel korral on esimeseks argumendiks (vastab parameetrile kuju) pall, teisel Juku, kolmandal Kraps. 

Kuna objekt Kraps ei ole seotud muutujaga, kasutatakse viitamiseks konstruktsiooni Shapes("Kraps"). Igal 

pöördumisel on ka parameetritele n ja h vastavatel argumentidel erinevad väärtused.  

Parameetritega protseduuri poole saab pöörduda korduvalt, kasutades erinevaid argumente. See 

võimaldab vähendada programmi mahtu.  

Scratchi praeguses versioonis puudub võimalus parameetrite ja argumentide kasutamiseks. Sellepärast 

peab moodustama (kopeerima) kolm peaaegu ühesugust skripti vastavate spraitide jaoks. Näidatud on 

neist kaks – palli ja Krapsu skriptid. Väikesed erinevused (korduste arv, hüppe kõrgus ja pausi pikkus) 

pannakse paika projekti loomise ajal.  

Exceli objektide kasutamine VBAs 

Rakenduse struktuur ja põhiobjektid 

Exceli rakendus kujutab endast omavahel seotud objektide kogumit: töövihikud, töölehed, lahtriplokid, 

graafilised kujundid jms. Ühetüübilised objektid kuuluvad ühte klassi. Igal klassil on kindel nimi: Workbook, 

Worksheet, Range, Shape jne. VBA toega Exceli rakenduses 

võib olla kasutusel mitu töövihikut. Töövihik võib koosneda 

ühest või mitmest töölehest. Töölehti saab lisada ja eemaldada 

nii „käsitsi“ Exceli vastavate käskudega kui ka VBA programmi 

käskude abil. Igal töölehel on üle 15 miljardi lahtri (!), millest 

saab moodustada praktiliselt piiramatu hulga lahtriplokke. 

Lahtriplokina on käsitletav suvaline lahtrite kogum. Töölehe 

pinnale saab paigutada praktiliselt piiramatu hulga 

graafikaobjekte ehk kujundeid. Neid saab joonestada Exceli 

juurde kuuluvate joonestusvahenditega ja importida 

graafikafailidest. Tegevusi graafikaobjektidega saab täita 

„käsitsi“ ja programmi käskudega.  

Klasse on Excelis tunduvalt rohkem, kui skeemil tooduid. Näiteks võib lisaks töölehtedele kasutada 

diagrammilehti. Diagramme saab paigutada ka töölehe pinnale, töölehele saab panna ka mitmesuguseid 

ohjureid (juhtelemente) – kerimisribad, märkeruudud, komboboksid jms. Siin puutume vähemalt esialgu 

nö programmilisel tasemel kokku ainult graafiliste kujunditega, lahtriplokkide ja ohjuritega. Piirdume ainult 

rakendustega, kus on kasutusel üks töövihik. Selles võib olla praktiliselt piiramatu hulk töölehti ning igal 

töölehel võib olla omaette rakendus. Vaatamata nimetatud piirangutele, saab luua väga sisukaid ja 

mahukaid rakendusi. Kui põhioskused omandatud, saab vajaduse korral laiendada kasutatavaid vahendeid 

ja tegevusvaldkonda. 

Scratchi skriptid on alati seotud kindla objektiga (spraidi või lavaga) ning seega pole vajadust ega isegi 

võimalust neile otseselt viidata. VBA protseduur aga ei ole seotud konkreetse objektiga ning üks protseduur 

saab määrata tegevusi mitme objektiga ning seetõttu peab alati näitama objekti, millega vastav tegevus 

täidetakse. Viitamiseks objektidele kasutatakse klassi nimesid, kinni pidades teatud reeglitest. 

Objektidel on kindel valik omadusi ja meetodeid. Ühe klassi objektidel on sama valik. Igal omadusel ja 

meetodil on kindel nimi, mida kasutatakse neile viitamiseks. Viitamisel kasutatakse erinevates 

programmeerimissüsteemides laialt levinud viisi, kus omadused ja meetodid seotakse objektiga punkti abil: 

 objekt.omadus ja objekt.meetod [argumendid] 

Objekt on viit objektile. Esitus sõltub objekti klassist ja viitamisviisist. Omadused ja meetodid esitatakse 

nimede abil. Mõnedel meetoditel võivad olla ka argumendid.  

Mõned näited: 

 Shapes(“auto“).Left = 0 kujundi auto omaduse Left väärtuseks võetakse 0. 

 Shapes(“pall“).IncrementTop 20 meetod muudab kujundi ülemise serva koordinaati 20 võrra. 

Range(“punkte“). Value = 0 lahtri punkte väärtuseks võetakse 0. 

Mõnikord kasutatakse ka alamobjekte ja nende omadusi. Näiteks graafilise kujundi värvuse määramiseks 

kasutatakse järgmist konstruktsiooni: Shapes(“ufo“).Fill.ForeColor.SchemeColor = 3. 

Täpsemalt vaadeldakse neid küsimusi vastava klassi objektide kirjeldamisel. 

Tööleht – klass Worksheet 

VBA programmide abil saab määrata tegevusi Exceli töölehtedega ja ka töövihikutega, mille koosseisu lehed 

kuuluvad, kuid neid võimalusi praegu eriti põhjalikult ei vaadelda. Töölehti ja töövihikuid kasutatakse antud 

juhul peamiselt VBA rakenduste konteineritena (hoidlatena) ning kasutajaliideste komponentidena. 

Töölehele paigutatakse graafikaobjektid, millega manipuleerivad programmid ning töölehe lahtreid 

kasutatakse programmide poolt kasutatavate andmete salvestamiseks ja säilitamiseks, aga ka  

kasutajaliideste loomisel.  

Protseduurid salvestatakse töölehe moodulisse või üldmoodulisse. Viimasel juhul 

eeldatakse, et programmi töö ajal on aktiivne kasutajaliidesega tööleht, kus asuvad 

VBA protseduuride poolt kasutatavad objektid. Ühes töövihikus võib olla realiseeritud 

mitu sõltumatut rakendust. Sellisel juhul peaks need asuma erinevatel lehtedel, kuigi 

põhimõtteliselt ei ole välistatud ka ühise lehe kasutamine. Salvestades töövihiku 

välismällu, saame säilitada nii rakenduse töö tulemused kui ka programmid.  

Siin ei vaadelda ka töölehe omaduste (näiteks nime) muutmist programmide abil. 

Seda saab teha lihtsalt klõpsates lehe lipikut (lehesakki) ja tippides vajaliku nime. 

Praktiliselt ei pea arvestama isegi töölehe selliste omadustega nagu ridade ja 

veergude arv ning lehe mõõtmed, sest need on meie lihtsate rakenduste jaoks 

meeletult suured (ridu üle miljoni, veerge ca 16 tuhat). Töölehe virtuaalne laius on ca 

300 meetrit ja kõrgus lausa 5 kilomeetrit (!). Võrrelge kasvõi Scratchi lavaga. 

Olulised on töölehe koorinaadisüsteem ja 

mõõtühikud, eriti töötamisel graafika-

objektidega. Andmed nende kohta on toodud 

pildil. Koordinaatsüsteemi nullpunkt on töö-

lehe ülemises vasakpoolses nurgas. 

Horisontaaltelg (X) on suunatud paremale, 

vertikaaltelg (Y) alla. Pikkuse ühikuks on 

trükindusest pärit üksus point (punkt), mis 

võrdub 1/72 tolliga ehk ca 0,35 millimeetriga.  

Erinevalt Scratchist, saab VBA protseduuridega teha ka nö mõõtkavas jooniseid ja skeeme, kasutades 

vajadusel ka vastavaid mõõtühikute (ja koordinaatide) teisendusi. 

Mõnikord on vaja viidata VBA programmis töölehele, selleks võib kasutada järgmisi konstruktsioone: 

 Sheets(nimi) või Worksheets(nimi), mis on samaväärsed. 

Nimi esitatakse tüüpiliselt stringkonstandina: Sheets(“Plats“), Worksheets(“rada_1“). 

Teatud määral tutvume siin töölehe sündmustega, mida saab lihtsalt kasutada paindlike ja dünaamiliste 

rakenduste loomiseks. Objektisündmuste käsitlus erinevates programmeerimissüsteemides on üsna 

ühesugune. 

Graafilised kujundid – klass Shape 

Viitamine objektidele 

Graafikaobjekte saab töölehele lisada Exceli käskudega vahelehe Insert (Lisa) grupist Illustrations 

(illustratsioonid). Käsk Shapes (kujundid) võimaldab lisada baaskujundeid: sirgjoone lõik, murdjoon, 

ristkülik, hulknurk, ovaal jm. Kujunditest saab moodustada liitkujundeid, redigeerida ja vormindada neid. 

Käsk Picture (Pilt) võimaldab importida kujundeid graafikafailidest. Graafikaobjekte saab lisada, redigeerida 

ja vormindada ka VBA käskudega. 

Tööleht 

nimi 

ridade arv 

veergude arv 

laius, kõrgus 

aktiivne lahter 

... 

aktiveerimine() 

teisaldamine() 

kopeerimine() 

eemaldamine() 

peitmine() 

... 

Kõik töölehel paiknevad objektid: MS Drawing abil tehtud joonised, imporditud pildid ja teiste rakenduste 

objektid, hõlmatud diagrammid, ohjurid ehk juhtelemendid (käsunupud, kerimisribad, märkeruudud jm), 

kuuluvad klassi Shape. Neile viitamiseks võib kasutada erinevaid viise. Üks esmaseid ja peamisi on:  

Shapes (nimi),  

kus nimi on objekti (kujundi) nimi. Tüüpiliselt esitatakse nimi tekstikonstandina, mis paigutatakse jutu-

märkide vahele. Nimi on kujundil alati olemas. Excel moodustab nime kujundi loomisel või importimisel. 

Vaikimisi on nime kuju tüüp number, kus tüüp sõltub kujundi tüübist (alamklassist) ja number on tüübist 

sõltumatu järjenumber, mis suureneb automaatselt iga uue kujundi lisamisel. Mõned näited:  

Rectangle 1, Rectangle 13, Oval 4, Straight Connector 18, Picture 7. 

Aktiivse kujundi nime võib näha nimeboksis. Viimast saab kasutada ka nime muutmiseks. Selleks peab 

muutma kujundi aktiivseks (klõpsates seda), tippima nimeboksis nime ja vajutama klahvile Enter. 

Omapoolse nime määramine on sageli otstarbekas, sest süsteemi poolt pandud nimed on tüüpiliselt üsna 

pikad ja ei ütle midagi objekti olemuse kohta antud rakenduses. Näiteks sirgjoone lõiku nimetakse 

Straight Connector (sirgkonnektor, Exceli varasemates versioonides oli Line).  

Kujunditele viitamise näited:  

Shapes(“Rectangle13”), Shapes (“ring”), Shapes (“pall”), Shapes (“Juku”)  

Kui protseduur ei asu selle töölehe moodulis, millel on kujund, peab viidale kujundile eelnema ka viit lehele. 

Seda peab näiteks kasutama siis, kui protseduur asub üldmoodulis.  

Sheets (“Mäng”).Shapes(“pall”), ActiveSheet.Shapes(“Juku”)  

Esimesel juhul viidatakse konkreetsele lehele, kasutades selle nime. Teisel juhul viidatakse aktiivsele lehele, 

kasutades rakenduse omadust ActiveSheet. 

Objekti saab Set-lause abil siduda muutujaga. See võimaldab lühendada viitu. Set-lause üldkuju on 

 Set muutuja = objekt 

Mõned näited:  Set auto = Shapes("auto") : Set J = Shapes("Juku"). 

Shape-objekti põhiomadused 

Name  nimi 

 Left  vasak serv 

 Top  ülemine serv 

 Width  laius 

 Height kõrgus 

 Rotation  pöördenurk 

 Visible  nähtavus 

Fill.ForeColor esiplaani värvus 

Kujundi täitevärvuse muutmiseks võib kasutada järgmist omaduse määrangut: 

objekt.Fill.ForeColor.SchemeColor = värvi number 

Värvikoodid on järgmised:  0 – must, 1 – valge, 2 – punane, 3 – roheline, 4 – sinine jne kuni 80.  

Proovige teha protseduur, mille abil saab näha värvikoode. 

Shape-objekti põhimeetodid 

IncrementLeft dx vasaku serva juurdekasv Select valimine (aktiveerimine) 

IncrementTop dy ülemise serva juurdekasv Cut lõikamine 

IncrementRotation dn pöördenurga juurdekasv Copy kopeerimine 

ScaleHeight k, False kõrguse skaleerimine Delete eemaldamine 

ScaleWidth k, False laiuse skaleerimine 

Näiteid 

Objekti omaduste muutmine. Töölehe suvalises kohas asub graafikaobjekt nimega kast. Protseduur viib 

kujundi punkti (0; 0) ja annab sellele kindlad mõõtmed (100 * 50 punkti). 

 

Sub Koju() 

  Shapes("kast").Rotation = 0 

  Shapes("kast").Left = 0 

  Shapes("kast").Top = 0 

  Shapes("kast").Width = 100 

  Shapes("kast").Height = 50 

  Shapes("kast").Fill.ForeColor.SchemeColor = 0 

End Sub 

Objekti sidumine muutujaga. Meetodite kasutamine. Töölehe suvalises kohas asub graafikaobjekt nimega 

kast. Protseduuri igakordsel täitmisel teeb kujund ühe sammu paremale ja alla, pöördub ning muudab 

suurust ja värvust. Koordinaatide ja pöördenurga muutused ning skaleerimise teguri võib valida suvaliselt. 

Värvuse määramiseks kasutatakse juhuarve. Et vähendada programmi teksti pikkust, on objekt seotud 

muutujaga. 

Sub Liigu_1() 

  Dim K As Shape 

  Set K = Shapes("kast") 

  K.IncrementLeft 10 

  K.IncrementTop 5 

  K.IncrementRotation 15 

  K.ScaleWidth 1.03, False 

  K.ScaleHeight 1.03, False 

  K.Fill.ForeColor.SchemeColor = Rnd() * 80 

End Sub 

Muutuja K on seotud kujundiga kast. Tänu millele lüheneb 

lausete pikkus. Võrdle eelmise näitega. Kuna muutuja K on 

deklareeritud tüübiga Shape, pakub süsteem sisestamisel 

selle objekti omaduste ja meetodite loetelu (spikrit), kui 

objektimuutuja järel sisestatakse punkt.  

Tegevuste määramisel kasutatakse peamiselt meetodeid. 

Esimesel kolmel juhul võib kasutada ka omadusi.  

Näiteks: K.IncrementLeft 10  =>  K.Left = K.Left + 10 

Kujundi sujuv liikumine. Protseduur Liigu_XY viib antud kujundi sujuvalt punkti koordinaatidega x, y. 

Protseduur on analoogne Scratchi käsuplokile [liigu t sek x: … y: …] . 

Sub Liigu_XY(kuju As Shape, x, y, d, p) 

  Dim xj, yj, L, dx, dy, i 

  xj = kuju.Left:  yj = kuju.Top 

  L = Sqr((x - xj) ^ 2 + (y - yj) ^ 2) 

  If L = 0 Then Exit Sub 

  dx = d * (x - xj) / L: dy = d * (y - yj) / L 

  For i = 1 To L / d 

      kuju.IncrementLeft dx 

      kuju.IncrementTop dy 

      paus p 

  Next i 

End Sub 

Parameetrid: kuju – graafiline kujund (klass Shape),  

x, y – sihtkoha koordinaadid, d – liikumise samm,  

p – pausi pikkus sammude vahel. Kahele viimasele 

parameetrile vastavate argumentide väärtustega saab 

reguleerida liikumise kiirust. 

Muutujad: xj, yj – kujundi algkoordinaadid,  

L – kaugus sihtkohast, dx, dy - sammu projektsioonid, 

i – abimuutuja. 

Protseduur leiab kujundi kauguse sihtkohast L ning 

sammu projektsioonid dx ja dy. Liikumist juhitakse 

For-lause abil, kus juhtmuutuja lõppväärtuseks on L/d. 

Antud protseduuri abil võiks näiteks muuta sujuvaks palli 

ja Juku liikumised näites „Jalka“. 

Lahtrid ja lahtriplokid – klass Range 

Lahtriplokina ehk Range-objektina võib käsitleda suvalist töölehe lahtrite kogumit. Üksiklahter kujutab 

endast lahtriploki erijuhtu: ühest lahtrist koosnev lahtriplokk. Esialgu piirdumegi peamiselt üksikute lahtrite 

kasutamisega. Viitamiseks lahtrile on mitmeid võimalusi, põhivariant on järgmine:  

Range ( nimi ),  

kus nimi on lahtri nimi, mis tüüpiliselt esitatakse stringkonstandina. Nimi määratakse lahtrile tavaliselt 

Exceli vahenditega, kuid seda saab teha ka VBA protseduuri abil. Mõned viitamise näited:  

Range (“a”),  Range (“pindala”),  Range(“aeg”), Range (“lööke“). 

Alternatiivina võib kasutada konstruktsiooni [nimi], kus lahtriploki nimi (ilma jutumärkideta) paigutatakse 

nurksulgude vahele: [a], [pindala], [aeg], [lööke] (protseduuris ei tohi sel juhul olla samanimelisi mujutujaid) 

ning töölehe omadust Cells(rida, veerg) või Cells(nr)  

Kui on vaja viidata teise lehe lahtritele, peab viidale lahtrile eelnema viit lehele, näiteks: 

 Sheets(“Hinnakirjad“).Range(“Arvutid“), Sheets(“Maksud“).Range(“tulumaks“) 

Lahtritele ja lahtriplokkidele nimede määramiseks on mitmeid viise. Kõige lihtsamaks on nimeboksi 

kasutamine. Nimede määramine toimub siin analoogselt graafikakujunditele nimede panemisega. Valida 

välja lahter või lahtriplokk, tippida nimeboksis nimi ja vajutada klahvi Enter. Vahelehel Formulas (Valemid) 

on käsk Name Manager (Nimehaldur), mille alusel kuvatav dialoogiboks sisaldab mitmeid võimalusi nimede 

määramiseks ja muutmiseks. Nimed saab määrata korraga mitmele lahtrile käsuga Create from Selection 

(Loo valikust). 

Range-objektil on suur hulk omadusi, järgnevalt on toodud ainult mõned olulisemad:  

Address lahtriploki (lahtri) aadress  

Name lahtriploki (lahtri) nimi  

Value lahtris salvestatud väärtus  

Formula lahtris olev valem  

Height  lahtriploki (lahtri) kõrgus 

Width lahtriploki (lahtri) laius 

Left lahtriploki vasaku serva kaugus töölehe vasakust servast  

Top lahtriploki ülemise serva kaugus töölehe ülemisest servast  

VBA programm saab lugeda töölehe lahtrites olevaid väärtusi, leida nende alusel tulemused ja kirjutada 

need vastavatesse lahtritesse. Kasutaja, sisestades lahtrisse mingi väärtuse, muudab selle omadust Value. 

Sama teeb ka VBA protseduur lahtrisse väärtuse kirjutamisel.  

Omadus Value on lahtriploki nn vaikimisi võetav omadus. See tähendab, et kui omadust objekti järel ei ole 

näidatud, arvestatakse, et tegemist on omadusega Value.  

Seega 

Range (“pikkus“).Value = 13 ja 

Range (“pikkus“) = 13 on samaväärsed 

Lahtriploki meetodeid kasutatakse näiteks toodud rakendustes suhteliselt harva. Nimetame siin vaid 

mõned: valimine (Select), kopeerimine (Copy), eemaldamine (Delete).  

Näide: Võrdse pindalaga ristkülik ja ring 

Programm leiab läbimõõdu ringi jaoks, mille pindala on võrdne antud ristküliku pindalaga ning arvutab 

ristküliku ümbermõõdu ja ringjoone pikkuse suhte. Makro muudab ka ristküliku ja ringi mõõtmeid vastavalt 

andmetele. Mõõtühikud on sentimeetrites. Selle näiteks on toodud kolm varianti. 

Variant 1 – kasutatakse ainult objektide omadusi 

Rakenduse kasutajaliides on toodud alloleval pildil. Töölehel on nimedega lahtrid andmete jaoks. 

Algandmed: ristküliku mõõtmed a ja b; tulemused: ringi läbimõõt d ja suhe ning abiandmed: ristküliku 

pindala pind ja ümbermõõt ümber. 

Samuti on töölehel graafilised kujundid: ristkülik 

ja ring nimedega rist ja ring. Olgu märgitud, et 

kujundite esialgsed mõõtmed ei ole üldse 

olulised, need paneb paika programm.  

Rakendus demonstreerib kasutajaliidese ele-

mente Exceli töölehel. Väga olulised on siin 

nimedega lahtrid, kuhu saab salvestada vajalikke 

andmeid. Vormindamiseks saab kasutada Exceli 

vahendeid. Ka graafilised kujundid lisatakse 

Exceli käskudega. 

Sub Rist_Ring_1() 

'  Võrdsete pindalaga ristkülik ja ring. Arvutused 

   Range("pind").Value = Range("a").Value * Range("b").Value  

   Range("ümber") = 2 * (Range("a") + Range("b"))   

   Range("d") = Sqr(4 * Range("pind") / 3.14159) 

   Range("suhe") = Range("ümber") / (3.14159 * Range("d")) 

 ' Kujundite mõõtmete muutmine 

   Shapes("rist").Width = Range("a") * 72 / 2.54 

   Shapes("rist").Height = Range("b") * 72 / 2.54 

   Shapes("ring").Width = Range("d") * 72 / 2.54 

   Shapes("ring").Height = Shapes("ring").Width 

End Sub 

Programm koosneb ühest protseduurist, 

milles kasutatakse ainult objektide 

omadusi. Kõik andmed (algandmed, 

tulemused ja abiandmed) on töölehe 

lahtrites. Töölehel on ka programmi 

käivitamisnupp. 

Programm loeb lahtritest a ja b külgede 

pikkused (formaalselt on lahtrite a ja b 

omadused Value), leiab nende korrutise 

ning kirjutab selle lahtrisse pind 

(formaalselt omistab väärtuse lahtri 

omadusele Value). 

Kuna omadus Value võetakse ka vaikimisi, on see edaspidi ära jäetud. Järgmised kolm lauset leiavad 

ümbermõõdu, läbimõõdu ja suhte väärtused ning kirjutavad need töölehe vastavatesse lahtritesse. 

Viimased neli lauset panevad paika kujundite mõõtmed. Arvestatakse, et mõõtmed on sentimeetrites. 

Tegur 72 / 2.54 teisendab sentimeetrid ekraaniühikutesse: 1 punkt = 1 / 72 tolli. 

Variant 2 – muutujate kasutamine 

Rakendus lahendab sama ülesannet, mida tegi eelmine variant, kuid muudab lisaks ka kujundite 

täitevärvust. Siin kasutatakse muutujaid, kuhu salvestatakse algandmete, tulemuste ja abiandmete väär-

tused, tänu millele ei ole abiandmete salvestamiseks vaja eraldi 

lahtreid töölehel. 

Viitamiseks kujunditele kasutatakse objektimuutujaid. 

Tegemist on üsna tüüpilise VBA toega rakendusega Excelis 

arvutusliku iseloomuga ülesannete jaoks. Algandmete sises-

tamiseks ja säilitamiseks kasutatakse töölehe lahtreid, kust 

programm loeb need VBA muutujatesse. Arvutuste ajal asuvad 

kõik vajalikud andmed: algandmed, abiandmed ja tulemid, VBA 

tööpiirkonna andmeplokis. Vajalikud tulemused kirjutab programm 

töölehele. Soovi korral saab andmed hõlpsasti printida. Andmete ja ka programmi säilitamiseks ei ole 

vajalikud eraldi meetmed vaid piisab töövihiku salvestamisest. Andmete vormindamise saab teha lihtsalt ja 

kiiresti Exceli vahenditega. 

Sub Rist_Ring_2() 

' Võrdsete pindalaga ristkülik ja ring 

  Const pi = 3.14159, cm_p = 72 / 2.54 

  Dim a, b, d, S, P, suhe ' muutujad 

' Algandmete lugemine töölehelt 

  a = Range("a"): b = Range("b") 

' Arvutamine 

  S = a * b: P = 2 * (a + b) 

  d = Sqr(4 * S / pi): suhe = P / (pi * d) 

' Tulemite kirjutamine töölehele 

  Range("d") = d: Range ("suhe") = suhe 

' Kujundite mõõtmete ja värvuse muutmine 

  Dim rist As Shape, R As Shape 

  Set rist = Shapes("rist"):   Set R = Shapes("ring") 

  rist.Width = a * cm_p: rist.Height = b * cm_p 

  R.Width = d * cm_p: R.Height = d * cm_p 

  rist.Fill.ForeColor.SchemeColor = 70 * Rnd() 

  R.Fill.ForeColor.SchemeColor = 70 * Rnd() 

End Sub 

Lausega Const määratletakse kaks nimega konstanti: 

pi ja cm_p. Viimast kasutatakse sentimeetrite teisen-

damiseks punktidesse. Dim lause abil on määratletud 

muutujad. Nende väärtuse tüüp ei ole näidatud ning 

VBA valib need ise. 

Algandmed loetakse töölehe lahtritest nimedega a ja 

b ning salvestatakse muutujatesse a ja b. 

Arvutamisel kasutatakse muutujatesse salvestatud 

algandmete väärtusi ja muutujaisse salvestatakse ka 

abiandmete ja tulemite väärtused. Tulemused 

kirjutatakse muutujatest töölehe lahtritesse. 

Deklareeritakse objektimuutujad rist ja R ning 

Set-lause abil seotakse nendega töölehel asuvad 

kujundid rist ja ring, mis võimaldab lühendada 

järgnevates lausetes viiteid objektidele. 

Viitamiseks graafikaobjektidele kasutatakse muutujaid ja ka omaduste väärtused on praegu salvestatud 

muutujates. Viimases kahes lauses muudetakse kujundite täitevärvust. Värvide numbrid määratakse 

juhuarvude abil. 

Objektimuutujate jaoks on tüübi esitamine Dim-lauses (siin 

As Shape) kasulik programmilausete sisestamisel. Kui tüüp on 

määratud, pakub VB editor pärast objektiviida järele punkti 

sisestamist vastava objekti omaduste ja meetodite loetelu 

(spikrit), millest saab lihtsalt ja kindlalt valida vajaliku omaduse või 

meetodi.  

Variant 3 – alamprotseduuride kasutamine 

Antud rakendus võimaldab lisaks eelmistes tehtutele kasutada mastaapi, muuta algandmeid spinnerite abil 

ja paigutada kohakuti ristküliku ja ringi keskpunktid. 

Programm demonstreerib parameetrite nimede 

kasutamist objektidele viitamisel. Samuti tutvustatakse 

siin sündmuste ja sündmusprotseduuride kasutamist. 

Sub Rist_Ring_3() 

' Võrdse pindalaga ristkülik ja ring 

  Const pi = 3.14159 

  Dim a, b, S, P, d, mas 

  a = Range("a"):  b = Range("b") 

  mas = Range("mastaap") 

  S = a * b:    P = 2 * (a + b) 

  d = Sqr(4 * S / pi) 

  Range("d") = d 

  Range("suhe") = P / (pi * d) 

  Tee_skeem Shapes("rist"), _ 

        Shapes("ring"), a, b, d, mas 

End Sub 

Sub Tee_skeem(rist As Shape, R As Shape, a, b, d, m) 

' Kujundite mõõtmete ja asukoha muutmine 

  Const cm_p = 72 / 2.54 

   rist.Width = a * cm_p / m 

   rist.Height = b * cm_p / m 

   R.Width = d * cm_p / m 

   R.Height = R.Width 

   R.Left = rist.Left + rist.Width / 2 - R.Width / 2 

   R.Top = rist.Top + rist.Height / 2 - R.Height / 2 

   rist.Fill.ForeColor.SchemeColor = 70 * Rnd() 

   R.Fill.ForeColor.SchemeColor = 70 * Rnd() 

End Sub 

Programm koosneb kahest protseduurist. Peaprotseduur loeb töölehelt algandmed, teeb arvutused ja 

käivitab alamprotseduuri Tee_skeem, andes viimasele argumentidena viidad graafikaobjektidele rist ja ring 

ning nende mõõtmed ja mastaabi. Viitamisel kujunditele kasutatakse parameetrite nimesid ja see on täiesti 

analoogne muutujate kasutamisele. 

Sündmused ja sündmusprotseduurid 

Süsteem reageerib mõnede objektidega seotud sündmustele (näiteks hiireklõps), käivitades vastava 

sündmusprotseduuri, kui selline on varem ette valmistatud. Selles jaotises vaatleme lühidalt töölehe ja 

mõnede AxtiveX juhtelementide (käsunupud, spinnerid, …) sündmusprotseduure. 

Private Sub CommandButton1_Click() 

   Call Rist_Ring_3  ' võib lihtsalt Rist_Ring_3 

End Sub 

 

Private Sub CommandButton2_Click() 

    MsgBox "Tere!" 

    nimi = InputBox("Mis on Sinu?") 

    MsgBox nimi & "? See on väga ilus nimi!" 

End Sub 

Jaotises „Käsunuppude loomine ja kasutamine“ 

vaadeldi makro sidumist ActiveX-käsunupuga. 

Selleks kasutatakse sündmusprotseduuri päisega  

Sub CommandButtonN_Click() 

N on nupu järjenumber (CommandButtonN on 

ActiveX-objekti nimi).  

 

Kui käivitamist vajav protseduur on juba olemas, paigutatakse sündmusprotseduuri ainult pöördumislause 

– käivitatava protseduuri nimi.  

Soovi korral võib vajalikud laused paigutada otse sündmusprotseduuri sisse nagu on näidatud protseduuris 

CommandButton2_Click(). 

Käsunupu jaoks on veel terve rida sündmusi, millele see võib reageerida vastava protseduuri abil. Igal 

sündmusel on kindel nimi: DblClick, MouseDown, MouseUp jms, mis ilmub automaatselt sündmus-

protseduuri malli päises, kui sündmus valitakse loetelust.  

Analoogselt toimub sündmusprotseduuride kasutamine teiste töölehele paigutatavate ohjurite jaoks. Olgu 

märgitud, et ohjureid saab kasutada ka ilma VBA kasutamiseta. 

Private Sub Spinner_a_Change() 

   Rist_Ring_3 

End Sub 

 

Private Sub Spinner_b_Change() 

   Rist_Ring_3 

End Sub 

Eespool toodud näites olid kasutusel spinnerid , mille abil 

saab muuta lahtrite a ja b väärtusi. Spinner seotakse lahtriga 

omaduse LinkedCell abil.  

Sündmusprotseduur Change võimaldab käivitada vastavad 

tegevused kohe, kui muudetakse lahtri väärtust 

(formaalselt spinneri väärtust).  

Ka töölehega on seotud terve rida sündmusi, mis võimaldavad käivitada vastavad sündmusprotseduurid: 

 Activate – lehe aktiveerimine, 

 Deactivate – lehe deaktiveerimine (lahkumine antud lehelt), 

 Change – muutus mingi lahtri väärtus, 

 SelectionChange – muutus aktiivseks uus lahter. 

Private Sub Worksheet_Activate() 

   MsgBox "Tere! Kuidas hästi elad?" 

   MsgBox "Hakkame tööle!" 

End Sub 

Kui tööleht, mille moodulis asub antud sündmusprotseduur, 

muutub aktiivseks, käivitub sündmusprotseduur 

Worksheet_Activate ja kuvatakse toodud teated.  

Private Sub Worksheet_Deactivate() 

     Salvesta 

     End 

End Sub 

Kui lahkutakse antud töölehelt, rakendub 

sündmusprotseduur Worksheet_Deactivate, käivitatakse 

protseduur Salvesta ja pärast selle töö lõppu katkestatakse 

käsuga End kõikide protseduuride töö. 

Sündmus Change tekib, kui muutub töölehe mõne lahtri sisu. Uue väärtuse võib lahtrisse kirjutada kasutaja 

või programm. Parameetri Target väärtuseks saab protseduur muudetud lahtri. 

Private Sub Worksheet_Change(ByVal Target As Range) 

   If Target = Range("a") Or Target = Range("b") Or _ 

        Target = Range("maastap") Then  Rist_Ring_3 

   End If 

End Sub 

Kui muudetud lahter oli a, b või mastaap, 

käivitatakse programm Rist_Ring_3.  

Kui taolist kontrolli ei tehta, tekib töölehele kirjutamisel korduv sündmusprotseduuri käivitamine.  

Sündmus SelectionChange tekib, kui muutub aktiivne lahter. Lahtrikursori asukohta muudetakse kas 

hiireklõpsuga, nooleklahvidega või programmis käsuga Select. Uue aktiivse lahtri (Range-objekt) saab 

protseduur parameetri Target väärtusena. Viimase omadusi ja meetodeid saab kasutada programmis. 

Private Sub Worksheet_SelectionChange(ByVal Target As Range) 

    Dim J As Shape 

    Set J = Shapes("Juku") 

    J.Left = Target.Left + Target.Width / 2 - J.Width / 2 

    J.Top = Target.Top - J.Height + Target.Height 

End Sub 

Kui muutub aktiivse lahtri asukoht, 

viiakse kujund Juku aktiivse lahtri 

kohale. Seega liigub Juku koos 

lahtrikursoriga. 

NB! Sündmusprotseduuride käivitamisevõimaluse saab välja (ja uuesti sisse) lülitada objekti Application 

omaduse EnableEvents abil. Kui Application.EnableEvents = False, siis sündmusprotseduurid ei käivitu. 

Töö tabelitega 

Tabel asub töölehe ristkülikukujulises sidusas lahtriplokis. Rakenduse loomisel peab arvestama, et tabel 

võib asuda töölehe suvalises piirkonnas ning erinevatel kasutamiskordadel võivad tabelil olla erinevad 

mõõtmed. Tegevuste määramiseks tabelitega on sageli vaja viidata programmis tervele tabelile ja/või selle 

elementidele: lahtrid, rivid (read), tulbad (veerud), suvalised osad tabelis.  

Andmete määratlemiseks on otstarbekas anda tabelile või selle mingi(te)le lahtri(te)le nimed.  

Viit- ehk objektimuutuja kasutamine (omistamine Set-lausega) võimaldab kasutada programmis oluliselt 

lühemaid lauseid.  

Tabeli mõõtmete ja asukoha määramisel saab kasutada lahtriploki vastavaid omadusi. 

Omadus Selgitus, näited 

Rows / Columns piirkonna kõik rivid / tulbad 

Rows(rn) 

Columns(tn) 

rivi/tulp etteantud numbriga: prk.Rows(1), prk.Rows(i),  

prk.Columns(13), prk.Columns(k+3) 

Cells piirkonna kõik lahtrid: k = prk.Cells.Count: Range(“mass“).Cells.Count 

Cells(rn, tn) lahter rivis rn, tulbas tn:  

x=prk.Cells(1, 1): For i=1 To m: S = S + T.Cells(i, 3): Next i  

.Cells võib jätta ära: x = prk (1, 1): For i=1 To: S=S+ T(i, 3): Next i 

CurrentRegion sidus mittetühi ristkülikukujuline piirkond lahtri (lahtriploki) ümber:  

Set T = Range(“talg“).CurrentRegion 

Offset(rnihe, tnihe) nihutatud piirkond: T.Offset(1, 2), T.Offset(1, 0) 

Resize(m, n) mõõtmete muutmine: T.Resize(5, 4), T.Resize(1, n)-tulp, T.Resize(m, 1)-rivi 

Cells.Count 

Rows.Count 

Columns.Count 

lahtrite, ridade, veergude arv piirkonnas (tabelis): 

k = Range(“nimed“).Cells.Count: m = A.Rows.Count: n = A.Columns.Count 

Vaatame jaotises „Makro kasutamine tabelis“ toodud ülesannet, kus leiti kehamassiindeks, saledus ning 

pikkuse, massi ja indeksi aritmeetiline keskmine mitme isiku jaoks. Erinevalt eelmisest, kus tabeli ja selle 

elementide (tulpade) identifitseerimiseks kasutati mitut nime, on siin mängus ainult üks nimi: tabeli 

esimese lahtri (ülemine vasakpoolne lahter) nimi tabalg. Tabeli määratlemisel mängib olulist rolli omadus 

CurrentRegion, mis määratleb mittetühja piirkonna lahtri (või lahtriploki) ümber. See lihtsustab üsna 

oluliselt tööd nn dünaamiliste (muutuvate mõõtmetega) tabelitega. Tulpade määratlemisel kasutatakse 

omadusi Offset ja Resize. 

Sub Statistika1() 

    Dim T As Range, m, i 

    Set T = Range("tabalg").CurrentRegion 

    m = T.Rows.Count 

    For i = 2 To m 

       T(i, 4) = kind(T(i, 2), T(i, 3)) 

       T(i, 5) = saledus(T(i, 2), T(i, 3)) 

     Next i 

     T(m + 2, 1) = "keskmine" 

     T(m + 2, 2) = p_kesk(T.Offset(1, 1).Resize(m - 1, 1)) 

     T(m + 2, 3) = p_kesk(T.Offset(1, 2).Resize(m - 1, 1)) 

     T(m + 2, 4) = p_kesk(T.Offset(1, 3).Resize(m - 1, 1)) 

End Sub 

Omadusega CurrentRegion määratud piirkond 

seotakse muutujaga T.  

Lause m = T.Rows.Count leiab rivide arvu tabelis 

m. NB! Sisaldab ka päiserida. Korduses on 

esimese rivi numbriks 2: tabeli keha esimene 

rivi.  

Omadusega Offset(1, tnihe) määratakse piir-

kond, mille algus on 1 rivi allpool ja tnihe tulpa 

paremal T algusest.  

Mõõtmed on samad nagu T-l (m * 5). Neid 

muudetakse omadusega Resize(m-1, 1). 

Saadud tulbad on argumendiks pöördumistel funktsiooni p_kesk poole. 

Andmed ja avaldised VBAs 

Andmete liigid ja tüübid 

Andmete põhiliikideks VBAs on stringid (sõned ehk tekstid), arvud, ajaväärtused ja tõeväärtused. Iga 

andmeliigi jaoks on määratletud võimalikud väärtused ja nende diapasoon, lubatavad tehted ja 

operatsioonid.  

Arvutisüsteemides on iga andmeliigi jaoks ette nähtud kindel esitusviis ehk vorming, mida kasutatakse 

väärtuste salvestamiseks mälus ja operatsioonide täitmisel protsessoris. Vorminguga on määratletud 

väärtuste salvestamiseks eraldatavate mäluväljade struktuur ja pikkused. Ühe andmeliigi jaoks võib olla 

kasutusel mitu erinevat vormingut ehk tüüpi. 

Stringi väärtuseks võib olla suvaline märgijada, mida saab arvutis esitada. Märkide valik ja nende 

arvutisisene esitus põhineb standarditega fikseeritud kodeerimissüsteemidel. Kasutatakse ASCII või 

Unicode kooditabelite süsteemi. ASCII süsteemis vastab igale märgile kindel 8-bitine arv (kood), mis 

esitatakse ühe baidi abil, Unicode'is – 16 bitine (2 baiti). Stringi (sõne) pikkusele (märkide arvule stringis) 

VBs praktikas segavaid piiranguid ei ole.  

Arvude salvestamiseks kasutatakse erinevaid vorminguid. Need võivad olla esitatud üldises tekstivormingus 

(ASCII-koodis), kus igale numbrile eraldatakse üks bait. Kuna aga erinevate märkide hulk on arvude esituses 

üsna väike (numbrid, arvu märk ja võimalik murdosa eraldaja), siis on nende salvestamiseks ja töötlemiseks 

ette nähtud erivormingud, mis on üldisest tekstivormingust ökonoomsemad. Täisarvude ja reaalarvude 

jaoks kasutatakse fikseeritud pikkusega välju ning erinevaid esitusviise.  

Täisarvud teisendatakse arvutis kahendsüsteemi ning esitatakse kahendnumbrite (bittide) jadana, ühte bitti 

kasutatakse arvu märgi esitamiseks. Arvu maksimaalne väärtus sõltub temale eraldatud välja pikkusest 

max = 2n-1 - 1, kus n on välja pikkus bittides. Kasutatakse kahe- ja neljabaidiseid välju (16 või 32 bitti), 

millele vastavad arvude maksimaalsed väärtused 215 - 1 = 32 767  ja  231 - 1 = 2 147 483 647.  

Reaalarvud esitatakse mantissi ja eksponendi abil: arv = m•pn, kus m on mantiss (arvu tüvi), n – eksponent 

(astendaja) ja p - arvusüsteemi alus (2, 10 või 16). Mantiss esitab arvu numbreid, eksponent koma mõttelist 

asukohta. Kasutatakse nelja- ja kaheksabaidiseid välju, millele vastavad esitustäpsused 6–7 numbrikohta 

(ühekordne täpsus) ja 15–16 numbrikohta (topelttäpsus) ning maksimaalsed väärtused 1037 ja 10307.  

Ajaväärtus koosneb üldjuhul kuupäevast ja kellaajast. Need salvestatakse ühe reaalarvuna. Arvu täisosa 

näitab päevade arvu alates 01.01.1900, murdosa esitab kellaaja alates keskööst. VBs saab kasutada aja-

väärtusi alates 01.01.100 kuni 31.12.9999. 

Tõeväärtusi on kaks: tõene (True) ja väär (False). VBs esitatakse need täisarvudena – väärtusele väär 

vastab 0, väärtusele tõene vastab tavaliselt -1, kuid sellena käsitletakse ka suvalist nullist erinevat väärtust.  

Programmi koostamisel saab määrata muutujate väärtuste jaoks sobivad esitusviisid ehk tüübid, kasutades 

spetsiaalseid deklareerimislauseid, millest peamine on Dim-lause.  

Järgnev Dim-lause deklareerib nelja muutuja väärtuste tüübid 

Dim n As Integer, a As Single, b As Double, nimi As String  

Lause määrab, et muutuja n väärtusteks võivad olla täisarvud, a väärtusteks ühekordse täpsusega 

reaalarvud, d väärtusteks topelttäpsusega reaalarvud ning muutuja nimi väärtusteks stringid. Arvmuutujate 

jaoks tulenevad siit ka väljade pikkused: n – 2 baiti, a – 4 baiti ning b – 8 baiti. Stringmuutuja nimi välja 

pikkus on muutuv ning sõltub talle omistatavatest väärtustest. 

Andmetüüp määratletakse deklaratsioonis võtmesõna abil. Igale tüübile vastab kindel väärtuste esitusviis 

ja diapasoon ning välja pikkus. Mitme andmetüübi määramiseks võib kasutada ka tüübitunnuseid - kindla 

tähendusega sümboleid, mis lisatakse muutuja nime lõppu. Näiteks on järgnev lause samaväärne eelmisega 

Dim n%, a!, b#, nimi$  

Andmete põhitüübid VBAs 

Tüübi 

nimetus 

Tüübi 

tähis 
Väärtuse tüüp 

Välja pikkus 

(Bait) 
Väärtuste diapasoon 

Integer % Täisarv 2 -32 768 ... 32 767 

Long & Pikk täisarv 4 
-2 147 483 648 ... 

 2 147 483 647 

Single ! 
Ühekordse 

täpsusega reaalarv 
4 

suurim väärtus umbes 1037 

täpsus 6..7 numbrikohta 

Double # 
Topelttäpsusega 

reaalarv 
8 

suurim väärtus umbes 10307 

täpsus 15..16 numbrikohta 

String $ String (tekst) 
märkide arv 

+ 10 B 
0 kuni ca 2 mln märki 

Date  Kuupäev 8 01.01.100 ... 31.12.9999 

Boolean  Tõeväärtus 2 True või False 

Variant 

arvuga 
 

Suvaline arvu 

tüüp 
16 

suurim väärtus umbes 10307 

täpsus 15...16 numbrikohta 

Variant 

stringiga 
 String (tekst) 

märkide arv 

+ 22 B 
0 kuni umbes 2 mln märki 

Objekti-

muutujad 
 

Viit objektile. 

Määratletakse 

Set-lausega 

4  

Vaikimisi, s.t kui muutuja tüüp ei ole deklareeritud, võetakse tema tüübiks universaalne andmetüüp 

Variant. Taolistele muutujatele võib omistada suvalist tüüpi väärtusi. Väljade pikkused võetakse nende 

jaoks varuga ning väärtuse esitusviisi valib interpretaator sõltuvalt omistatava väärtuse tüübist. Näiteks 

arvu salvestamiseks, sõltumata tema tüübist ja suurusest, eraldatakse alati väli pikkusega 16 baiti. Variant-

tüübi kasutamine arvandmete jaoks on üldiselt ebaökonoomne nii mäluruumi kasutamise kui ka programmi 

töökiiruse poolest. Selle tüübi korral on vajalik mälumaht keskmiselt 3...5 korda suurem ning arvutuste 

kiirus 2...3 väiksem kui spetsiaalselt arvude jaoks ettenähtud tüüpide kasutamisel. Objektimuutujate jaoks 

määratakse tüüp klassi nime abil: Range, Shape, Worksheet jne.  

Skalaarandmed 

Konstandid 

Konstandi väärtus näidatakse programmis ning ta ei muutu programmi täitmise ajal. Iga andmeliigi jaoks on 

ette nähtud kindlad konstantide esitamise reeglid. 

Arvkonstandid esitatakse tavaliste kümnendarvudena või kümne astmega. Reaalarvudes kasutatakse 

murdosa täisosast eraldamiseks punkti.  

13     -345   647.234     -35.67       2.1E6 = 2.1106   1e-20 =10-20 

Stringkonstandis kasutatakse suvalisi märke, mida saab arvutisüsteemis esitada. Konstandi väärtus 

paigutatakse jutumärkide vahele.  

"a"    "Pindala"   "x1="    "Mis on Sinu nimi?"    “Ei” 

Ajakonstant sisaldab üldjuhul kuupäeva ja kellaaega. Konstant paigutatakse numbrimärkide (#) vahele. 

Väärtuste esitamiseks kasutatakse erinevaid vorminguid, põhivariant on järgmine: # kk/pp/aa tt:mm:ss # 

# 05/13/99 14:23:45 #        # 12/24/99 #       # 13:45 # 

Tõeväärtusi on ainult kaks, need esitatakse võtmesõnade True (tõene) ja False (väär) abil. 

Võib kasutada nimeta ja nimega konstante. Nimeta konstant esitatakse otse avaldises või lauses 

5 * (a^2 + b^2),     3.14159 * d^2 / 4, "Summa=" & S 

Nimega konstant deklareeritakse (määratletakse) Const-lause abil, mille struktuur on järgmine: 

Const nimi  [As tüüp] = väärtus { , nimi  [As tüüp] = väärtus } 

tüüp esitatakse tüübi nimetust tähistava võtmesõna abil Integer, Long, Single, Double, String, Boolean, 

Date või Variant. Kui tüüp pole näidatud, siis võetakse selleks Variant. Olgu märgitud, et tüübi otsest 

määratlemist konstantide jaoks kasutatakse võrdlemisi harva. Deklaratsioonide näiteid 

Const pi = 3.14159, Nmax = 1000, pea = "Viktoriin"  

Konstandi nime võib kasutada erinevates avaldistes ja lausetes viitamiseks vastavale väärtusele  

L = 2 * pi * r : S = pi * r^2.  

Deklaratsioonis konstandi väärtuse muutmisel muutub see (tema skoobi piires) kõikjal, kus tema nimi 

esineb. Erinevalt muutujast ei saa nimega konstandile omistada väärtusi programmi täitmise ajal. 

Konstante võib deklareerida protseduuri sees ning mooduli alguses väljaspool protseduure. Esimesel juhul 

saab neid kasutada ainult protseduuris, kus need on deklareeritud. Teisel juhul saab neid kasutada kõikides 

antud mooduli protseduurides, kus sama nime pole deklareeritud muuks otstarbeks. 

VBAs on rida sisekonstante. Igal taolisel konstandil on kindel tähendus, väärtus ja nimi. Neid kasutatakse 

kindlates kohtades ja kindlas tähenduses, peamiselt mõne sisefunktsiooni argumendina. Kõikide 

sisekonstantide nimed algavad tähtedega vb:  

vbBlack, vbRed, vbOK, vbCancel, vbYes, vbSunday 

Muutujad  

Muutujad võib jagada kolme rühma: 

 lihtmuutujad ehk skalaarmuutujad, 

 objektimuutujad ehk viitmuutujad, 

 struktuurmuutujad. 

Lihtmuutuja (ehk lihtsalt muutuja) on nimega varustatud koht arvuti mälus - järjestikku baitide rühm: 

mäluväli ehk mälupesa, kuhu programm saab täitmisel ajal salvestada väärtusi (arve, tekste jm) ja hiljem 

kasutada neid näiteks uute väärtuste leidmiseks või ka muuks otstarbeks.  

Väljad muutujatele eraldab VBA interpretaator oma tööpiirkonna andmeplokis. Välja pikkus (baitides) 

sõltub muutuja tüübist, mis määratakse programmis otseselt või kaudselt. Igal ajahetkel saab muutujal olla 

ainult üks väärtus. Väärtuse salvestamist antud muutuja pesas nimetatakse muutujale väärtuse 

omistamiseks. Seni kuni muutujale pole omistatud väärtust, on see määramatu.  

Objektimuutujale eraldatakse samuti VBA tööpiirkonnas mäluväli ehk pesa. Välja pikkus on alati neli baiti. 

Erinevalt tavamuutujast (ehk lihtmuutujast) ei salvestata sinna väärtusi, vaid nn viit objektile: objekti 

omaduste vektori aadress mälus. Tänu selle, saab muutuja nime abil viidata objektile ning selle omadustele 

ja meetodile. See on reeglina oluliselt lühem ja lihtsam, kui otsene viitamine. 

Struktuurmuutujale (massiivmuutuja jm) vastab mitu omavahel seotud ja teatud organisatsiooni omavat 

mälupesa (elementi). Massiivmuutujaid ehk massiive käsitletakse jaotises „Massiivid“. 

Protseduuride sisemuutujad võib kasutusele võtta deklareerimata, esitades nende nimed lausete sellistes 

kohtades, kus muutujate kasutamine on lubatud (näiteks omistamislausetes). Selle kohta öeldakse, et 

muutujad on deklareeritud kaudselt. Kui deklaratsioonid puuduvad, teeb VBA interpretaator konteksti järgi 

kindlaks kõik protseduuris esinevad muutujad, eraldab neile väljad ja kasutab neid protseduuri täitmise ajal. 

Muutuja tüübiks võetakse Variant.  

Enamasti on otstarbekas määratleda muutujad otse, kasutades vastavaid deklaratsioonilauseid. 

Deklaratsiooniga saab määrata järgmised muutuja omadused:  

 nimi, 

 tüüp, 

 skoop ehk määramispiirkond, 

 väärtuste eluiga.  

Peamiseks lauseks muutujate deklareerimiseks on Dim-lause, mille kuju on järgmine: 

Dim nimi  As tüüp  [ , nimi  As tüüp  ] … 

kus tüüp esitatakse lihtmuutujate jaoks tüübi nimetust tähistava võtmesõna Integer, Long, Single, Double, 

String, Boolean, Date või Variant abil. Objektimuutujate jaoks esitatakse tüüp klassi nime abil: Range, 

Shape, Worksheet jmt. Kui tüüpi pole näidatud, siis võetakse selleks Variant.  

Deklaratsioonide näiteid 

Dim a As Single, b As Single, n As Integer, x As Double,  t As Boolean 

Dim enimi As String, pnimi As String, sünniaeg As Date, palk As Double 

Mõnede tüüpide korral saab muutuja tüübi Dim-lauses määrata ka spetsiaalsete tüübitunnuste abil: 

$ - String, % -  Integer, & - Long, ! –  Single, # -  Double. 

Kasutades tüübitunnuseid, võib ülaltoodud Dim-laused esitada kompaktsemalt järgmiselt:  

Dim a!, b!, n%, x#, t As Boolean 

Dim enimi$, pnimi$, sünniaeg As Date, palk#  

NB! Tõeväärtustel ja ajaväärtustel tüübitunnused puuduvad, nende jaoks kasutatakse tüübi määramiseks, 

kui vaja, vastavaid võtmesõnu.  

NB! Tüüp näidatakse iga muutuja jaoks eraldi tema nime järel. Näiteks kehtib lauses  

Dim a, b, c As Double, d, e As Double 

tüüp Double ainult muutuja c ja e jaoks, teiste muutujate (a, b ja d) tüübiks võetakse Variant. Võtmesõnade 

kasutamisel kipuvad Dim-laused venima üsna pikaks, kui muutujaid on rohkem. Sellepärast on 

tüübitunnuste kasutamine sageli mugavam. Näiteks eelpool toodud lause, kui tahetakse kõikide muutujate 

jaoks määrata tüüp Double, näeb tüübitunnuse kasutamisel välja järgmiselt: 

Dim a#, b#, c#, d#, e# 

Järgnev Dim-lause  

Dim J As Shape, pall As Shape, e_tab AS Range 

määratleb kaks objektimuutujat viitamiseks graafikaobjektidele (J ja pall) ja ühe muutuja viitamiseks 

lahtriplokile (e_tab). 

Et objektimuutujaid saaks kasutada viitamiseks, peab need tingimata siduma Set-lausete abil vastavate 

objektidega, näiteks: 

Set J = Shapes(“Juku“) : Set pall = ActiveSheet.Shapes(“Juku“) 

Set e_tabel = Range(“Edetabel“) 

Kuigi muutujate deklareerimine ei ole VBs kohustuslik, on soovitav seda siiski teha. Kusjuures lihtmuutujate 

jaoks ei ole tüüpide määratlemine eriti vajalik, sest saavutav mõnebaidine kokkuhoid pole tänapäeva 

arvutite jaoks nimetamisväärne. Hõlbustamaks tähevigade leidmist nimedes, on otstarbekas siiski dekla-

reerida kõik muutujad (võib ilma tüüpideta) ja paigutada mooduli algusesse korraldus: 

Option Explicit 

mis määrab, et muutujate deklareerimine on antud moodulis kohustuslik. Kui moodulis esineb dekla-

reerimata muutuja, väljastab VB käivitamisel veateate:  

  Variable not defined – muutuja ei ole defineeritud – ja märgistab deklareerimata muutuja nime.  

Objektimuutujate korral on tüübi (ehk klassi nime) esitamine Dim-lauses väga kasulik. Kui protseduuri teksti 

koostamisel sisestada peale objektiviita punkt, näitab redaktor antud klassi objektide omaduste ja 

meetodite loetelu, kust saab valida vajaliku elemendi. See hoiab kokku aega ja vähendab vigade võimalust.  

Andmete skoop ja väärtuste eluiga 

Andmete skoop (ingl scope) määratleb protseduurid ja moodulid, millest on juurdepääs andmeüksusele 

(parameetrile, nimega konstandile või muutujale) tema nime abil. Eristatakse kolme taset: 

 protseduuri tase, 

 mooduli tase, 

 projekti tase.  

Protseduuritasemega andmeid nimetatakse ka lokaalseteks andmeteks, mooduli- ja projektitasemega 

andmeid globaalseteks andmeteks. 

Parameetrite jaoks on skoobiks alati ainult see protseduur, mille päises (Sub- või Function-lauses) need on 

esitatud (deklareeritud) ning seda muuta ei saa. Konstantide ja muutujate skoope saab määrata 

deklaratsioonidega. 

Protseduuritaseme andmeid ehk lokaalseid andmeid saab kasutada ainult antud protseduuris. Need 

deklareeritakse protseduuri sees: konstandid Const-lause, muutujad Dim-lausega. Lihtmuutujaid saab 

kasutada protseduuri sees ka deklareerimata. Antud protseduuris on deklareerimata muutuja lokaalne siis, 

kui samanimelist muutujat või konstanti pole deklareeritud globaalsel tasemel.  

Moodulitaseme andmed ehk mooduli ühisandmed deklareeritakse mooduli alguses. Nendele on juurde-

pääs kõikidest mooduli protseduuridest, milles ei ole deklareeritud samanimelist andmeüksust. Konstandid 

deklareeritakse Const-lausega, muutujad Dim-lausega.  

Projektitaseme andmeid saab kasutada kõikides moodulites ja protseduurides, kus ei ole deklareeritud 

samanimelist andmeüksust. Need deklareeritakse suvalise üldmooduli alguses lausega Public. 

Muutujate väärtuste eluiga tähendab perioodi, mille jooksul need väärtused eksisteerivad. Lokaalsete 

muutujate väärtuste eluiga on vaikimisi antud protseduuri täitmise aeg. Pärast protseduuri töö lõppu tema 

sisemuutujate väärtused kustutatakse. Et muutuja väärtus säiliks, peab tema deklareerimiseks kasutama 

Dim-lause asemel Static-lauset. Globaalsete muutujate väärtused säilivad seni, kuni täidetakse End-lause 

või suletakse rakendust sisaldav fail. 

Avaldised ja VBA sisefunktsioonid 

Avaldiste struktuur ja liigid 

Avaldis määrab, mis tehted (operatsioonid) ja millises järjekorras on vaja väärtuse leidmiseks täita. Üldjuhul 

koosneb avaldis:  

 operandidest, 

 tehtesümbolitest, 

 ümarsulgudest.  

Erijuhul võib avaldis koosneda ainult ühest operandist. Operandideks võivad olla:  

 konstandid, 

 lihtmuutujad, 

 objektide omadused, 

 struktuurmuutujate elemendid, 

 funktsiooniviidad.  

Nimeta konstandi väärtus (arv, string jm) esitatakse otse avaldises. Nimega konstant deklareeritakse 

lausega Const ning avaldises kasutatakse selle nime.  

Lihtmuutujad esitatakse samuti nimede abil, struktuurmuutujate elementide esitus sõltub andmekogumi 

liigist (massiiv, kirje jmt). Objekti omadus esitatakse kujul:  

objekt [.omadus] 

kus omadus esitatakse vastava nime abil: Value, Left, Top, Address jne. Vaikimisi võetava omaduse võib ära 

jätta, näiteks omadus Value Range-objektil. 

Funktsiooniviit esitatakse kujul: 

nimi (argument [, argument ] …) 

kus nimi on VBA sisefunktsiooni või kasutajafunktsiooni nimi (Sin, Sqr, CDbl jmt). VBA funktsioonide nimed 

ei kuulu reserveeritud võtmesõnade hulka, kuid muuks otstarbeks neid kasutada ei ole mõistlik, sest see 

võib tekitada segadust. Funktsiooniviites esinev argument näitab funktsioonile edastatavat väärtust. 

Argumendid võivad olla esitatud avaldiste abil. Argumentide arv, tüüp ja esitusjärjekord sõltuvad 

konkreetsest funktsioonist. 

Tehted (operatsioonid) ja tehtemärgid (operaatorid):  

 aritmeetika; ^ ,  * , / , \ , Mod , + , -   

 stringitehted;    & või + 

 võrdlused;   = , <> , < , <= , > , >=   

 loogika; Not, And, Or  

Tehete liigid on siin toodud prioriteetide kahanemise järjekorras. Aritmeetika- ja loogikatehete prioriteedid 

kahanevad vasakult paremale. Avaldise väärtuse leidmisel arvestatakse tehete prioriteete liikide vahel ning 

aritmeetika- ja loogikatehete puhul ka liigi sees. Üldjuhul võivad avaldises esineda tehted kõikidest liikidest. 

Avaldises  a + b > c And a + c > b And b + c > a esinevad aritmeetika-, võrdlus- ja loogikatehted. Väärtuse 

leidmisel täidetakse kõigepealt aritmeetika-, siis võrdlus- ning lõpuks loogikatehted. 

Tehete järjekorra muutmiseks võib kasutada ümarsulge. Sulgudes asuva avaldise väärtus leitakse eraldi. 

Ümarsulgudes esitatakse ka funktsiooniviitade argumendid.  

Sõltuvalt andmete liigist, kasutatavatest operatsioonidest ja leitava väärtuse liigist võib avaldised jagada 

järgmistesse rühmadesse: arvavaldised, tekst- ehk stringavaldised, loogikaavaldised.  

Arvavaldised ja matemaatikafunktsioonid 

Arvavaldiste operandide väärtusteks on arvud ning neis kasutatakse aritmeetikatehteid ning funktsioone, 

mis tagastavad arvväärtusi. Aritmeetikatehted ja nende prioriteedid on järgmised. 

Prioriteet Tehtesümbolid (operaatorid) Selgitus 

1 ^ Astendamine  a^n  

2 - Unaarne miinus  -a * -b + c 

3 * , / Korrutamine, jagamine  a * b,  a / b 

4 \ Jagatise täisosa  a \ b,  13 \ 5 = 2 

5 Mod Jagatise jääk  a Mod b,  13 Mod 5 = 3 

6 + , - Liitmine, lahutamine  a + b, a - b 

 

Tehete prioriteetide rakendamise näiteid 

-3^2 * 5 + 18 / 2 * 3 = -9 * 5 + 9 * 3 = -45 + 27 = -18, 

(-3)^2 * 5 + 18 / (2 * 3) = 9 * 5 + 18 / 6= 48 

4^3^2 = 64^2 = 4096   64^1/3 = 641/3 = 64/3    64^(1/3) = = 4 

Matemaatikafunktsioonid 

Funktsioon Selgitus:  a – arvavaldis 

Sqr(a) Ruutjuur    Sqr(b^2 - 4*a*c) = (b^2 - 4*a*c)^(1/2) 

Log(a) 
Naturaallogaritm (ln a)  Log(a) / Log(10) = log10a 

NB! Kümnendlogaritmi VBAs ei ole! 

Exp(a) 
ea (e = 2,71828…)   

(Exp(-x) + Exp(2 * x)) / 2 = (e-x + e2x) / 2 

Abs(a) Absoluutväärtus  Abs((a-x)/(a+x)) 

Sin(a), Cos(a), Tan(a) 
Sin(x)+Cos(2*x)+Tan(x^2)-Cos(2*x)^2 = sin x+cos 2x+tan x2 -cos2 2x 

Argument radiaanides 

Atn(a) 

arctan radiaanides (-/2<x</2). 

Atn(a/Sqr(1-a^2))=arcsin a  

2*Atn(1) - Atn(a/Sqr(1-a^2)) = arccos a.  4*Atn(1) =  

Sgn(a) Arvu märk  Sgn(5)=1, Sgn(0)=0 (a=0), Sgn(-5)= -1 (a<0)   

Rnd() 
Juhuslik arv x: 0 <= x < 1.  

n=Int((b-a+1)*Rnd()+a) - täisarv vahemikus a <= n <= b 

Teisendusfunktsioonid 

Funktsioon Selgitus:  a – arv- või tekstavaldis 

Asc(a) Esimese märgi ASCII-kood. Asc(“Abi“)=65, Asc(1)=48 

CDbl(a) Teisendus topelttäpsusega realarvuks. | a | < 10307 

CInt(a) Teisendus täisarvuks. -32768 ≤ a ≤ 32767. Ümardatakse. 

CLng(a) Teisendus pikaks täisarvuks. -231 ≤ a ≤ 231 

CSng(a) Teisendus ühekordse täpsusega reaalarvuks. | a | < 1037 

CStr(a) Teisendus stringiks 

CVar(a) Teisendus Variant tüüpi 

Val(a) Teksti teisendus arvuks. Murdosa jäetakse ära. 

Int(a) Lähim täisarv, mis on väiksem kui a  Int(4.9)=4, Int(-4.9)=-5 

Fix(a) Arvu täisosa  Fix(4.9) = 4   Fix(-4.9) = -4 

Round(a [, mp]) Ümardab väärtuse mp numbrikohani murdosas 

Format(a, for) 

Arvu teisendamine etteantud vormingusse. for  –  vorming. Peamine variant 

“0.0{0}“. Võimaldab määrata murdosa pikkuse:  

Format(45.67375, "0.000") = > 45.674 

Stringavaldised ja funktsioonid 

Stringavaldiste operandide väärtuseks on stringid, neis võib kasutada stringitehet ja stringifunktsioone. 

Stringitehet & nimetatakse sidurdamiseks. See võimaldab ühendada stringe. Sidurdamiseks tuleks 

teisendada arvud tekstivormingusse (operaatori & kasutamisel teeb VB seda automaatselt). Sidurdustehte 

operaatorina võib kasutada ka märki  + , kui mõlemad operandid on stringid. 

Näiteid 

"Peeter" & " " & "Kask" annab Peeter Kask,    35.7 & " " & 2.5 annab 35.7 2.5 

Kui S=5378.75, x1=2.538, x2=-1.34, siis  

"Summa= " & S annab Summa= 5378.75,       

 "x1= " & x1 & "  x2= " & x2  =>  x1= 2.538  x2= -1.34  

Stringifunktsioonid  

s – stringavaldis, näidetes S = “Visual Basic“ 

Left(s, n) 
eraldab stringist n vasakpoolset märki.   

Left(S,6)=Visual 

Right(s, n) 
eraldab n parempoolset märki.   

Right(S, 5)=Basic 

Mid(s, m, n)          
eraldab n märki alates märgist m.   

Mid(S, 8, 3)=Bas 

Len(s) 
leiab stringi pikkuse   

Len(S) = 12 

InStr(n, s1, s2) 
leiab positsiooni, millest algab string s2 stringis s1. n näitab otsingu algust, 

vaikimisi 1.  InStr(S, “a”) = 5 

UCase(s) 
muudab väiketähed suurtähtedeks.  

UCase(S)=VISUAL BASIC 

LCase(s) 
muudab suurtähed väiketähtedeks.  

LCase(S)=visual basic 

Space(n) 
moodustab n tühikust koosneva stringi.   

Space(80) 

String(n, märk) moodustab stringi, mis sisalda n näidatud märki  

Ltrim(s), Rtrim(s), 

Trim(s) 

eemaldavad tühikut vastavalt stringi algusest, lõpust ning lõpust ja algusest   

Trim(Inputbox("Sisestage nimi")) 

Chr(kood) 
ASCII koodile vastav märk   

Chr(65) = A 

Asc(s) 
esimesele märgile vastav ASCII kood   

Asc("A") = 65 

Val(s) teisendab stringi arvuks. Murdosa jäetakse ära.  

Str(s) 
teisendab arvu stringiks   

Str(3.14159*r*r) 

Format(a, for) 
teisendab väärtuse etteantud formaadiga tekstiks  

Format(34.36732,"0.00") => 34.37 

Operatsioonides stringidega kasutatakse ka mõningaid erilauseid: 

Mid(s1, m[, n]) = s2 

Asendab stringis s1 n märki, alates positsioonist m, märkidega stringist s2. Kui n puudub, kasutatakse 

stringi s2 tervikuna. 

LSet s1 = s2 salvestab stringi s2 stringi s1 algusesse (vasakule). 

RSet s1 = s2 salvestab stringi s2 stringi s1 lõppu (paremale). 

Võrdlused ja loogikaavaldised 

Võrdlused on käsitletavad loogikaavaldiste erijuhtudena, nende kuju on: 

avaldis1 operaator avaldis2 

Võrdlusoperaatorid on:  = , <> , < , <= , > , >= 

avaldis1 ja avaldis2 on arv- või stringavaldised. Ühes võrdluses esinevad avaldised peaks üldjuhul kuuluma 

samasse liiki. Võrdluses võib olla ainult üks operaator. Võrdluse tulemiks on alati tõeväärtus True (tõene) 

või False (väär).  

Võrdluste näiteid 

x <= 0,  b*b - 4*a*c < 0,  x*x + y*y > r*r, UCase(vastus) = "EI" 

NB! Stringide võrdlemisel eristatakse suur- ja väiketähti! 

Loogikaavaldise üldkuju on järgmine: 

avaldis LTS avaldis [LTS avaldis]… 

Siin on avaldis võrdlus- või loogikaavaldis ja LTS loogikatehte sümbol. Peamised loogikatehted on Or, And ja 

Not. Nende tähendused on 

Or  või. Tehte a Or b  väärtus on tõene (True), kui vähemalt ühe operandi väärtus on tõene, vastupidisel 

juhul on tulem väär (False). 

And  ja. Tehte a And b tulem on tõene (True) ainult siis, kui mõlema operandi väärtused on tõesed, 

vastupidisel juhul on tehte tulem väär (False). 

Not  mitte. Tehte Not a tulem on tõene (True) siis, kui a väärtus on väär (False) ja väär (False) vastupidisel 

juhul. 

Loogikaavaldiste näiteid 

x >= 2 And x <= 13,  x < 2 Or x >13 

a+b > c And a+c > b And b+c > a 

Mõned loogikafunktsioonid 

IsDate(a) Tõene, kui argument (a) on kuupäev 

IsEmpty(a) Tõene, kui argumendi väärtus on tühi 

IsMissing(param) Tõene, kui parameeter puudub 

IsNumeric(a) Tõene, kui avaldise väärtus on arv 

IsObject(a) Tõene, kui argumendiks on objekt 

Ajaavaldised ja ajafunktsioonid 

Ajaväärtus sisaldab üldjuhul kahte osa: kuupäev ja kellaaeg. Ajaväärtused salvestatakse reaalarvudena: 

täisosa – päeva järjenumber alates baasajast (01:01:1900), murdosa – kellaaeg alates keskööst päeva 

osades. 

Ajaväärtuste kui reaalarvudega saab täita põhimõtteliselt suvalisi operatsioone, kuid praktilist tähendust 

omavad ainult liitmine, lahutamine ja võrdlemine: 

aeg2 – aeg1 NOW – #01/01/2000# 

kuupäev2 – kuupäev1 Date – sünniaeg 

kellaaeg2 – kellaaeg1 finaeg – startaeg 

kuupäev + arv Date+100 

kuupäev – arv Date-100 

jooksev aeg – algaeg Timer() – algaeg 

NB! Kahe ajaväärtuse vahe saadakse päevades, see on arv, mille täisosa on päevade arv kahe ajaväärtuse 

vahel ja murdosa kellaaeg päeva osades. 

Mõned ajafunktsioonid 

Now() Jooksev aeg arvutis kujul pp.kk.aaaa hh:mm:ss 

Date() Jooksev kuupäev kujul  pp.kk.aaaa 

Time() Jooksev kellaaeg kujul  hh:mm:ss  

Timer() Keskööst möödunud aeg sekundites täpsusega 0,01 sek 

Year(aeg) Aasta number ajaväärtuses 

Month(aeg) Kuu number ajaväärtuses 

Day(aeg) Päeva number ajaväärtuses 

Hour(aeg) Tunnid ajaväärtuses 

Minute(aeg) Minutid ajaväärtuses 

Second(aeg) Sekundid ajaväärtuses 

DateSerial(a, k, p) Teeb kuupäeva aasta (a), kuu (k) ja päeva (p) alusel 

TimeSerial(h, m, s) Teeb kellaja tundide (h), minutite (m) ja sekundite (s) alusel 

Omistamine ja omistamislause 

Omistamise olemus ja omistamislause põhivariandid 

Omistamine on üks fundamentaalsemaid tegevusi, mida arvuti saab programmi toimel täita. See seisneb 

mingi väärtuse või aadressi salvestamises arvuti sisemälu etteantud väljas või pesas – muutujas. Välja 

(pesa) eelmine väärtus (kui oli) kaob.  

Üheks peamiseks vahendiks väärtuse salvestamiseks (omistamiseks) on kõikides programmeerimiskeeltes 

omistamislause. Tüüpiliselt eelneb väärtuse salvestamisele selle leidmine (tuletamine) etteantud avaldise 

abil. Kusjuures avaldise väärtuse leidmisele kaasneb enamasti muutujate ja/või omaduste varem 

salvestatud väärtuste lugemine. Omistamine võib kaasneda ka mõnede muude lausete ja meetodite 

täitmisele, näiteks korduslause For...Next. 

VBAs võib eristada omistamislause kasutamise kuut põhivarianti: 

 väärtuse omistamine lihtmuutujale, 

 väärtuse omistamine objekti omadusele, 

 viida omistamine objektimuutujale, 

 väärtuse omistamine struktuurmuutuja (massiiv, kirje) elemendile, 

 viida omistamine struktuurmuutuja elemendile, 

 ühe struktuurmuutuja omistamine teisele struktuurmuutujale. 

Järgnevas peatume esimesel kahel.  

Väärtuse omistamine lihtmuutujale ja objekti omadusele 

Esimese kahe omistamise ja vastavate omistamislausete olemus on praktiliselt sama. Muutujale eraldatav 

mäluväli (pesa) kujutab endast objekti, mille omadust väärtus omistuslause täitmisel muudetakse. Objekti 

igale omadusele eraldatakse omaduste vektoris mäluväli, kuhu omistamisel salvestatakse vastav väärtus. 

Esimese kahe variandi jaoks on omistamislausete kujud järgmised: 

muutuja = avaldis 

objekt.omadus = avaldis 

Siin muutuja esitatakse nime abil, objekt määratakse objektiviida või objektimuutuja abil ning omadus 

näidatakse vastava nime abil. Mõlemal juhul on avaldise olemus ja esitamise reeglid samad. Tuletame 

meelde, et avaldiste operandideks võivad olla konstandid, muutujad, objektide omadused, massiivi 

elemendid, funktsiooniviidad (funktsioonid) VBA sisefunktsioonidele või kasutaja funktsioonidele. Avaldis 

võib koosneda ka ainult ühest operandist. 

Omistamislausete näiteid: 

k = 0:    x = 2.5:    pi = 3.14159:    nimi = "A. Kask":    auto.Left = 25   

Erijuht: muutujale või omadusele omistatakse konstandi väärtus – konstandi väärtus salvestatakse 

programmiplokist andmeplokki või omaduse vektori vastavasse mäluvälja.  

NB! Öeldakse, et „k-le omistatakse null“ või „x-le omistatakse 2.5“, aga mitte „k võrdub nulliga“ või 

„x võrdub 2.5-ga“. 

x = y : ring.Top = Shapes(“rist“).Top : ufo.Left = x :  a = Range(“a“) 

Erijuht: muutujale või omadusele omistatakse teise muutuja või omaduse väärtus – paremas pooles oleva 

muutuja või omaduse väärtus kopeeritakse vasakus pooles oleva muutuja või omaduse mäluvälja.  

NB! Ka siin on õige öelda, et „y-i väärtus omistatakse x-le“ (lühemalt: „y omistatakse x-le“), aga mitte 

„x võrdub y-ga“. 

x = a + i * h : y = 3 * Sin(x) - Sqr(x^4 + 5) 

Üldjuht: leitakse paremas pooles oleva avaldise väärtus ja tulemus omistatakse vasakus pooles olevale 

muutujale, st salvestatakse antud muutuja mäluväljas (pesas). 

k = k + 1: S = S + y: n = n – k : F = F * k: auto.Left = auto.Left + h 

Erijuht: sama muutuja või omadus esineb omistuslause vasakus ja paremas pooles. Tegemist on muutuja 

või omaduse uue väärtuse leidmise ja asendamisega eelmise (jooksva) väärtuse alusel. Näiteks lause 

k = k + 1 täitmisel loetakse k jooksev väärtus, liidetakse sellele 1 ja saadud tulemus võetakse k uueks 

väärtuseks, so k väärtust suurendatakse ühe võrra. 

Tahaks rõhutada, et omistamislauset ei tohi mitte mingil juhul samastada võrrandi, võrdusega või valemiga, 

mida alguses sageli kiputakse tegema välise sarnasuse tõttu (eriti omistamisel lihtmuutujale).  

Võrduses või võrrandis on vasak ja parem pool samaväärsed.  

Näiteks võrdused x = y ja y = x on samaväärsed aga omistamislaused x = y ja y = x on lausa vastupidise 

toimega. Olgu näiteks pesade x ja y sisu vastavalt 5 ja 1. Lause x = y täitmisel kopeeritakse arv 1 pesast y 

pesasse x ja peale seda on mõlemas pesas väärtus 1, väärtus 5 pesas x kaob. Kui aga tegemist on 

omistamislausega y = x, kopeeritakse väärtus (5) pesast x pesasse y ja mõlemas pesas on nüüd väärtus 5. 

Veel kaks sarnast aga vastupidise toimega lauset 

Juku.Left = pall.Left ja pall.Left = Juku.Left 

Esimese juhul “viiakse“ objekt Juku objekti pall juurde: Juku vasak serv võetakse võrdseks palli vasaku 

servaga, teisel juhul, vastupidi – “tuuakse“ pall Juku juurde.  

Näiteks omistamislaused 2 = x  ja  x + 5 = y on mõttetud ja lubamatud. Ei saa omistada väärtust konstandile 

või avaldisele ehk salvestada (!) konstandis või avaldises midagi. Omistuslause vasakus pooles võib olla 

ainult muutuja nimi või objekti omadus! 

Võrrandid k = k + 1 või n = n – k ei oma mingit mõtet matemaatikas: ei saa mingi suurus olla võrdne 

iseendaga pluss või miinus midagi, väljaarvatud triviaalne juht, kui see midagi võrdub nulliga. 

Andmete lugemine töölehelt ja kirjutamine töölehele 

Reeglina on otstarbekas arvutustes kasutatavad lähteandmed lugeda eelnevalt töölehelt VB muutujatesse. 

Muutujate kasutamine võimaldab loobuda vahetulemuste ja abiandmete salvestamisest töölehele. Ka 

tulemused on tüüpiliselt otstarbekas salvestada muutujatesse ja sealt kirjutada töölehele. Muutujate 

kasutamine suurendab programmide töökiirust kümneid ja isegi sadu kordi. See on väga oluline, kui on 

tegemist suuremate ülesannete ja andmehulkadega. 

Andmevahetuseks Exceli töölehtede ja VBA vahel saab kasutada erinevaid vahendeid. Siinpakutavad on 

üksikväärtuste korral ühed lihtsamad ja enim kasutatavad. Tegemist on erikujuliste omistamislausetega. 

Väärtuste lugemiseks töölehelt võib kasutada järgmise kujuga omistamislauset: 

muutuja = Range("nimi").Value või muutuja = Range("nimi") või muutuja = [nimi] 

Formaalselt on tegemist töölehe lahtri omaduse Value lugemisega ja selle salvestamisega VB muutujasse. 

Kuna omadus Value on lahtri jaoks vaikimisi võetav omadus, jäetakse see tavaliselt ära, näiteks:  

a = Range("a"): b = Range("b"): L = Range("pikkus") 

Väärtuste kirjutamiseks töölehele võib kasutada järgmise kujuga lauset: 

Range("nimi").Value = avaldis või  Range ("nimi") = avaldis või [nimi] = avaldis 

Formaalselt on tegemist töölehe lahtri omaduse Value muutmisega. Üldjuhul leitakse avaldise väärtus ja 

saadud tulemus kirjutatakse töölehe lahtrisse. Ka siin võib omaduse Value ära jätta. 

NB! Kui muutuja nimi langeb kokku lahtri nimega ei saa kasutada viitamiseks lahtritele konstruktsiooni 

[nimi], vaid peab kasutama põhivarianti Range ("nimi"). 

Dialoogibokside kasutamine 

Dialoogibokse võib kasutada teadete väljastamiseks ning üksikväärtuste lugemiseks ja väljastamiseks. Siin 

vaadeldakse nende kasutamise lihtsamaid võimalusi. 

VBA siseprotseduur MsgBox võimaldab väljastada teateid ja üksikuid väärtusi. Tema lihtsaim variant on: 

MsgBox stringavaldis 

Lause täitmisel peatub programmi täitmine ning kuvatakse teateboks stringavaldise väärtusega ja nupuga 

OK. Kui kasutaja klõpsab nuppu OK või vajutab klahvile Enter, eemaldatakse boks ja programmi töö jätkub. 

Üksikute väärtuste lugemiseks võib kasutada VBA sisefunktsiooni InputBox, mille põhivariant on järgmine: 

muutuja = InputBox(teade [,päis, pakkumine]) 

Funktsioonil on üks kohustuslik argument - teade, mis võib üldjuhul olla stringavaldis. Lause täitmisel peata-

takse programmi töö ning ilmub sisendboks, milles on kuvatud argumendi teade väärtus. Dialoogiboksis on 

tekstiväli, kuhu kasutaja saab tippida sisestatava väärtuse, ning kaks nuppu OK ja Cancel. Kui kasutaja 

klõpsab nuppu OK või vajutab klahvile Enter, omistatakse tekstiväljas olev väärtus antud muutujale. Kui 

kasutaja klõpsab nuppu Cancel või vajutab klahvile Esc, omistatakse muutujale tühi string. 

Argumendi päis väärtus (kui ta esineb) kuvatakse sisendboksi päises. Argument pakkumine on mõeldud 

vaikimisi võetava väärtuse soovitamiseks. See kuvatakse sisendboksi tekstiväljas ja kui kasutaja ei asenda 

seda, vaid klõpsab kohe nuppu OK, võetakse see muutuja väärtuseks. Näiteks kuvab lause 

n = InputBox ("Katsete arv", "Korrutamine", 13) 

sisendboksi teatega „Katsete arv“, boksi päises on tekst „Korrutamine“ ning tekstiväljas arv 13. Kui kasutaja 

klõpsab kohe nuppu OK, võetakse muutuja n väärtuseks 13. 

Sisendboksist loetavat väärtust käsitletakse alati tekstina. Arvude puhul võib see põhjustada probleeme, 

kui arve kasutatakse liitmistehetes. Kuna stringide sidurdamise tehtesümbolina on lubatud kasutada ka 

märki +, võib arvude liitmise asemel toimuda nende sidurdamine.  

Näiteks makro Test1 täitmisel, kui a väärtuseks sisestada 20 ja b väärtuseks 10, kuvatakse järgmine vastus: 

Keskmine=1005. Lause c = (a + b) / 2 täitmisel sidurdatakse a väärtus b väärtusega ja saadakse 2010, mis 

jagamisel kahega annab tulemuseks 1005.  

Et vältida taolisi asju, peaks sisestatud väärtuse teisendama arvuvormingusse: 

muutuja = Val(InputBox(teade [,päis, pakkumine])) 

Makros Test2, kus arvude sisestamisel kasutatakse funktsiooni Val, toimub arvude liitmine ning samade 

väärtuste korral (20 ja 10) väljastatakse vastus: Keskmine = 15. Kui on tegemist täisarvudega, võib Val 

asemel kasutada funktsiooni Int, reaalarvude korral aga funktsiooni CDbl.  

Makrod demonstreerivad ka võimalikku reaktsiooni juhule, kui kasutaja klõpsas nuppu Cancel.  

Sub Test1() 

a = InputBox("a", , 20) 

If a = "" Then End 

b = InputBox("b", , 10) 

If b = "" Then End 

c = (a + b) / 2 

MsgBox "Keskmine=" & c 
End Sub 

Sub Test2() 

 a = Val(InputBox("a", , 20)) 

 If a = 0 Then End 

 b = Val(InputBox("b", , 10)) 

 If b = 0 Then End 

 c = (a + b) / 2 

 MsgBox "Keskmine=" & c 
End Sub 

Juhtimine 

Valikud ja valikulaused 

Valikulaused võimaldavad määrata tegevuste (lausete) valikulist täitmist sõltuvalt etteantud tingimustest ja 

kriteeriumitest. Need kujutavad endast liit- ehk struktuurlauseid, sisaldades teisi liht- ja/või liitlauseid.  

VBAs on kaks If-lauset: mitmerealine ja üherealine. 

Lausete struktuur ja täitmise põhimõtted 

Mitmerealise If-lause üldkuju ja täitmise põhimõte on järgmised: 

 

If tingimus Then 

     [ if_laused ] 

[ ElseIf tingimus2 Then 

      elseif_laused 

[ Else  

      else_laused ] 

  End If 

 

If-, ElseIf-, Else- ja End If-osalaused peavad olema eraldi ridadel. Tingimused esitatakse võrdluste või 

loogikaavaldiste abil. Laused võivad olla suvalised liht- ja liitlaused, sh ka If-laused. 

Lause täitmisel kontrollitakse kõigepealt tingimust If-osalauses, kui see on tõene, täidetakse if_laused, kõik 

ülejäänu jääb vahele. Vastupidisel juhul kontrollitakse järjest tingimusi ElseIf-osalausetes (kui neid on 

olemas) ning kui leitakse esimene tõene, täidetakse järgnevad laused, kõik ülejäänu jääb vahele. Kui ükski 

tingimus ei ole tõene, täidetakse else_laused (kui need on olemas). 

Ülaltoodud If-lause üldist varianti nimetatakse sageli ka mitmeseks valikuks: mitmest võimalikust tegevuste 

rühmast valitakse tingimuste alusel välja üks.  

Lause üldkujust tulenevad ka kaks sageli kasutatavat varianti, mis võimaldavad määratleda valiku kahest ja 

valiku ühest.  

Kahendvalik  Valik ühest 

If tingimus Then 

     laused_1 

Else  

     laused-2  

End If 

 
If tingimus Then 

     laused 

End If 

 

Sama tüüpi valikuid võimaldab kirjeldada ka üherealine If-lause, mille üldkuju on: 

If tingimus Then laused_1  [ Else laused_2 ] 

Üherealist lauset tasub kasutada siis, kui sisemised laused on lühikesed lihtlaused. Näiteks: 

If a > b Then m = a Else m = b 

If x > 0 Then k = k + 1: Sp = Sp + x 

If y > max Then max = y 

If n > nmax Then Exit Do 

Loomulikult saab toodud laused esitatada mitmerealistena, näiteks esimene lause näeks välja nii: 

If a > b Then 

   m = a 

Else 

   m = b 

End If 

Nagu näha, tuleb siin ridu juurde, sest If-, Else- ja End If-osalaused peavad olema eraldi ridadel. Üherealisel 

lausel peavad aga kõik elemendid olema ühel real ja End If-osalauset ei ole ja ei tohi olla. 

Näited 

Näide: Kolme arvu mediaan 

Leida (valida välja) kolmest arvust suuruse poolest keskmine ehk mediaan. Allpool on toodud ülesande 

lahendamise tegevusskeem (algoritm) ja VBA funktsioon Med. Funktsioonil on kolm parameetrit: a, b ja c. 

Kasutusel on kaks mälupesa (muutujat) min ja max. Kõigepealt omistatakse muutujale min minimaalne 

väärtus esimesest kahest arvust (a ja b) ja muutujale max maksimaalne väärtus nendest. Edaspidise jaoks ei 

ole tähtis kumb arvudest (a või b) on suurem. 

Edasi võrreldakse min ja max väärtusi kolmanda arvuga c. Kui c on suurem kui max, on tegemist olukorraga, 

kus min ≤ max ≤ c ja mediaaniks on max. Kui aga c ≤ max, on kaks võimalust. Kui c > min on min ≤ c ≤ max ja 

mediaaniks on c, kui ei ole, siis c ≤ min ≤ max ja mediaaniks on min. 

 

Funktsioonis on kasutusel kaks mitmerealist If-lauset. Esimene esindab nö klassikalist kahendvalikut, teises 

on kasutusel ka ElseIf-osa.  

Function Med1(a, b, c) 

    Dim max, min 

    If a < b Then min = a: max = b Else min = b: max = a 

    If c > max Then Med1 = max Else If c > min Then Med1 = c Else Med1 = min 

End Function 

Function Med(a, b, c) 

    Dim max, min 

    If a < b Then 

        min = a: max = b 

    Else 

        min = b: max = a 

    End If 

    If c > max Then 

        Med = max 

    ElseIf c > min Then 

        Med = c 

    Else 

        Med = min 

    End If 

End Function 

Võrdluseks on toodud sama ülesande jaoks funktsioon Med1, kus kasutatakse kahte üherealist If-lauset. 

Võib tähele panna, et kuigi funktsioon Med1 on kompaktsem (eeskätt ridade arvude poolest), on selle 

loetavus ja ülevaatlikus halvem kui eelmisel funktsioonil, seda eriti viimase lause puhul. 

Function Med2(a, b, c) 

    Dim max, min 

    If a < b Then min = a: max = b _ 

                 Else min = b: max = a 

    If c > max Then Med2 = max _ 

                     Else If c > min Then Med2 = c _ 

                                              Else Med2 = min 

End Function 

Loetavust ja ülevaatlikkust saab pika rea puhul parandada, 

kasutades käsu jätkumise tunnust: tühik ja allkriips ( _ ). 

Näiteks on toodud sama funktsiooni jaoks veel üks variant. 

Funktsioonis Med2 on kasutusel samuti kaks üherealist 

If-lauset, mis on liigendatud mitmele reale.  

Nagu juba varem öeldud, peaks eelistama mitmerealist 

If-lauset. 

Näide: Ruutvõrrand 

Programm leiab ruutvõrrandi ax2 + bx + c = 0 reaalarvulised lahendid x1 ja x2. Arvestatakse võimalusega, et 

lahendid võivad puududa, kui diskriminant D = b2 – 4ac on väiksem kui 0. Kontrollitakse ka, et a ei oleks null. 

Ruutvõrrandi lahendamine on realiseeritud parameetritega protseduuriga Ruut_Vrd. Sisendparameetriteks 

on võrrandi kordajad a, b ja c, väljundparameetriteks lahendid x1 ja x2 (kui need on) ja abimuutuja tun. 

Viimase väärtuseks on 1 (lahendid on olemas) või 0 (lahendid puuduvad).  

Kõigepealt leitakse protseduuris diskriminandi D väärtus. If-lauses kontrollitakse, kas D on nullist väiksem, 

kui jah, võetakse muutuja tun väärtuseks 0 ja Else-osa laused jäävad täitmata. Vastupidisel juhul võetakse 

muutuja tun väärtuseks 1 ja leitakse x1 ja x2 väärtused. 

Sub Ruut_Vrd(a, b, c, x1, x2, tun) 

    Dim D 

    D = b ^ 2 - 4 * a * c 

    If D < 0 Then 

        tun = 0 

    Else 

        tun = 1 

        x1 = (-b - Sqr(D)) / (2 * a) 

        x2 = (-b + Sqr(D)) / (2 * a) 

    End If 

End Sub 

Sub Ruut_Pea() 

  Dim a, b, c, x1, x2, tunnus 

  a = Range("a") 

  If a = 0 Then MsgBox "a ei tohi olla 0!": End 

  b = Range("b"): c = Range("c_") 

  Ruut_Vrd a, b, c, x1, x2, tunnus 

  If tunnus = 0 Then 

      x1 = "": x2 = "" 

      MsgBox "Lahendid puuduvad!" 

   End If 

   Range("x_1") = x1: Range("x_2") = x2 

End Sub 

Peaprotseduuris arvestatakse, et võrrandi kordajad on töölehe lahtrites nimedega a, b ja c_. Protseduur 

loeb väärtused töölehelt ja salvestab vastavates muutujates. Kõigipealt loetakse a väärtus ning kui see 

võrdub nulliga, kuvatakse teade ja programmi töö lõpetatakse. Siin on kasutusel üherealine If-lause, milles 

on kaks lihtlauset: MsgBox ja End. Need mõlemad täidetakse, kui tingimus on tõene, või jäetakse vahele, 

kui tingimus ei ole tõene. Peale alamprotseduuri täitmist kontrollitakse If-lauses muutuja tunnus väärtust. 

See saab väärtuse parameetrilt tun ja saab olla 0 või 1. Kui tunnus võrdub nulliga, omistatakse muutujatele 

x1 ja x2 tühi väärtus ning kuvatakse vastav teade. Kui see ei ole 0, jäävad kehtima tagastatud väärtused. 

Viimased laused kirjutavad töölehe lahtritesse x1 ja x2 väärtused, sõltumata sellest, kuidas need tekkisid. 

Kordused 

VBs on korduste kirjeldamiseks mitu lauset: 

 Do...Loop-lause 

 For-lause 

 For Each-lause 

 Lausetel on mitu varianti, eriti kehtib see Do…Loop lause kohta 

Do … Loop-lause. Lõputu kordus 

  Do  

      laused  

  Loop 

 

kordus 

  tegevused  

lõpp kordus 

 

 

Do ja Loop vahel olevate lausetega määratud tegevusi täidetakse põhimõtteliselt lõputult. Korduse saab 

lõpetada täites protseduuri, milles on End-lause või vajutades klahvidele Ctrl+Break. Katkestamiseks võib 

korduse sees olla käsk Exit Do. 

Näide: Foor 1 

Imiteeritakse foori ühe sektsiooni tööd. Töölehel on graafikaobjekt (näiteks ring) nimega tuli ning 

käsunupud Tööle ja Stopp. Protseduuri Tööle moodustab lõputut kordust määrav Do…Loop-lause. Pidevalt 

toimub pöördumine protseduuri Muuda poole, mis muudab tule värvi. Parameetriteks ja argumentideks on 

värvi number ja põlemise kestus. Foori saab välja lülitada, käivitades protseduuri Aitab. 

  Sub Tööle () 

    Do  

       Muuda  2, 5  ' punane, 5 sek 

       Muuda  5, 2  ' kollane, 2 sek 

       Muuda  3, 4  ' roheline, 4 sek 

       Muuda  5, 2  ' kollane, 2 sek 

    Loop 

  End Sub  

 

  Sub Muuda (värv, p) 

      Shapes("tuli").Fill.ForeColor.SchemeColor = värv 

      paus p 

  End Sub  

  Sub Aitab () 

      Muuda 0, 1 ' must  

     End 

  End Sub  

Do … Loop-lause. Lõputu kordus katkestusega 

Do 

  laused_1  

  If tingimus Then Exit Do  

  laused_2  

Loop 

 

 

kordus  

    tegevused 1 

    kui tingimus siis välju 

    tegevused 2 

lõpp kordus 

Korduse moodustab Do…Loop-lause. Selle sees peab olema vähemalt üks valikulause, mis sisaldab 

katkestamiskäsku Exit Do (võib olla ka Exit Sub või Exit Function). Igal kordamisel kontrollitakse tingimust. 

Kui see on tõene, katkestab Exit Do korduse ja täitmine jätkub Do…Loop-lausele järgnevast lausest. Laused 

Exit Sub või Exit Function katkestavad vastava protseduuri töö.  

Katkestamise lauseid võib olla rohkem kui üks ning tingimus võib olla antud üldisemal kujul. 

NB! Scratchis lõpetab käsk [peata skript] skripti töö, aga ainult korduse katkestamiseks käsku ei ole.  

tegevused 

 

Sub paus(pp) 

  Dim pl 

  pl = Timer() + pp 

  Do 

     DoEvents 

  Loop While Timer() < pl 

End Sub 



57 

Näide: Auto 1 

 

Töölehel on kujundid nimedega auto ja fin. 

Programmi toimel „sõidab“ auto töölehe vasakust 

servast finišijooneni (fin). Kasutaja saab töölehe 

lahtris anda sammu, millest sõltub liikumise kiirus. 

Programm kirjutab lahtrisse aeg sõitmise aja. 

Sub Auto_1() 

  Dim auto As Shape, algaeg, h 

  Set auto = Shapes("auto") 

  h = Range("samm") 

  auto.Left = 0: algaeg = Timer() 

  Do 

      Range("aeg") = Timer() - algaeg 

      auto.IncrementLeft h / 2 + Rnd() * h 

      If auto.Left > Shapes("fin").Left Then Exit Do 

      paus 0.01 

  Loop 

  MsgBox "Olen kohal!" 

End Sub 

Lahtrist samm loetakse väärtus ja omistatakse 

muutujale h. Auto vasaku serva väärtuseks võetakse 

null ning muutuja algaeg väärtuseks taimeri jooksev 

väärtus. 

Põhiosa tegevustest toimub Do…Loop korduses. Igal 

kordamisel kirjutatakse lahtrisse aeg jooksev aeg. 

Auto vasaku serva koordinaati muudetakse juhuarvu 

abil, kasutades muutuja h väärtust. If-lauses kontrolli-

takse, kas auto vasak serv on suurem finišijoone vasa-

kust servast. Kui jah, katkestatakse kordamine ja 

kuvatakse teade „Olen kohal!“ 

Do … Loop-lause. Eelkontrolliga ja järelkontrolliga kordused 

 

Do Until tingimus 

    laused 

Loop 

 
 

Do While tingimus 

    laused 

Loop 

 
 

Do  

    laused 

Loop Until tingimus 

 
 

Do  

    laused 

Loop While tingimus 

 
Tegemist on nelja väga sarnase korduste määratlemise lausega. Kui Do-osalauses kasutatakse täiendit 

Until, toimub kordamine seni, kuni tingimus saab tõeseks. Täiendi While korral toimub kordamine, kui 

tingimus on tõene. Järelkontrolliga lausetes täidetakse grupis olevaid lauseid vähemalt üks kord. 

Eelkontrolliga korduses võivad need jääda ka täitmata, kui tingimus on Until-tüüpi korduses kohe tõene või 

väär While-tüüpi korduse korral. Milline variant valida, sõltub sageli programmi koostaja soovist. Lausete 

seas võib olla üks või mitu katkestamislauset: Exit Do, Exit Sub või Exit Function.  

Toodud näidetes on protseduuris Auto_2 kasutusel eelkontrolliga Until-kordus, protseduuris Auto_3 on 

järelkontrolliga Do While-kordus. Esimesel juhul, kui auto on juba sihtkohas, ei liigu see programmi 

käivitamisel paigast: tingimus on kohe tõene. Teisel juhul samas olukorras teeb auto ühe sammu ja siis 

selgub, et tingimus on väär ja kordus lõpetatakse. 

tõene

laused

väär väär

laused

tõene

Sub Auto_2() 

  Dim auto As Shape, algaeg, h 

  Set auto = Shapes("auto") 

  h = Range("samm") 

  auto.Left = 0: algaeg = Timer() 

  Do Until auto.Left  >= Shapes("fin").Left 

      Range("aeg") = Timer() - algaeg 

      auto.IncrementLeft h / 2 + Rnd() * h 

      paus 0.01 

  Loop 

End Sub 

 Sub Auto_3() 

  Dim auto As Shape, algaeg, h 

  Set auto = Shapes("auto") 

  h = Range("samm") 

  auto.Left = 0: algaeg = Timer() 

  Do  

      Range("aeg") = Timer() - algaeg 

      auto.IncrementLeft h / 2 + Rnd() * h 

      paus 0.01 

  Loop While auto.Left  <  Shapes("fin").Left 

End Sub 

Näide: Palli liikumine  

Protseduurid, mis juhivad palli lendu üles aeglustusega (kuni kiirus jõuab nulli) ning kukkumist kiirendusega 

kuni pall jõuab maale.  

Sub Yles() 

 ' Liikumine üles aeglustusega 

  Dim h, d, pall As Shape 

  Set pall = Shapes("pall") 

  h = 30: d = 1 

  Do Until h <= 0 Or pall.Top <= 0 

     pall.IncrementTop -h 

     h = h - d 

     paus 0.1 

  Loop 

End Sub 

 Sub Alla() 

 ' Liikumine alla kiirendusega 

  Dim h, d, pall As Shape 

  Set pall = Shapes("pall") 

   h = 0: d = 1 

   Do While pall.Top < Shapes("maa").Top 

       pall.IncrementTop h 

       h = h + d 

       paus 0.1 

   Loop 

End Sub 

For ... Next-lause. Juhtmuutujaga kordus  

 

For v = av1 To av2 [ Step av3 ] 

     laused_1 

    [Exit For] 

    laused_2  

Next v  

 

 

 

kordus v = a1..a2 [,a3] 

 

      tegevused  

 

  lõpp kordus 

 

v on nn juhtmuutuja. av1, av2 ja av3 võivad üldjuhul olla avaldised. av1 on juhtmuutuja algväärtus, av2 

lõppväärtus, av3 muutmise samm. Kui sammu ei ole antud, võetakse selle väärtuseks 1.  

Näide: Auto ringliiklus 

Programm imiteerib auto ringliiklust. Kasutusel on kolm graafikaobjekti auto, Juku ja Kraps, kolm lahtrit: 

ringe, ring ja aeg ning neli muutujat: auto, ringe, ring ja algaeg.  

tegevused

* v =av1, av2, av3

Sub Auto_4() 

  Dim auto As Shape, ringe, ring, algaeg 

  Set auto = Shapes("auto") 

  auto.Left = 0 

  ringe = Range("ringe") ' ringide arv 

  algaeg = Timer()                 

  For ring = 1 To ringe  ' For-lause algus 

     Range("ring") = ring  ' ringi number 

     Do Until auto.Left > 1000     

         Range("aeg“) = Timer() - algaeg   

         auto.IncrementLeft  20 + Rnd() * 10 

         paus 0.02  

     Loop           

     auto.Left = 0                        

  Next ring                    ' For –lause lõpp 

  Hyppa Shapes("Juku"), 3 , 40    

  Hyppa Shapes("Kraps"), 5 , 30  

End Sub 

Auto viiakse töölehe vasakusse serva. Töölehe lahtrist 

ringe loetakse väärtus ja omistatakse muutujale ringe. 

Seda kasutatakse For-korduses juhtmuutuja ring lõpp-

väärtusena. 

Igal kordamisel liigub auto töölehe vasakust servast 1000 

punkti kaugusele. Ringi number (juhtmuutuja väärtus), 

kirjutatakse lahtrisse ring. Liikumise juhtimiseks kasu-

tatakse Do Until-kordust. 

Jooksev aeg kirjutatakse 

lahtrisse aeg. 

Peale sõidu lõppu teevad 

Juku ja Kraps mõned 

hüpped protseduuri 

Hyppa toimel, kus 

kasutatakse samuti 

For-kordust. 

Näide: Naturaalarvude ruutude summa 

Funktsioon Rea_Sum leiab naturaalarvude ruutude summa alates 

algväärtusest n1 kuni n2. Algus ja lõpp antakse parameetritele vastavate 

argumentidega pöördumisel. 

Põhiosa funktsioonist moodustab juhtmuutujaga kordus. Kõigepealt 

võetakse muutuja S algväärtuseks 0. For-lauses muudetakse juhtmuutuja 

k väärtust alates n1-st n2-ni ning liidetakse järgmise väärtuse ruut 

summale. Vastavalt VB reeglitele, omistatakse tagastatav väärtus 

funktsiooni nimele. 

Näide „Ühemuutuja funktsiooni väärtuste aritmeetiline keskmine“ 

Allpool on toodud kaks funktsiooni, mis võimaldavad leida ühemuutuja funktsiooni y = F(x) väärtuse lõigul 

[a, b]. Lõik jagatakse n võrdseks osaks pikkusega h = (b – a)/n. Keskmise saamiseks leitakse funktsiooni 

väärtuste summa ja jagatakse punktide arvuga n+1. Funktsiooni väärtuste yi leidmisel muudetakse x 

väärtusi sammuga h alates a väärtusest kuni b väärtuseni. Eeldatakse, et funktsiooni väärtuse leidmiseks 

antud x väärtuse jaoks on olemas VB funktsioon F(x). 

Function Fkesk_1(a, b, h) 

   Dim  S, x, n 

   S = 0 : n = 0 

   For x = a To b + h / 2 Step h 

      S = S + F(x) 

      n = n + 1 

   Next x 

   Fkesk_1 = S / n 

End Function 

Function Fkesk_2(a, b, n) 

   Dim i, S, h, x 

   h = (b - a) / n: S = 0 

   For i = 0 To n 

      x = a + i * h 

      S = S + F(x) 

   Next i 

   Fkesk_2 = S / (n + 1) 

End Function 

Funktsioonis   Fkesk_1 on parameetriteks 

otspunktid a ja b ning  samm h. 

Funktsioonis Fkesk_2 on sammu asemel 

jaotiste arv n. 

Funktsioonis Fkesk_1 on For-lause juht-

muutujaks reaalarvuline suurus x, mille 

algväärtus on a, lõppväärtus b ja muutu-

mise samm on h.  

Kuigi teoreetiliselt on x lõppväärtus b, siis praktiliselt on selleks võetud b + h/2, st pool sammu rohkem. Asi 

on selles, et avaldisega (b–a)/n leitav sammu h tegelik väärtus saab tulla ka veidi erinev teoreetilisest, 

milleks võib olla lõpmatu murd. Selle tõttu võib x väärtuse suurendamisel korduses tekkida olukord, kus 

viimane väärtus, mis peaks teoreetiliselt olema võrdne b-ga, on sellest natuke suurem ja sellisel juhul 

viimast juhtmuutuja väärtust ei kasutata. Näiteks praegu jääks viimane punkt keskmise arvutamisest välja. 

Poole sammu lisamine lõppväärtusele tagab, et viimane väärtus leiab igal juhul kasutamist. 

Sub Hyppa(kuju, n, h) 

Dim i 

For i = 1 To n 

   kuju.IncrementTop -h 

   paus 0.1 

   kuju.IncrementTop h 

   paus 0.1 

Next i 

End Sub 

Function Rea_Sum(n1, n2) 

   Dim k, S 

   S = 0 

   For k = n1 To n2 

      S = S + k ^ 2 

   Next  k 

   Rea_Sum  = S 

End Function 

Näide: Ühemuutuja funktsiooni maksimaalne väärtus 

Funktsioon Fmax_1 leiab etteantud ühemuutuja funktsiooni 

maksimaalse väärtuse antud lõigul [a; b]. Funktsiooni 

parameetriteks on lõigu otsapunktid a ja b ning jaotiste arv n. 

Maksimumi algväärtuseks võetakse funktsiooni väärtus vasakus 

otsapunktis a. For-lauses arvutatakse järjest funktsiooni väärtusi 

jaotuspunktides. Iga leitud y väärtust võrreldakse maksimumi 

jooksva väärtusega (maks) ja kui y väärtus on suurem, võetakse 

see uueks maksimumi väärtuseks. Viimane muutujale maks 

omistatud väärtus ongi tulemuseks ja see omistatakse 

funktsiooni nimele. 

 

For Each-lause 

For Each-lause võimaldab täita mingid tegevused antud kogumi iga elemendi jaoks. Kogumiks võib olla 

Exceli objektide kollektsioon (lahtriplokk, töölehed, kujundid jne) või massiiv.  

For Each element In kogum 

   laused_1 

   [Exit For] 

   laused_2 

Next element 

 

kordus Iga elemendi jaoks kogumis 

    tegevused 

lõpp kordus 

 

Kogumi element esitatakse objektimuutuja või Variant-tüüpi muutuja abil. Lausete seas võib olla üks või 

mitu Exit For-lauset (reeglina mingi valikulause sees), mis võimaldavad katkestada korduse ja anda täitmis-

järje järgmisele lausele enne kordamiste nö normaalset lõppu. Nii korduse kui ka protseduuri või prog-

rammi lõpetamiseks võib kasutada ka lauseid Exit Sub, Exit Function või End. 

Täitmine: For Each- ja Next-osalausete vahel olevaid lauseid täidetakse iga kogumi elemendi korral, kui 

mõni lause nagu Exit For, End vmt, ei katkesta täitmist. 

Näide:  Lahtriploki elementide summa  

Funktsioon SumP leiab etteantud piirkonna (lahtriploki) lahtrite 

väärtuste summa. Piirkond esitatakse parameetri prk abil, mille tüübiks 

on Range.  

Funktsioonis on määratletud kaks muutujat:  

lahter – Range-tüüpi objektimuutuja viitamiseks piirkonna igale lahtrile,  

S – lihtmuutuja summa jooksva ja lõppväärtuse salvestamiseks.  

Summa kogumiseks kasutatakse For Each-lauset. Igal kordamisel 

liidetakse järgmise lahtri väärtus summale – muutuja S väärtusele.  

tegevused

* iga elemendi jaoks

Function SumP(prk As Range) 

   Dim lahter As Range, S 

   S = 0 

   For Each lahter In piirkond 

       S = S + lahter.Value 

   Next lahter 

   SumP = S 

End Function 

 Function Fmax_1(a, b, n) 

   Dim h, x, y, maks, i 

   h = (b - a) / n 

   maks = F(a) 

   For i = 1 To n 

      x = a + i * h 

      y = F(x) 

      If y > maks Then maks = y 

    Next i 

    Fmax_1 = maks 

End Function 

Eespool toodud funktsiooni võib esitada ka lihtsamalt, saavutades samal ajal suurema universaalsuse. 

Funktsioon SumPM võimaldab leida nii Exceli töölehe 

piirkonna kui ka VB massiivi elementide summa.  

Siin on parameetri kogum tüübiks Variant. Sellele vastavaks 

argumendiks võib olla nii Exceli töölehe lahtriplokk kui ka VB 

massiiv. Samuti on siin Variant-tüüpi ka muutuja element. 

Lauses S = S + element on jäetud ära omadus Value. See ei 

olnud vajalik ka eelmises funktsioonis, kuna lahtriploki jaoks 

võetakse see vaikimisi.  

Näide: Maksimaalne element piirkonnas või massiivis 

Funktsioon MaxPM võimaldab leida maksimaalse väärtuse 

Exceli lahtriplokis või VBA massiivis. 

 Parameeter kogum on Variant-tüüpi. Sama tüüpi on ka 

muutuja elem, mida kasutatakse viitamiseks piirkonna või 

massiivi elementidele, ja muutuja maks, mida kasutatakse 

maksimaalse elemendi jooksva ja lõppväärtuse 

salvestamiseks.  Maksimumi maks algväärtuseks võetakse 

siin väga väike arv (-1020), mis on kindlasti väiksem kõigist 

väärtustest massiivis. Alternatiiviks on võtta maksimumi 

algväärtuseks kogumi esimene element. 

For Each-lauses võrreldakse igat elementi elem maksimumi jooksva väärtusega maks ja kui mingi elemendi 

väärtus on sellest suurem, võetakse element uueks maksimumi väärtuseks: omistatakse muutujale maks, 

asendamaks eelmist väärtust. 

Näide: Väärtuse asukoha otsimine vektoris 

Funktsioon Otsi_Nr leiab otsitava väärtusega võrdse väärtuse järjenumbri antud vektoris Vek. Kui 

otsitavaga võrdset väärtust vektoris ei ole, tagastab funktsioon väärtuse 0.  

Funktsioonil on kaks parameetrit: x – otsitav väärtus, Vek – vektor, milles toimub otsimine 

Otsitava väärtuse järjenumbri k algväärtuseks võetakse 1. 

For Each-lause igal kordamisel võrreldakse jooksvat elementi 

elem otsitava väärtusega x.  

Kui need on võrdsed, võetakse k jooksev väärtus funktsiooni 

väärtuseks ja katkestatakse nii For Each-lause kui ka funk-

tsiooni täitmine.  

Vastupidisel juhul suurendatakse k väärtust ühe võrra ja 

jätkatakse kordamist. Kui kõik elemendid on läbi vaadatud ja 

ükski ei võrdunud otsitavaga, lõpeb kordamine ja tulemuseks 

võetakse 0. 

Function SumPM (kogum) 

   Dim element, S 

   S = 0 

   For Each element In kogum 

      S = S + element 

   Next element 

   SumPM = S 

End Function 

Function MaxPM(kogum) 

  Dim elem, maks 

  maks = -10 ^ 20 

  For Each elem In kogum 

     If elem > maks Then maks = elem 

 Next elem 

 MaxPM = maks 

End Function 

Function Otsi_Nr(x, Vek) 

  Dim k, elem 

  k = 1 

  For Each elem In Vek 

     If elem = x Then  

Otsi_Nr = k 

Exit Function 

     End If 

     k = k + 1 

 Next elem 

 Otsi_Nr = 0 

End Function  

Näide: Graafikaobjektide asukoha muutmine 

For Each-lause kasutamisel võib tekkida olukord, kus lauses määratavaid tegevusi ei pea täitma kõikide 

antud kollektsiooni liikmetega. Eriti tüüpiline on see graafikaobjektide kollektsiooni (klassi) Shapes puhul, 

sest sellesse kollektsiooni kuuluvad kõik töölehe pinnal paiknevat objektid nagu diagrammid, joonised, 

pildid, ohjurid (sh käsunupud) ja isegi lahtrite kommentaarid.  

Vajalike objektide valimiseks võib kasutada erinevaid tingimusi:  

 objektid asuvad mingis kindlaks alas (piirkonnas), 

 kuuluvad kindlasse liiki (joonis, pilt, ohjur vms), 

 nende nimed algavad mingite kindlate märkidega vms. 

Makros Liikuge eeldatakse, et töölehe kindlas piirkonnas, ristkülikus nimega plats, asub mingi hulk suvalist 

tüüpi graafilisi kujundeid. Nende sees on ka ovaalid või ringid, mille nimed algavad tekstiga „Oval“. 

Makro Liikuge muudab juhuarvude abil 

ainult ovaalide asukohta. 

 For Each-lause on paigutatud lõpmatut 

kordust määrava Do...Loop-lause sisse. 

Programmi töö katkestamiseks on protse-

duur, mis sisaldab End-lauset. 

 For Each-lause määrab korduse üle kõikide 

kujundite töölehel – kollektsioon Shapes. 

Sisemine If-lause valib aga ainult need, mis 

asuvad kujundi plats sees ja mille nimede 

alguses on „Oval“. Liigutatavate kujundite 

asukoht määratakse juhuslike arvudega, 

arvestades ka nende mõõtmeid, nii et need 

jääksid platsi piiridesse. 

Sub Liikuge() 

   Dim kuju As Shape, plats As Shape, x, y 

   Set plats = Shapes("plats") 

   Do 

      For Each kuju In Shapes 

         If On_Sees(kuju, plats) And _ 

            Left(kuju.Name, 4) = "Oval" Then 

            x = plats.Left + Rnd() * (plats.Width - kuju.Width) 

            y = plats.Top + Rnd() * (plats.Height - kuju.Height) 

            kuju.Left = x: kuju.Top = y 

            paus 0.1 

         End If 

      Next kuju 

   Loop 

End Sub 

Massiivid 

Järgnevalt vaadeldakse massiivide olemust ja põhiomadusi, massiivide deklareerimist ning viitamist massiivi 

elementidele. 

Massiivide olemus ja põhiomadused 

Massiiv ehk massiivmuutuja on ühetüübiliste elementide (väärtuste) järjestatud kogum. Massiiv tähis-

tatakse ühe nimega. Pöördumiseks tema üksikelemendi poole kasutatakse nime koos indeksitega, mis 

näitavad elemendi asukohta massiivis. Massiivi igale elemendile eraldatakse arvuti mälus eraldi väli (pesa). 

Massiivi põhiomadused on:  

 nimi 

 dimensioon 

 indeksite rajad 

 elementide tüüp 

 dünaamilisus 

Massiivi nime esitamiseks kehtivad samad reeglid nagu teiste nimede jaoks. 

Dimensioon ehk mõõt iseloomustab massiivi organisatsiooni  – määrab elementide paigutuse kogumis ning 

indeksite arvu, mis on vajalik elementide asukoha määramiseks. Ühemõõtmeline massiiv on näiteks 

nummerdatud elementide jada, kus elemendi asukoha jadas saab määrata ühe indeksi (järjenumbri) abil. 

Sellist massiivi nimetatakse sageli ka vektoriks või jadaks. Selle analoogiks Excelis on töölehe rida või veerg 

või nende osad. Scratchis vastab sellele loend. Kahemõõtmelises massiivis, mida nimetatakse ka 

maatriksiks, moodustavad elemendid tabeli. Elemendi asukoht on üheselt määratav kahe indeksi abil, 

milleks on  rea- ja veerunumber. Excelis vastab kahemõõtmelisele massiivile tööleht või ristkülikukujuline 

lahtriplokk. Võib kasutada ka kolme ja enama dimensiooniga massiive. Maksimaalne dimensioonide arv 

VBAs võib olla 60! 

Indeksite rajad näitavad iga dimensiooni jaoks indeksi minimaalse ja maksimaalse väärtuse. Indeksi 

minimaalseks väärtuseks võib põhimõtteliselt olla suvaline arv, kuid enamasti on selleks 0 või 1. Indeksi 

maksimaalne väärtus ei ole piiratud.  

Elementide tüüp määrab nende esitusviisi ehk vormingu. Formaalselt peab massiivi kõigil elementidel 

olema ühesugune tüüp. Kui selleks on Variant, võib ühes massiivis säilitada erinevat liiki väärtusi – arve, 

tekste, kuupäevi jms. 

Dünaamilisus. Selle omaduse järgi jagunevad massiivid kahte liiki: 

 fikseeritud massiivid 

 dünaamilised massiivid 

Fikseeritud massiivile eraldatakse koht (pesad) mälus enne selle protseduuri täitmist, kus massiiv on 

määratletud, ja selle rajad peavad olema määratud konstantide abil. Dünaamilisele massiivile eraldatakse 

mälu protseduuri täitmise ajal ning massiivi rajade määramiseks võib kasutada muutujaid ja avaldisi.  

Enamasti kasutatakse dünaamilisi massiive, sest nende abil saab luua universaalsemaid ja paindlikumaid 

programme, mis ei sõltu massiivide mõõtmetest. 

Massiivide deklareerimine 

Massiivide omadused määratakse deklaratsiooniga. Kõik massiivi muutujad peavad tingimata olema 

deklareeritud. Deklareerimisviis sõltub teatud määral sellest, kas on tegemist fikseeritud või dünaamilise 

massiiviga.  

Fikseeritud massiive kirjeldatakse Dim-lausega: 

Dim nimi (rajad, rajad … ) As tüüp … 

Rajad võib deklaratsioonis esitada kujul min To max, kus min on dimensiooni indeksi minimaalne väärtus 

(alumine raja) ja max sama indeksi maksimaalne väärtus (ülemine raja). Need väärtused peavad olema 

esitatud täisarvuliste konstantide abil. Rajade arv näitab massiivi dimensiooni. Kui indeksi minimaalväärtust 

ei ole näidatud, võetakse selleks null. Korraldusega Option Base 1 mooduli alguses saab määrata, et 

vaikimisi võetav indeksite minimaalne väärtus moodulis on 1.  

Massiivi elementide tüüp määratakse samade võtmesõnade abil nagu lihtmuutujatel: Integer, Double vms. 

Võib kasutada ka tüübitunnuseid: %, # jne. Kui tüüp puudub, siis valitakse selleks Variant. 

Deklaratsioonide näiteid 

Dim X(100) As Long, Y(100) As Double, nimed(300), V(1 To 500)  

Lauses on deklareeritud neli ühemõõtmelist massiivi. Massiivides X ja Y on mõlemas 101 elementi  

(i = 0, 1, ... 100), elementide tüübiks on esimeses massiivis Long, teises Double. Massiivis nimed on 

301 elementi  ja massiivis V  500 elementi. Mõlema massiivi elementide tüübiks on Variant.  

Dim  T%(5, 8), A(1 To 20, 1 To 30) As Double, palgad(1 To 1000, 1 To 8)  

Selles lauses on deklareeritud kolm kahemõõtmelist massiivi. Massiivis T on 6 rida (rn = 0, 1, ... , 5) ja 

9 veergu (vn = 0, 1, ... , 8), elementide tüübiks on Integer. Massiivis A on 20 rida ja 30 veergu, elementide 

tüüp on Double. Massiivis palgad on 1000 rida ja 8 veergu. 

Dünaamilisi massiive võib deklareerida korduvalt. Alguses, kui massiivi mõõtmed ei ole veel teada, võib 

selle deklareerida Dim-lausega, milles dimensioon ja rajad ei ole määratud ning nime järel on tühjad sulud 

Dim Kor( ), V( ), W( ) 

Taolise deklaratsiooni alusel massiivile mälus kohta ei eraldata ja seda ei saa kasutada. Dünaamilise massiivi 

dimensiooni ja rajade määramine ning mälu eraldamine toimub ReDim-lausega  

ReDim nimi (rajad, rajad, …) As tüüp … 

Massiivi deklaratsiooni kuju on sama nagu Dim-lauses, ainult siin võib rajade määramisel kasutada 

muutujaid ja avaldisi. 

Näiteks määrab järgmine lause 

ReDim V(2 * k + 1), W(2 * k + 1), Kor(1 To m, 1 To n)  

massiivide V, W ja Kor mõõtmed ja eraldab neile mälu, kui eelnevalt on muutujatele k, m ja n omistatud 

väärtused. 

Viitamine massiivi elementidele ja massiividele 

Viitamine massiivi elementidele toimub kujul 

nimi (indeks, indeks, …) 

Selles on nimi on massiivi nimi ja indeks konstant, muutuja või avaldis, mis näitab vastava indeksi väärtust. 

Indeksite arv peab võrduma massiivi dimensiooniga ja indeksite väärtused peavad olema deklaratsioonis 

määratud rajade piirides.  

Näiteid eespool deklareeritud massiivide jaoks: 

X(0),  X(13),  Y(100),  X(i),  Y(2 * j - 1),  T(0, 0),  A(i, j), A(1, k + 2) 

Massiivi elementidele võib omistada väärtusi ja neid võib kasutada avaldistes analoogselt lihtmuutujatega. 

Näiteks leiavad laused 

For i = 0 To n 

   Y(i) = Sin(3 * X(i))^2 - Cos(X(i) / 2) + 2 * X(i) 

Next i  

funktsiooni y = sin2(3x) - cos(x/2) + 2x väärtused ja salvestavad need massiivi Y, kasutades muutuja väärtusi 

massiivis X. 

Sub Kopi(A(), B(), m, n) 

   Dim i, j 

   For i = 1 To m 

      For j = 1 To n 

          B(i, j) =  A(i, j) 

      Next j 

   Next i 

End Sub 

Protseduur Kopi omistab elementhaaval maatriksi A 

elemendid maatriksi B elementidele (omistab maatriksi A 

maatriksile B ehk kopeerib maatriksi A maatriksisse B). 

Protseduurile edastatakse parameetrite abil maatriksid ja 

nende mõõtmed: m – ridade arv, n – veergude arv. Kuna A ja 

B on kahemõõtmelised massiivid, kasutatakse viitamiseks 

nende elementidele kahte indeksit: i – rea number, j – veeru 

number. 

Massiivi elementidele saab viidata ka For Each-lause muutuja abil.  

Function Pos_Kesk(A) 

  Dim elem, S, n 

  S = 0: n = 0 

  For Each elem In A 

     If elem > 0 Then  S = S + elem : n = n + 1 

  Next element 

  If n = 0 Then   Pos_Kesk = 0 _ 

               Else Pos_Kesk = S / n 

End Function 

Funktsioon Pos_Kesk võimaldab leida suvalise massiivi 

positiivsete elementide aritmeetilise keskmise.  

Protseduuri parameeter on Variant-tüüpi ja pöördumisel võib 

sellele vastata suvalise dimensiooni ja mõõtmetega massiiv.  

Enam veel – seda funktsiooni saab kasutada ka töölehe 

lahtriploki summa leidmiseks. 

For Each-lauses saab massiivi elemente ainult kasutada (lugeda), kuid mitte muuta. 

Viitamiseks tervele massiivile kasutatakse massiivi nime indeksiteta, kuid seda saab teha ainult teatud 

erijuhtudel:  

 Parameetrites ja argumentides. Nimele võivad järgneda ka tühjad sulud.  

 Ühe massiivi omistamisel teisele. Massiivide dimensioonid, rajad ja tüübid peavad olema 

ühesugused. Näiteks ülaltoodud protseduuri asemel võiks kasutada hoopis ühte lihtsat 

omistamislauset: B = A 

 Massiivi omistamisel töölehe lahtriplokile (piirkonnale) või vastupidi. 

Näide:  Operatsioonid vektoritega 

Programmid Mas_1 ja Mas_2 teevad põhimõtteliselt seda sama – genereerivad funktsiooni Rnd() abil 

juhuslikest arvudest (1 kuni 100) kaks vektorit (ühemõõtmelist massiivi) X ja Y, liidavad elementhaaval need 

vektorid, saades tulemuseks vektori Z. Kõik kolm vektorit kirjutatakse töölehele, kasutades baaslahtrina 

lahtrit nimega V_alg. Lõpuks leiavad mõlemad programmid vektori Z elementide aritmeetilise keskmise ja 

kirjutavad selle töölehele. 

Väliselt erinevad programmid suhteliselt vähe, kuid sisuline erinevus on põhimõtteline. Esimeses 

programmis kasutatakse fikseeritud massiive, teises dünaamilisi. Programmis Mas_1 kasutatakse nii 

massiivide rajade määramisel kui ka For-lausetes juhtmuutuja i lõppväärtuse esitamisel nimega konstanti n, 

mille väärtuseks on võetud 10. Elementide arvu muutmiseks peab muutma programmi. Programmis Mas_2 

loetakse vektorite elementide arv n töölehelt. Seega võib igal täitmisel ette anda suvalise arvu elemente 

programmi muutmata. Vektorite kirjutamiseks kasutatakse selles programmis  kompaktsemat varianti.  

Sub Mas_1() 

' Fikseeritud massiivid 

Const n = 10 

Dim X(n), Y(n), Z(n), i, S, kesk 

 ' Vektorite genereerimine 

For i = 1 To n 

  Y(i) = Int(Rnd() * 100) + 1 

  X(i) = Int(Rnd() * 100) + 1 

Next i 

' Vektorite liitmine 

For i = 1 To n 

   Z(i) = Y(i) + X(i) 

Next i 

' Vektorite kirjutamine töölehele 

For i = 1 To n 

   Range("V_alg").Cells(i, 1) = X(i) 

   Range("V_alg").Cells(i, 2) = Y(i) 

   Range("V_alg").Cells(i, 3) = Z(i) 

Next i 

' Keskmise leidmine 

S = 0 

For i = 1 To n 

   S = S + Z(i) 

Next i 

kesk = S / n 

Range("keskmine") = kesk 

End Sub 

Sub Mas_2() 

' Dünaamilised massiivid 

 Dim i, n, S, kesk, prk  

 n = Range("elemente") 

 ReDim X(n), Y(n), Z(n) 

 ' Vektorite genereerimine 

 For i = 1 To n 

   Y(i) = Int(Rnd() * 100) + 1 

   X(i) = Int(Rnd() * 100) + 1 

 Next i 

 ' Vektorite liitmine 

 For i = 1 To n 

    Z(i) = Y(i) + X(i) 

 Next i 

 ' Vektorid töölehele 

 Set prk = Range("V_alg") 

 For i = 1 To n 

    prk(i, 1) = X(i) 

    prk(i, 2) = Y(i) 

    prk(i, 3) = Z(i) 

 Next i 

 ' Keskmise leidmine 

 S = 0 

 For i = 1 To n 

    S = S + Z(i) 

 Next i 

 kesk = S / n 

 Range("keskmine") = kesk 

End Sub 

Sub Mas_3() 

Dim n, kesk, prk As Range 

n = Range("elemente") 

ReDim X(n), Y(n), Z(n) 

Tee_Vek X(), n, 1, 100 

Tee_Vek Y(), n, 1, 100 

Liida_Vek X, Y, Z, n 

Set prk = Range("prk") 

prk.ClearContents 

Kir_Vek X, n, prk(1, 1) 

Kir_Vek Y, n, prk(1, 2) 

Kir_Vek Z, n, prk(1, 3) 

kesk = Kesk_Vek(Z, n) 

Range("keskmine") = kesk  

End Sub 

Sub Tee_Vek(V(), n, a, b) 

 Dim i 

 For i = 1 To n 

   V(i) = a+Int(Rnd()* (b - a + 1)) 

 Next i 

End Sub 

Sub Liida_Vek(V1, V2, V3, n) 

  Dim i 

  For i = 1 To n 

     V3(i) = V1(i) + V2(i) 

  Next i 

End Sub 

Function Kesk_Vek(V, n) 

  Dim i, S 

  For i = 1 To n 

    S = S + V(i) 

  Next i 

  Kesk_Vek = S / n 

End Function 

Sub Kir_Vek(V, n, prk) 

  Dim i 

  For i = 1 To n 

    prk(i) = V(i) 

  Next i 

End Sub 

Näide:  Maatriksi ridade aritmeetiline keskmine  

Sub Mas_3() 

  Dim m, n, i, j, S, prk As Range 

  m = Range("ridu") :  n = Range("veerge") 

  ReDim A(1 To m, 1 To n), KV(1 To m) 

  For i = 1 To m :  For j = 1 To n 

      A(i, j) = Int(Rnd() * 100) + 1 

  Next j  :  Next i 

   For i = 1 To m 

     S = 0 

     For j = 1 To n : S = S + A(i, j) :  Next j 

     KV(i) = S / n 

   Next i 

   Range("M_alg").Resize(m, n).Value = A 

   For i = 1 To m  

      Range("M_alg").Cells(i, n + 2) = KV(i) 

   Next i 

End Sub 

Programm Mas_3 leiab ristkülikmaatriksi A iga rea 

elementide aritmeetilise keskmise ja salvestab saadud 

keskmised vektoris KV. Kasutatakse dünaamilisi massiive. 

Maatriksi ridade ja veergude arv loetakse töölehelt.  

Massiivide deklareerimisel on ReDim-lauses alumised rajad 

esitatud ilmutatud kujul. Andmete kirjutamisel töölehele 

määratakse maatriksi ja vektori asukoht töölehel ühe lahtri 

M_alg abil. 

Lausega  

Range("M_alg").Resize(m, n).Value = A  

omistatakse kahemõõtelise massiivi A väärtused sobivate 

mõõtmetega lahtriplokile kasutades lahtriploki omadust 

Resize(m, n).  

Vektori asukoht töölehel määratakse samuti lahtri M_alg 

suhtes. Vektor kirjutatakse töölehe veergu, mis asub n + 2 

lahtri võrra sellest lahtrist paremal. Vektori esimene element 

kirjutatakse töölehe samale reale, millel on lahter M_alg.  

Massiivide kasutamine parameetrite ja argumentidena 

Massiivide kasutamist parameetrina esines ka eespool olnud näidetes, kuid seal ei peatutud nende 

esitamise reeglitel ega vaadeldud massiivide kasutamist argumentidena. Käesolevas jaotises esitatakse 

ülevaade nendest küsimustest. 

Parameetermassiiv deklareeritakse protseduuri päises järgmiselt: nimi [( )]  As tüüp  

Parameetermassiivi nimele järgnevad tühjad sulud. Juhul kui parameeter on Variant-tüüpi, võib sulud ära 

jätta. Dimensiooni ja rajasid siin näidata ei tohi.  

Järgnevalt on esitatud protseduuride päised 

Sub Mat_Vek (A(), B(), C(), m, n)  ja Sub Mat_Vek (A, B, C, m, n)  

on põhimõtteliselt samaväärsed. Mõlemal juhul on parameetrite tüübiks Variant, kuid teisel juhul ei ole 

näha, et A, B ja C on massiivid. Seda võib näha protseduuris, kust toimub pöördumine antud protseduuri 

poole. Muide, parameetermassiivi deklaratsioonist ei ole näha ka seda, kas tegemist on fikseeritud või 

dünaamilise massiiviga. Ka see pannakse paika protseduuris, kust toimub pöördumine. 

Parameetermassiivile vastavaks argumendiks peab olema massiiv. Argumentide loetelus võivad selle nime 

järel olla tühjad sulud sõltumata tüübist, kuid need ei ole kohustuslikud. 

Toodud protseduuri fragmendis on näha andmete 

deklareerimine ja pöördumine alamprotseduuri 

poole. Massiivid T, V1 ja V2 on dünaamilised – need 

on deklareeritud ReDim-lausega. Massiivide rajad on 

määratud muutujatega r ja v, mille väärtused loetakse 

eelnevalt töölehelt. Protseduuris on näidatud 

võimalikud pöördumised alamprotseduuri Mat_Vek 

poole, mille päis on esitatud ülalpool. Teine variant on 

esitatud kommentaarina. Pöördumistes kasutatakse 

Sub Demo_Mas() 

   Dim r, v 

   r = Range(”ridu”) : v = Range(”veerge”) 

   ReDim T(1 To r, 1 To v), V1(1 To v), V2(1 To r) 

   ... 

   Mat_Vek T(), V1(), V2(), r, v    

  ’ Mat_Vek T, V1, V2, r, v 

    ... 

End Sub 

Variant-tüüpi argumentmassiive T, V1 ja V2, mis vastavad parameetritele A, B ja C. Argumendid on 

deklareeritud peaprotseduuris. Tühjade sulgude kasutamine argumentides ei ole kohustuslik.  

Protseduuri päises Sub Mat_Vek1 (A#(), B#(), C#(), m, n) on massiivide (elementide) tüübiks määratud 

reaalarvud – Double. Sellisel juhul peavad massiivide nimedele tingimata järgnema tühjad sulud. Kuna 

argumentide tüübid peavad vastama parameetritele, peab protseduuris, kust pöördumine toimub, olema 

massiivid deklareeritud samade tüüpidega, näiteks nii: ReDim T#(1 To r, 1 To v), V1#(1 To v), V2#(1 To r). 

Pöördumisel võivad sulud olla, kuid need ei ole kohustuslikud: 

Mat_Vek T(), V1(), V2(), r, v   või  Mat_Vek T, V1, V2, r, v 

Mitmeprotseduurilistes programmides deklareeritakse massiivid, mida kasutatakse erinevates alam-

protseduurides, reeglina peaprotseduuris, sest selle tööpiirkond eksisteerib programmi töö lõpuni. 

Alamprotseduuris deklareeritud massiivid eemaldatakse peale selle töö lõppu. 

Massiivi kasutamisel parameetritena tekib üsna sageli vajadus kasutada selle rajasid – indeksite 

minimaalseid ja maksimaalseid väärtusi erinevate dimensioonide jaoks. Seda läheb vaja näiteks korduste 

kirjeldamiseks. Rajade kindlakstegemiseks kasutatakse kahte viisi: 

 alamprotseduur teeb ise kindlaks massiivide vajalikud rajad, 

 rajad esitatakse antud protseduuris parameetrina, nende tegelikud väärtused edastab kutsuv 

protseduur argumentide abil. 

Massiivide rajade leidmiseks on VBAs vastavad funktsioonid: 

LBound(massivi_nimi [, dimensioon]) ja UBound(massivi_nimi [, dimensioon]). 

Funktsioon LBound leiab indeksi minimaalse väärtuse antud dimensiooni jaoks, UBound – maksimaalse. 

Dimensiooni number esitatakse teise argumendi abil. Kui argument puudub, loetakse selle väärtuseks 1. 

Vaatleme võrdluseks kahte funktsiooni vektoris (ühemõõtmelises massiivis) minimaalse väärtuse 

leidmiseks. Esimeses funktsioonis on üks parameeter – vektor V. Kasutades ülalpool vaadeldud funktsioone, 

leitakse massiivi rajad n1 ja n2, mida kasutatakse For-lauses juhtmuutuja i alg- ja lõppväärtuse 

määramiseks. 

Function MinV_1(V()) 

  Dim i, n1, n2, mini 

  n1 = LBound(V) :   n2 = UBound(V) 

  mini = V(n1) 

  For i = n1 + 1 To n2 

     If V(i) < mini Then mini = V(i) 

  Next i 

  MinV_1 = mini 

End Function 

Function MinV_2(V(), n1, n2) 

  Dim i, mini 

  mini = V(n1) 

  For i = n1 + 1 To n2 

     If V(i) < mini Then mini = V(i) 

  Next i 

  MinV_2 = mini 

End Function 

Function MinV_3(V(), n) 

  Dim i, mini 

  mini = V(1) 

  For i = 2 To n 

     If V(i) < mini Then mini = V(i) 

  Next i 

  MinV_3 = mini 

End Function 

 

Teises funktsioonis on parameetrites lisaks vektorile näidatud ka 

indeksi minimaalne ja maksimaalne väärtus. Funktsioon MinV_2 on 

laiemate võimalustega kui MinV_1, sest seda saab kasutada vektori 

elementide suvalises vahemikus [n1; n2]. Praktikas on indeksi 

minimaalne väärtus enamasti 1. Funktsioon MinV_3 arvestab 

sellega. Siin on parameetriteks vektor V ja indeksi maksimaalne 

väärtus n (tavaliselt ka elementide arv vektoris). Indeksi minimaalne 

väärtus (1) on antud funktsioonis fikseeritud. 

Massiivide lugemine töölehelt ja kirjutamine töölehele 

Töötades Exceli keskkonnas sageli tekib vajadus lugeda massiivide elemente töölehelt ja kirjutada neid 

sinna. Allpool on toodud tüüpprotseduurid massiivide lugemiseks ja kirjutamiseks. Parameetrile prk võib 

vastata kas piirkond või selle esimene lahter. Eeldatakse, et indeksite minimaale väärtus on 1. 

Lugemine töölehelt Kirjutamine töölehele 

Sub Loe_Tab(A, m, n, prk) 

  Dim i, j 

  For i = 1 To m 

    For j = 1 To n 

        A(i, j) = prk(i, j) 

    Next j 

  Next i 

End Sub 

Sub Kir_Tab(A, m, n, prk) 

  Dim i, j 

  For i = 1 To m 

    For j = 1 To n 

      prk(i, j) = A(i, j) 

    Next j 

  Next i 

End Sub 

Sub Loe_Tulp(V, n, prk) 

  Dim i 

  For i = 1 To n 

    V(i) = prk(i, 1) 

  Next i 

End Sub 

Sub Kir_Tulp (V, n, prk) 

  Dim i 

  For i = 1 To n 

    prk(i, 1) = V(i) 

  Next i 

End Sub 

Sub Loe_Rivi(V, n, prk) 

  Dim i 

  For i = 1 To n 

    V(i) = prk(1, i) 

  Next i 

End Sub 

Sub Kir_Rivi(V, n, prk) 

  Dim i 

  For i = 1 To n 

    V(i) = prk(1, i) 

  Next i 

End Sub 

Viitamiseks töölehe lahtritele kasutatakse protseduurides konstruktsiooni prk.Cells(i, j) asemel 

kompaktsemat varianti prk(i, j). 

Ühemõõtmeliste massiivide lugemisel/kirjutamisel eristatakse nende paigutust töölehel, kusjuures rivi on 

horisontaalne, tulp vertikaalne. Paneme tähele indeksite kasutamist töölehe piirkonna jaoks. Formaalselt 

käsitletakse piirkonda kahemõõtmelisena. Juhul kui parameetrile prk seatakse vastavusse piirkond (mitte 

selle esimene lahter), võib kasutada nii tulba kui ka rivi jaoks samu protseduure, kus viitamiseks töölehe 

piirkonnale kasutatakse ühte indeksit.  

Sub Loe_Vek(V, n, prk) 

  Dim i 

  For i = 1 To n 

    V(i) = prk(i) 

  Next i 

End Sub 

Sub Kir_Vek(V, n, prk) 

  Dim i 

  For i = 1 To n 

    V(i) = prk(i) 

  Next i 

End Sub 



 

 


